
Re-Editor's Foreword
--------------------

As tribute to OSS this fully edited version is provided
to keep one of the best and fastest BASIC versions on
ATARI 8-bit computers accessible to all being interested
in this great piece of programming software.

BASIC XL was released by OSS in 1983.
OSS merged into a division of ICD, Inc. in 1988.
ICD sold their ATARI 8-bit product line to FTe in 1993.
FTe disappeared supposedly in 1996.
The author Bill Wilkinson passed away in November 2015.
The current status is that the Wilkinson Family gave
allowance to atariwiki.org to use OSS material.

Many of those famous A8 hardware and software products
from  OSS  and  ICD  are  still  available  thanks  to  a
worldwide active A8 community. So is this manual.

This learning book went through someone's scanner and
ended up as an already OCR-ed PDF in my e-mail. To get
it more handy and to have a chance to squeeze out the
bugs I found it worth to be re-edited as it is a good
example of how to get people attracted to make more use
out of their Atari computer.

The new book keeps the original layout for reference to
the  original  paperback  and  of  course  for  historical
reasons.  All  entries  from  the  original's  table  of
contents are found on the same page.

Typos and errors found were corrected.

BASIC XL and other OSS programming languages work best
together with SpartaDOS X V. 4.47 or newer provided by
DLT.

Enjoy learning to program in BASIC XL
                ... and may your A8 always be with you!

GoodByteXL, February 2022

P.S.: Special thanks for help to Roland.



Preface
-------

The computer age is upon us. From young children to
senior citizens, the computer and the need for "computer
literacy" have  caught the  attention and  curiosity of
many  people.  And,  for  better  or  worse,  computer
literacy  has  come  to  mean  "learning  to  program  in
BASIC". However, learning to program is no easy task.
Most  books  are  written  in  computerese.  You  have  to
understand computers in order to understand the books.

Not so with "30 Days to Understanding BASIC XL." While
avoiding computer  "jargon" as  much as  possible, this
book is intended to introduce you to the fundamental
concepts  of  programming  a  computer  using  the  BASIC
language.

BASIC (Beginner's All-purpose Symbolic Instruction Code)
is an introductory computer language. Because BASIC uses
many  common  English  words,  it  is  often  considered
the  easiest  of  computer  languages  for  the  beginning
programmer to learn.

Unfortunately, there are as many dialects of BASIC as
there are of English. While this book focuses on the
particular dialect known as BASIC XL, it nevertheless
attempts to teach concepts universal to most computer
languages.

Once you have mastered BASIC, you can then delve into
the other mysterious computer languages like C, Action,
and Pascal. For now, BASIC is the language you need to
understand in order to do elementary programs on your
ATARI Home Computer.

Contrary to what you may have read previously, it is our
belief that not everyone is intended to be a computer
programmer.  However,  even  if  you  do  not  become  a
programmer, you should become aware of computers and how
they  work.  Hopefully,  the  knowledge  you  gain  from
reading this book will make you feel more relaxed around
computers.

By  working  through  this  book  you  will  become  more
familiar with your ATARI Home Computer. In addition, you
will learn common BASIC statements. This knowledge will
permit you to use commercially produced products. You do
not have to be a programmer to use programs developed by
other people.

Finally, the intent of this book is to introduce you to
BASIC XL and beginning computer concepts. In the process
we hope to make you computer literate.

INTRODUCTION
------------

Before you begin working in "30 DAYS to UNDERSTANDING
BASIC  XL"  there  are  several  details  concerning  the
organization and the format of this book that you should
understand.

Chapters I through V are introductory chapters, and they
describe concepts  that must  be understood  before any
real programming techniques can be learned. Beginning
programming methods are covered in Chapters VI through
XV. Also, please note: in Chapters XII, XIII and XIV we
explain some equipment that can be used along with your
home computer. If you already own these devices, the
explanation  should  enhance  your  understanding  of  how
they  operate.  If  you  do  not  own  any  additional
equipment,  you  should  read  these  chapters  to  obtain
information on support products available to expand the
usage of your home computer. In Chapter XV we assume
you are using a color monitor or color television screen
as your computer display. If your ATARI Home Computer is
not attached to a color display, please work through
the chapter anyway. The only difference will be in the
lack of color.

In  Chapters  XVI  through  XXI  additional  beginning
concepts are explained. The last section of the book,
Chapters  XXII  through  XXVIII,  further  explores  the
graphics and sound capabilities of the ATARI computer.
The last chapters, Chapters XXIX and XXX, are a summary
of concepts covered and a final example. Our example is
a relatively easy video game which will indicate what
you can produce with a little creativity and the BASIC
commands covered in this book.

To make the learning process easier for you, we have
organized each chapter in the same manner. First, each
chapter begins with a glossary. Study this before you
start  reading  the  main  body  of  the  chapter.  The
glossary  provides  an  introduction  to  the  terms  and
concepts covered within the chapter.

Following  explanatory  paragraphs  are  sections  labeled
"Instruction". Please complete these sections. Usually
you will be asked to type commands or statements into
the computer. Instruction sections are intended to give
you hands-on experience or, in computer talk, give you
experience  "interfacing"  with  your  ATARI  computer
system.

At the end of each chapter are exercises. The answers
are  also  provided  at  the  end  of  the  book.  These

exercises are intended to provide you with additional
practice  and  to  integrate  new  concepts  with  those
previously explained.

In order to visually emphasize certain words and ideas
we  have  employed  several  different  print  styles  or
fonts. First, {braces} indicate a key on the computer's
keyboard. For example {RETURN} refers to the key on the
keyboard  and  also  means  "push  the  key  labeled
'RETURN'".

Sometimes  one  key  may  have  two  or  more  names  and
functions associated with it. In most cases we use only
one of the names to designate a particular key.

When necessary we use bold type to emphasize key words
and italics to give instructions or make commentary.

In some chapters we have enclosed additional information
in  boxes.  These  boxes  are  intended  to  provide  the
reader  with  supplementary  information.  Although
interesting and informative, these boxes are ancillary
to the explanations provided within the main body of the
text.

Finally, this book is to be used with your ATARI Home
Computer. Insert the OSS BASIC XL cartridge. Turn on the
computer  and  your  monitor.  If  you  do  not  understand
these instructions,  please first  refer to  your ATARI
Operators Manual. Let's begin.

CHAPTER I
---------

          GETTING TO KNOW YOUR COMPUTER:
                ALL THE CONCEPTS THAT ARE FIT TO PRINT

Glossary:
---------

Cursor       The  cursor  is  a  white  (or  light  blue)
------       square on the monitor which indicates where
             the next  letter or  number you  type will
             appear.

Keyboard     The area on your Atari Home Computer which
--------     Contains letters, numbers, and additional
             special keys.

{RETURN}     The return on your computer keyboard. By
--------     hitting  this  key,  you  tell  the  computer
             that  you  have  completed  entering  an
             instruction or an answer.

Error        A  mistake;  the  Atari  Home  Computer  will
             indicate  an  error  when  you  have  made  a
             typographical  mistake,  misplaced
             punctuation, misspelled a word, or given a
             command or statement in a form which the
             computer does not recognize.

Syntax       The  pattern,  structure  or  rules  which
             describe  the  language  the  computer
             understands.

{DELETE}     This key will "erase" any letter you type
             by accident; it moves the cursor one space
             to  the  left  each  time  you  hit  the  key,
             deleting the letter. symbol. or number it
             replaces.

{BACK S}     This means back space and is the same as
             {DELETE}  as  described  in  the  above
             definition.

Statement    Is the action part of a computer language
             just  as  a  verb  .is  the  action  part  of
             English.

PRINT        A common command that tells the computer
             you want to see something on the screen.

Character    Any number, letter, or special symbol on
---------    the keyboard.

String       Any  group  of  characters  inside  quotation
------       marks.

After  you  have  properly  connected  your  Atari  Home
Computer,  placed  the  BASIC  XL  cartridge  into  the
computer, and turned on the computer you will notice
that  the  monitor  displays  the  word  READY.  Located
directly under the R is a white square. This square is
the  cursor.  It  indicates  where  the  next  letter  or
 number  you  type  will  appear.  A  cursor  is  like  a
pointer, and it will move as you type, indicating where
the  next  character  will  appear  on  the  screen.  Your
 Atari  computer  keyboard  is  similar  to  an  electric
typewriter's keyboard. It has some extra keys and some
additional special features, but in general the letters
and numbers are in the same location.

First, locate the {RETURN} key. You will find it on the
right side, second row from the top. Press the {RETURN}
key. By hitting {RETURN} you are telling the computer
that you are there. Also, you are telling the computer
that  you  are  entering  or  have  entered  an  answer  or
information.  When  typing  on  the  computer's  keyboard,
usually nothing will happen until you press the {RETURN}
key!

Your computer monitor or TV screen should say READY.
(If it does not, check the Atari Operators Manual to be
sure  your  computer  is  connected  properly.)  If.  you
continue to press the {RETURN} key, the cursor will move
down the left hand margin. If you press the {SPACE BAR},
the cursor will move toward the right hand margin.

Instruction:
------------
You Type:
---------

        PINT
        Press the {RETURN} key

Your computer should say
        ERROR - PINT[]

([] is actually an inverse video space.)
This is a sample of a Syntax Error.

Syntax  refers  to  the  well-defined  set  of  rules  that
comprises or makes up a computer language. Actually any
language - including English - has such a set of rules;
try to understand this, proper without syntax the. Did
you have trouble understanding that? The computer isn't
as smart as you, so you have to be even more careful
about what you tell it!

In the example shown above, you have told the computer
something it does not understand. Whenever you make a
typographical error, misplace punctuation, or misspell a
word,  the  computer  will  display  the  line  you  typed
together with the word "ERROR". It will show you the
location where it thinks your error is by reversing the
colors of the character and background at that spot.

Please note: For now, if you make a typographical error,
simply retype the line. If you notice the error before
you have pushed {RETURN}, use the {BACK S} key. {BACK S}
means "back space" and is located above the {RETURN}
key. Use this key to back up to your error and then
retype the line from that character to the end of the
line.  The  complete  Atari  editing  system  will  be
discussed in Chapter 15.

        ---------------------------------------
        |  Notice  as  you  type information  |
        |  into   the   computer   and   the  |
        |  computer  responds,  there  is  a  |
        |  great  deal  of  data   on   your  |
        |  monitor or television screen.  If  |
        |  the   accumulation   of  data  is  |
        |  distracting,  press  the  {SYSTEM  |
        |  RESET}.   This  will   clear  the  |
        |  screen and the  word  READY  will  |
        |  reappear.                          |
        ---------------------------------------

Instruction:
------------
You Type:
---------

        PRINT
        Press the {RETURN} key.

A statement is the active part of any computer language.
Statements  translate  into  an  action  you  want  the
computer to perform. In BASIC there are many different
statements. PRINT is a common statement and indicates
that you would like to see some information on your
screen.

Your screen should look like this:

READY
PINT
ERROR PINT[]
PRINT

READY

You have just told the computer to PRINT nothing, and it
did exactly that; it PRINTed a blank line. If you want
to  see  something  on  the  screen  you  must  tell  the
computer to PRINT.

        ---------------------------------------
        |  Originally,     computers    were  |
        |  attached to printers; there  were  |
        |  no   monitors   or   TV  screens.  |
        |  Today, a  better  word  might  be  |
        |  "SHOW"  or  "DISPLAY".   However,  |
        |  we're still stuck with PRINT.      |
        ---------------------------------------

Instruction:
------------
You Type:
---------

        PRINT "HELLO"
        Press the {RETURN} key

Your screen should look like this:

        READY
        PRINT "HELLO"
        HELLO

        READY

Any words or numbers which you enclose in quotations
marks will be treated as a unit by the computer. Any
group of characters - that is, letters or numbers inside
quotation marks - is called a "string". When you tell
the computer to PRINT a string, it does exactly that.
Of course,  the computer'  doesn't PRINT  the quotation
marks. After all, if you told someone to say "HELLO",
you wouldn't expect them to say "QUOTATION-MARK HELLO
QUOTATION-MARK" , would you?

Being able to PRINT such quoted strings (which are also
sometimes called 'literal strings' or some other similar
name) is invaluable. Most dialogues with the computer
will involve several PRINTs of strings.

Before you begin our practice exercises, here are some
additional facts that might help you relax as well as
learn computer programming:

• Even  the  very  best  programmers  make  errors.
Errors are part of the learning process. Don't
let them upset you.

• As  you  type,  the  computer  will  automatically
continue or "wrap around" to the next display
line. Even if something you type in occupies
more than one display line, the computer treats
the line or lines as a unit. However, you may
not exceed more than three display lines without
starting a new PRINT statement.

• The only way the computer knows that you have
finished a thought or entry is by pressing the
{RETURN} key.

• If you press the {RETURN} key before you have
finished  a  thought--say  at  the  end  of  the
display line--the computer will execute what it
can. It may, however, consider what you have
typed  to  be  illegal  and  give  you  an  ERROR
message.

• It is normal for your computer to use UPPER-CASE
LETTERS. Lower case is discussed in our chapter
on editing.

• If you want to clear the screen of all confusing
characters, you may either turn your computer
off  and  then  back  on,  or  you  may  press  the
(SYSTEM RESET) key.

Exercise:
---------

1.  Tell the computer to PRINT your name.

2.  Tell the computer to PRINT a famous quotation like
    "GIVE ME LIBERTY OR GIVE ME DEATH".

Reminder: Don't forget to use quotation marks.

CHAPTER II
----------

                    ARITHMETIC:
                        COMPUTERS NEVER MAKE MISTAKES

Glossary:
---------

+ The symbol for addition which the computer
understands.

- The symbol for subtraction.

* The symbol for multiplication.

/ The symbol for division.

Using BASIC, a computer is capable of performing simple
arithmetic calculations such as: addition, subtraction,
multiplication, and division.

Instruction:
------------
You Type:
---------

PRINT 3+7

Press the {RETURN} key

Your screen should look like this:

READY

PRINT 3+7

10

READY

Instruction:
------------
You type:
---------

PRINT 9-5

Press the {RETURN} key

What does your screen look like?

Notice: it is neither necessary nor correct to hit the
equal sign key. The PRINT statement alone is sufficient
to  tells  the  computer  to  automatically  perform  the
calculation and display the results.

The symbol for multiplication is the asterisk (*). If
you  want  to  try  multiplying,  then  you  must  use  the
asterisk.

Instruction:
------------
You Type:
---------

PRINT 6*4

Press the {RETURN} key

The computer should reply with an answer of 24. But also
try typing these lines:

PRINT 6X4

PRINT (6) (4)

Even though such forms are used in Algebra and other
mathematics, they are not legal in BASIC. You must use
an asterisk to cause BASIC to multiply.

The symbol for division is the slash.

Type:  PRINT 24/2
       Press the {RETURN} key

The computer will respond with an answer of 12.

Instruction:
------------
You type:                             Computer Responds
---------                             -----------------

PRINT 72/9 {RETURN}                           8
PRINT 7*8 {RETURN}                            56
PRINT 2+4+6+8 {RETURN}                        20
PRINT 3*21 {RETURN}                           63
PRINT 3+4*2 {RETURN}                          11

Were you surprised by the last response? Did you expect
the  answer  to  be  14?  The  computer  normally  performs
arithmetic functions in left to right order, but all
multiplication  (*)  and/or  division  (/)  is  performed
before any  addition (+)  and/or subtraction  (-). Some
people call this the "MDAS rule" because the computer
executes  all  multiplication  and/or  division  before
executing addition and/or subtraction.

Instruction:
------------
You type:                             Computer Responds
---------                             -----------------

PRINT 850 {RETURN}                           850
PRINT "850" {RETURN}                         850
PRINT 2+6 (RETURN}                           8
PRINT "2+6" (RETURN}                         2+6

Does the last result surprise you?

Remember  a  string  is  any  group  of  characters  inside
quotation  marks.  When  confronted  with  a  string,  the
computer treats the string as a unit. Literally, what
goes in, comes out.

Instruction:
------------
You type:                             Computer Responds
---------                             -----------------

PRINT "5+9*7 CATS=1,000"(RETURN}      5+9*7 CATS=1,000
PRINT "XQR/%#$ (RETURN}               XQR/%#$
PRINT "ADV*2197@" {RETURN}            ADV*2197@
PRINT "800-421-0009"                  800-421-0009
PRINT 800-421-0009                    370
PRINT "999-38-0101"                   999-38-0101
PRINT 999-38-0101                     860

In  our  examples  above,  the  computer  performed  all
arithmetic  operations  it  confronted  EXCEPT  on  those
which were enclosed in quotation marks and just happened
to look  like operations:  Since anything  in quotation
marks is a string, even arithmetic expressions enclosed
with quotation marks become strings.

Notice it does not matter what you place in a string.
When told to reconstruct the string, the computer does
so exactly in the manner in which it was entered.

From now on, we won't usually show you when to press
{RETURN}.  Just  remember  to  press  the  key  when  you

finish an instruction to the computer or when you give
it  an  answer.  If  the  computer  appears  to  be  doing
nothing, press {RETURN}. In most cases, this will cause
the computer to do something and then give you control.

Exercises:
----------

1. Tell the computer to PRINT your name, address, and
phone number.

2. Write a PRINT statement that will allow the computer
to compute each of the following:

         a)  15 + 25

         b)  38 - 14

         c)  680 * 12

         d)  25 / 5

3. Write  a  PRINT  statement  that  will  display  your
social security number.

CHAPTER III
-----------

                 REMEMBERING NUMBERS:
                     VARY VERY IMPORTANT

Glossary:
---------

Variable     Anything which has the capacity to change.
--------     Examples include your age, the price of a
             stock, and the rate of speed at which you
             drive your car.

Numeric      A variable which contains only numbers.
Variable

LET          A  statement  which  assigns  a  value  to  a
             variable.

A variable is anything which is subject to change. The
weather is a variable because from day to day and season
to season, it is capable of changing. The prices of
food  and  gasoline  are  also  variables.  Given  certain
conditions, the price of any given item might change.
These are variables.

To better understand the concept of variables as used in
OSS BASIC XL, imagine a teacher standing at the door to
her  classroom  the  first  day  of  the  new  year  in  an
elementary school.

The first student arrives; she hands him a card with his
name on it and assigns him the first cubbyhole to store
his possessions. He tapes his name to the cubby, and
then he puts his jacket inside the cubby.

The  next  student  arrives  and  goes  through  the  same
procedure taking the second cubby, but she has a jacket,
notebook,  umbrella,  and  a  lunchbox  to  put  into  her
cubby.

As each student arrives, the next succeeding cubby is
assigned; each cubby is filled or not filled depending
on what the student has brought to school that day.
Anything which is placed in the cubby becomes associated
with that particular cubby hole, until something new is
added to the cubby.

In computer talk, a "value" is stored in a "variable".
Any child could have walked into class first and thus
been assigned  any "variable"  cubby. Also,  each child

could have brought any number of items which became the
"value" in the variable.

Each variable cubby is assigned a name. The name was
chosen in a meaningful manner, using the child's name
for  identification.  Computer  variables  function  much
like our "cubby" example.

Although there are different kinds of variables, each
numeric variable is represented by a name. In BASIC XL,
variable names always begin with a letter. A numeric
variable name may consist of one letter or a combination
of letters or numbers. Examples of legal variable names
are: N, X, STOCKPRICE, AC, or B3. There is no practical
limit (other than the line size) on the length of a
name. Some illegal variable names are: 1ABC, A%>, or
A]B.

Although a single letter is sufficient to identify a
variable, it is helpful to use variable names which are
more descriptive. For example, the single letter I is
sufficient  and  acceptable  as  the  name  of  a  variable
associated with income. However, it is just as easy and
more  meaningful  to  use  INCOME  as  the  name  of  the
variable for the value associated with income. As you do
more programming and as you use many variables within
the same program, the value of using meaningful names
for variables will become more apparent.

It is important to remember that each numeric variable
may hold a single number at any given time. Although the
value of a variable may change several times during the
course of a program, the variable can represent only one
value at a time.

In order to tell the computer you are going to use a
variable you must use a LET statement. A LET statement
assigns  a  value  to  a  variable.  The  form  of  a  LET
statement is:

      LET name of variable = value of variable

Instruction:
------------
You type:                             Computer Responds
---------                             -----------------

LET X = 14
PRINT X 14                                    14

LET X1 = 35
PRINT X1                                      35

LET APPLES = 90
PRINT APPLES                                  90

LET R = 9
PRINT R 9                                     9

In the above examples values were assigned to different
variables.  The  computer  stores  the  value  of  the
variable  in  its  memory.  When  given  additional
instructions, (e.g., PRINT) the computer recalled the
value of the variable and used it (e.g., PRINTed its
value).

Instruction:
------------
You type:                             Computer Responds
---------                             -----------------

LET A = 5
PRINT A                                       5

LET A = 7
PRINT A                                       7

Notice  in  the  examples  above  that  the  value  of  the
variable A changes from 5 to 7. Whenever the value of a
variable changes the computer remembers only the last
value. We could continue to change the value of our
variable A, but the computer would remember only the
most recent value and none of the previous ones.

You will recall from Chapter 2 that it was possible to
use more that one number in a PRINT statement as long as
the numbers were connected by an arithmetic operator.
The same is true for variables. A variable and a number
may be combined in a PRINT statement as long as they are
connected by an arithmetic operator.

Instruction:
------------
You type:
---------

PRINT A+1                                     8

(Remember, A had a value of 7, so the above statement is
equivalent to typing PRINT 7+1.) In addition, variables
may  be  used  in  conjunction  with  other  variables  and
numbers  as  long  as  they  are  used  in  an  arithmetic
expression using arithmetic operators.

LET B = A+1
PRINT B                                       8

Although we have named a new numeric variable B, the
computer still remembers the value of the variable A.
In the above example, we have asked the computer to add
1 to the value of A and to let that new value (7+1) be
the assigned as the value of the variable B.

Although A+1 = B is an acceptable expression in algebra,
it is illegal in BASIC XL. Remember, computers are not
as smart as people. To a computer B=A+l and A+l=B are
two very different expressions. The former is permitted
in a program; the latter will produce an error. The
statements  which  assign  a  value  to  a  variable,  MUST
follow this form:

        LET variable=expression.

Instruction:
------------
You type:                             Computer Responds
---------                             -----------------

LET A=B
PRINT A                                       8

Now, we are telling the computer to let the value of the
variable A be the same as the value of the variable B.
Notice that the computer remembers only the new value of
the  variable  Ar  the  old  value  of  A  is  completely
disregarded.

We  could  also  change  the  value  of  the  variable  in
another way.

Instruction:
------------
You Type:
---------

LET F = 15
LET X = F
PRINT X                                       15

The  computer  stores  the  value  of  a  variable  in  its
memory. Each time you change the value, the computer
also changes the value of the variable.

The really neat thing about a variable is its ability to
change. During the course of a program, the value of a
variable can change as many times as you would like.

The use of variables will become more apparent when we
discuss loops. But for now, remember that a variable is
anything that can change, and that the purpose of using
a variable is to put the burden of remembering on the
computer and not on your brain. For example, if you
wanted to keep track of the days in a year, how many are
left, and how many have already passed, you could keep
track  by  using  math  and  lots  of  paper.  However,  an
easier and quicker way would be to use the computer.

Exercise:
---------

1) What do the following LET statements do?

LET statements:

a) LET X = 15

b) LET B = 9 * 8 * 2

c) LET F2 = 111 * 3 + 5

d) LET Z = F2 - 15

2) Assign  the  sum  of  seven  and  five  to  a  variable
called  TOTAL.  Display  the  value  of  a  number  one
greater than that sum.

3) Try the following:

LET GROSSPAY = 40 * 8

LET DEDUCTIONS = 122

LET NETPAY = GROSSPAY - DEDUCTIONS

PRINT NETPAY

What does the computer display?

CHAPTER IV
----------

               DIRECT MODE VS. PROGRAMMING MODE:
                  THE BIRTH OF A PROGRAM

Glossary:
---------

Statement    Is an instruction to the computer. An example
---------    is PRINT as used in chapters 1, 2, and 3.

Direct       Allows the computer to respond immediately
Mode         to any instruction you give it.
------

Program      Is any valid combination of symbols, values
Line         and statements entered into the computer.
-------      It instructs the computer to do a task.

Program      Is a series of one or more statements which
             tells the computer exactly what task it is
             to perform.

Line         Is a one to five digit number entered at
Number       the beginning of a program line. Note that
             an  absence  of  a  line  number  implies  an
             instruction  which  is  executed  in  direct
             mode.

RUN          Is the command which tells the computer to
---          execute the program.

LIST         Is a command which tells the computer you
             want to see the program lines it has stored
             in its memory.

Up to this point, we have been using the computer in
"direct"  mode.  That  is  to  say,  when  we  gave  the
computer a command, we wanted the response immediately.
Examples of statements includes:

      PRINT "HELLO"
      PRINT 7*964.

A computer program is a series of instructions which
tells the computer to perform a sequence of tasks that
hopefully  will  produce  a  desired  output.  These
instructions are called statements. LET X = 15 is a LET
statement. As  stated earlier,  a LET  statement allows
you to assign a number value to a variable. These are
instructions for the computer.

It is important to tell the computer that you want to
give it a series of commands before it executes your
orders. Think about sending someone to the supermarket.
You need ten items: but as soon as you mention the first
one, the person runs out the door. He comes back five
minutes later with your one item. You explain to him
you need several products. This time he leaves after
you have mentioned three items. When he returns, you
are still missing six items. It would have been more
efficient if he had waited for the entire list in the
beginning.

The efficient way in which to tell the computer you have
a series of commands for it to execute is by assigning a
line number to each statement (command). When you type
10 or 20 or 30 before a statement, it tells the computer
that more commands might follow.

By using line numbers, you take the computer out of
direct mode and put it into programming mode. You do
not have to turn any knobs or change anything with the
computer: you just have to type a line number before
each statement. Each line of a BASIC program must have
a  number.  The  computer  executes  statement  lines  in
numerical order, regardless of the order in which they
were typed into the machine. Even if you type in line
30 first and then type in line 10, the computer will act
on line 10 first. In addition, you should know that it
is often customary - but certainly not necessary - to
number lines in increments of 10.

One reason for numbering program lines in increments of
10  is  to  allow  for  insertions.  If  you  begin  your
program with the line number 10 and continue to number
each line in increments of 10, you will have enough room
to  add  a  line  anywhere  in  the  program.  Perhaps  you
forgot a line in the beginning; your new line could be
assigned line number 5 or 6. Remember, the order in
which  you  enter  a  line  is  not  important.  What  is
important to the computer is the number of the line.

Once you begin to use line numbers, the computer will
not execute the lines with numbers until you type RUN.
The  command  RUN  indicates  to  the  computer  to  begin
processing the statements you have entered. Remember,
it normally processes the lines in numerical order.

Instruction:
------------
You type:                             Computer Responds
---------                             -----------------

PRINT "PROGRAMMING IS FUN"            PROGRAMMING IS FUN

Now type:
---------

10 PRINT "PROGRAMMING IS FUN"

Nothing happens until you also type RUN!

So now type RUN and press the {RETURN} key.

                                      PROGRAMMING IS FUN

Instruction:
------------
You type:                             Computer Responds
---------                             -----------------

10 LET A = 1
20 PRINT A
30 LET B = A + 1
40 PRINT B
50 LET C = B + 1
60 PRINT C
RUN
                                              1
                                              2
                                              3
                                              READY

As mentioned earlier the computer will act on the lines
in line number order, not necessarily the order in which
you  entered  the  lines.  Once  you  start  using  line
numbers, you are out of direct mode and in programming
mode.

For example, sometimes as you are programming you may
forget a line.

Instruction:
------------
You type:                             Computer Responds
---------                             -----------------

10 LET R = 1
20 PRINT R
30 LET S = l0
40 LET T = 15
50 PRINT T
60 LET V = R + T
70 PRINT V
RUN

                                              1
                                              15
                                              16
                                              READY

Note  that  nothing  happened  with  S!  So  let's  insert
another line.

You Type:
---------
35 PRINT S

Now  your  program  is  out  of  order  on  the  screen.
However, if you were to RUN the program, the computer
would PRINT:

1
10
15
16

READY

In order to get the computer to show you the program in
the correct order, type LIST and hit the {RETURN} key.

You Type:
---------
LIST {RETURN}                         Computer Responds
                                      -----------------

                                          10 LET R=1
                                          20 PRINT R
                                          30 LET S=l0
                                          35 PRINT S
                                          40 LET T=15
                                          50 PRINT T
                                          60 LET V=R+T
                                          70 PRINT V

Now your program is in the correct numerical sequence on
the screen.

Remember,  the  order  of  entry  of  the  lines  does  not
matter to the computer. Once again, the computer will
always act on the program in line number order, not the
order in which the lines were typed into the machine.

Before you begin the exercises below, please turn off
your computer.  This will  erase the  computers memory.
The next chapter will explain more about clearing the
screen and the computers' memory.

Exercises:
----------

1) Write  a  program  which  assigns  a  value  to  two
variables. Then have it PRINT out the sum of the
two variables and their product. Be sure to LIST
and RUN your program.

2) Turn your computer off and then on again and type in
the  same  program  (as  in  exercise  1)  again.  This
time, though, leave out the line which assigns a
value  to  the  second  variable.  LIST  and  RUN  the
program.  What  value  does  the  computer  assume  for
the unassigned variable?

3) Without turning off the computer reinsert the line
which  assigns  a  value  to  the  second  variable.
Remember this line should now be the second line of
your program. What happens when you LIST and RUN?

--27--



CHAPTER V
---------

                A NEW BEGINNING:
                     WIPING THE SLATE CLEAN

Glossary:
---------

NEW          A command Which tells the computer to erase
---          the portion of its memory that stores BASIC
             programs.

{CLEAR}      A key that when pressed at the same time as
-------      the shift key will erase the screen BUT NOT
             THE MEMORY.

{SYSTEM      A  key  that  when  pressed  will  clear  the
 RESET}      screen  and  place  the  computer  back  in
-------      direct mode.

;            Punctuation for the PRINT statement which
–-           tells the  computer to  PRINT more  on the
             same line instead of going to the beginning
             of the next line.

,            Allows  the  programmer  to  PRINT  words  or
–-           numbers in columns.

All the information you have typed into your Atari Home
Computer in program mode is stored in its memory. In a
program each line number must be distinctive. No two
program lines can share the same number. If you use the
same program line number, the computer remembers only
the one which was most recently entered. To see how
this works, try this program:

        10 PRINT "THIS IS THE ORIGINAL LINE 10"
        10 PRINT "THIS IS THE NEW LINE 10"
        RUN

The computer should respond:

      THIS IS THE NEW LINE 10

      READY

If you have been working with this book and have not
turned off your computer between Chapters IV and V, you
may have found additional information PRINTed on your
screen. If so, don't worry about it. That is what this
chapter is all about.

Once you type in a program, the computer will normally
remember  it  until  you  turn  the  power  off.  A  more
convenient  way  in  which  you  tell  the  computer  to
"forget" everything it has stored in its memory is by
typing the command NEW.

When you begin a new program, it is a good idea to first
type NEW.  This command  erases the  computer's memory.
By typing NEW you avoid having your new program mixed in
with your old program. Try this example:

Instruction:
------------
You type:
---------

NEW
10 PRINT "Y"
20 PRINT "E"
30 PRINT "L"
40 PRINT "L"
50 PRINT "O"
60 PRINT "W"
RUN

Does the computer do what you expected?

Now type:
---------

10 PRINT "B"
20 PRINT "L"
30 PRINT "U"
40 PRINT "E"
RUN

Notice that O W is PRINTed after B L U E. Why did that
happen? The extra letters were PRINTed because lines 50
and 60 from the last program are still in the computer's
memory. There are two things you can do to avoid this
problem. First, you could simply type "NEW" and press
{RETURN}  into  the  machine  and  then  retype  lines  10
through 40. Remember, though, once you type "NEW" you
can NOT get a program back. Turning off your computer
or  opening  the  cartridge  door  will  also  erase  your
program from the computer's memory.

However, you CAN save the program and just eliminate the
excess lines. To do this, just type the line number you
want to delete and press {RETURN}. In the example above
to get rid of the O W following B L U E, type,

        50 {RETURN}
        60 {RETURN}

Now type LIST and RUN. Notice that lines 50 and 60 are
no longer in the computer's memory and the O and W are
not PRINTed.

Also,  please  note,  there  is  a  difference  between
clearing the screen and clearing the computer's memory.
To clear the memory type "NEW." To clear the screen,
press the {CLEAR} key while holding down the {SHIFT}
key. Or, you can also clear the screen by pushing the
{SYSTEM RESET} key.

CAUTION! If a program is RUNning when {SYSTEM RESET} is
pushed, the program will halt but will remain intact.

In  addition  to  clearing  the  screen,  there  are  other
things that you can do to change the manner in which
items  appear  on  the  screen.  A  PRINT  statement
automatically  returns  the  computer  to  the  left-hand
margin. You can suppress the return by putting a comma
or  a  semicolon  after  the  last  value  in  the  PRINT
statement.

For example:
10 PRINT 8*3;
20 PRINT 956,
30 PRINT "CARS"

The only difference between the comma and the semicolon
is the semicolon tells the computer to stay on the same
line while the comma tells the computer to stay on the
same  line  and  move  to  the  next  columnar  position.
(Columnar positions are generally 10 characters apart,
but this can be changed by the user and is described in
the OSS BASIC XL Reference Manual).

Remember, if you want information PRINTed adjacent and
on the same line, use a semicolon(:). However, if you
want information PRINTed in columns on the same line,
use a comma.

Instruction:
------------
You type:                         Computer Responds
---------                         -----------------
NEW
10 PRINT 10*7,
20 PRINT 5+2;
30 PRINT "SEE"
RUN

                                  70         7SEE

It is also possible to have the computer display more
than one item using a single PRINT statement. The items
to be displayed may be separated by either a comma or a
semicolon. As you might expect, the comma will cause
the items to be PRINTed in columnar form. The semicolon
will cause the items to be PRINTed adjacent to each
other.

Instruction:
------------
You type:                         Computer Responds
---------                         -----------------

NEW

10 LET A = 12
20 LET B = A + 3
30 LET A = B
40 PRINT "THE VALUE OF A IS  " ; A
RUN

                                  THE VALUE OF A IS 15

You type:                   Computer Responds
---------                   -----------------

Now change line 40 to:

40 PRINT "The Value of A is" , A
RUN

                            THE VALUE OF A IS         15

Notice the difference a single punctuation mark makes.
Although  not  as  smart  as  people,  computers  are  very
consistent.  In  BASIC  XL,  the  computer  will  always
respond to statements using semicolons and commas in the
same way.

Exercises:
----------

1. Write a program which will PRINT the values of two
different  variables.  Have  the  computer  PRINT  the
values adjacent to each other on the screen.

Suggestion: Once you enter the answer to exercise 1, you
need only to change one line. The lines in which you
set the value of your variables may be reused for all
three exercises.

2. Modify the above program so that the two values will
be PRINTed about 10 spaces apart.

3. Modify the above program so that the sum (+), the
difference {-l and the product (*) of the values of
the two variables are printed on the same line.

CHAPTER VI
----------

                       REPETITION:
                           GETTING LOOPED

Glossary:
---------

GOTO         A  statement  which  tells  the  computer  to
----         jump or branch to a specified line number
             and then continues to follow the program
             instructions  in  the  usual  line  number
             order.

Loop         The repetition of one or more statements.
----

{BREAK}      A  key  which  when  pressed  stops  whatever
-------      the computer was doing.

Initializing Giving the first or the initial value to a
a variable   variable.
------------

Counter      Keeps  track  of  or  counts  the  number  of
-------      times a loop has been executed.

X=X + 1      A non algebraic expression that allows a
-------      variable to act as a counter.

The commands or statements we have discussed so far have
permitted  the  computer  to  work  in  a  "straight  line"
sequence. First do line 10, then line 20, etc. In order
to get the computer to repeat a process, you need to use
a GOTO statement. This statement must be followed by a
line number, and 'this instruction tells the computer to
jump or branch to the line number following the word
GOTO. For example:

10 PRINT "My Name is Sally"
20 GOTO 10

The computer will PRINT continuously MY NAME IS SALLY.
The computer executes line 10 then reads line 20. It
goes back to 10 then back to line 20 etc. The result is
a "loop".

A loop is the repetition of a sequence of instructions.
An  "endless  loop"  is  one  in  which  the  computer
continuously  loops  back  through  the  instructions  and
never ends. To stop an endless loop press the {BREAK} or
{SYSTEM RESET} key.

A GOTO statement is VERY useful; it allows the computer
to execute the same command, or set of commands, under
various circumstances.

      -----------------------------------------
      |   A  computer language purist might   |
      |   tell you  not  to  use  the  GOTO   |
      |   statement.  It is true that there   |
      |   are  other  commands  in BASIC XL   |
      |   which produce the same results as   |
      |   GOTO (but in what is considered a   |
      |   more  structured  and   "elegant"   |
      |   manner).   However,  these  other   |
      |   commands are  more  advanced  and   |
      |   require more background knowledge   |
      |   in   programming   than  we  have   |
      |   presented up to this time.          |
      -----------------------------------------

Instruction:
------------
You type:                         Computer Responds
---------                         -----------------

NEW
10 PRINT "N", "N-Squared"
20 LET N = 1
30 PRINT N, N*N
RUN

                                  N         N-SQUARED
                                  1         1
                                  READY

In analyzing this program, line 10 should not have been
a problem. We simply asked the computer to PRINT or
write on the screen "N" and, about 10 spaces later,
"N-SQUARED".

In Line 20, we have set a value for our variable N.
This is called initializing or giving a starting value
to a variable. Notice that this statement is executed
once.

Again, in Line 30, we have asked the computer to PRINT
or show the value of "N" and the value of "N*N" on the
screen.  Since  we  have  assigned  a  value  to  "N",  the
computer can execute these 3 lines as a program.

Instruction:
------------
You type:
---------

40 LET X•= N + 1
50 LET N•= X
60 GOTO 30
RUN

When you've seen enough, press the {BREAK} key.

In Line 40, we have identified another variable called
"X". Furthermore, we have said "LET X = N + 1". You
should read that as: Let the value of X equal whatever
the value of N is plus 1 more. In other words, whatever
the value of N is, the value of X is equal to one more
than N.

Line 50 indicates that N is to be given a value equal to
the  current  value  of  X.  This  is  to  say:  first  we
decided that N would have a value of one. Then, X was
assigned a value of N + 1 or 2. Now we are saying that
really we want the values of N and X to be the same.

All of this is done so that when the computer gets to
line 60 and then in turn goes back to line 30, the value
of N changes. Each time the computer executes line 30,
the  value  of  N  changes.  This  is  a  counter.
Essentially,  the  computer  counts  each  time  it  has
executed Line 30.

An easier way to write our program is:

 READY
 10 PRINT "N", "N-SQUARED"
 20 LET N•= 1
 30 PRINT N, N*N
 40 LET N = N + 1
 50 GOTO 30

Look at our new line 40. How can N = N + 1? This is
NOT an algebraic equation! What is meant here is LET
the new value of N become the old value of N plus one
more.

     LET N = N + 1

(new value) (old value)

Exercises:
----------

1) write a program which will display the number, the
square of the number, and double the number.

2) Write a program that will make the computer count
from one to forever.

3) Modify the above program so that it will PRINT your
name and count at the same time.

4) Modify the above program so that it will PRINT your
name and count at odd numbered increments only.

          Hint: LET X =X + 2

5) Modify  the  above  program  so  that  it  will  begin
counting with 5 and increases in odd increments.

Remember with these programs, when you have seen enough,
press the {BREAK} key. Also since exercises 3, 4, 5,
are  modifications  of  exercise  2,  just  retype  the
particular line that needs to be changed. Be sure to
LIST the program and then RUN it.

CHAPTER VII
-----------

                  RELATIONAL OPERATORS:
                      IF YOU CAN PASS THE TEST

Glossary:
---------

Relational Operators
--------------------

 <           less than
---

 >           greater than
---

 =           equal to
---

 <>          not equal to
----

 <=          less than or equal to
----

 >=          greater than or equal to
----

IF           a command which tells the computer to test
–-           the truth of a comparison.

THEN         a special key word which always follows an
----         IF  comparison.  When  IF  finds  a  true
             condition, the statement(s) following the
             word THEN are performed.

A computer does only what you specifically tell it to
do. It cannot make decisions "on its own". However, as
a programmer, you can formulate statements in such a way
that  the  computer  will  be  forced  to  make  certain
comparisons. Relational operators allow you to test two
values and to determine what relationship one has to the
other.  The  computer  will  then  decide  if  the
relationship  or  comparison  is  true  or  false.  The
relational operators most commonly used in BASIC are:

<              (less than)

4 < 6          read as: Is four less than 6?
200 < 400      read as: Is two hundred less than four
               hundred?
X < C          read  as:  Is  the  value  of  X  less  than
               the value of C?

>              (greater than)

9 > 5          read as: Is nine greater than 5?
15 > 0         read as: Is fifteen greater than zero?
Y > X          read as: Is the value of Y greater than
               the value of X?

=              (equal to)
4 = 2+ 2       read as: Is four equal to two plus two?
9 = 4 + 3 + 2  read  as:  Is  nine  equal  to  four  plus
               three, plus two?
A = B          read as: Is the value of A equal to the
               value of B

<>             (not equal to)

5 <> 10        read as: Is five not equal to ten?
A <> B         read as: Is the value of A not equal to
               the value of B?
DOG <> CAT     read as: Is DOG not equal to CAT?

For those of you who have studied algebra, the above
symbol is probably a new one. <> is the same as  ≠;
however, the computer will understand only <> since  ≠
is not available on the computer keyboard.

<=             (less than OR equal to)

19 <= 20       read as: Is the value of nineteen less
               than or equal to the value twenty?
R <= D         read as: Is the value of R less than or
               equal to the value of D?
N <= 1         read as: Is the value of N less than or
               equal to the value one?

>=             (greater than or equal to)

1 >= 0         read as: Is one greater than or equal to
               0?
H >= 4         read as: Is the value of H greater than
               or equal to four?
E >= L         read as: Is the value of E greater than
               or equal to the value of L?

Again, the last two symbols are the closest the computer
can come to the conventional math symbols.

In programming in BASIC, the IF statement allows you to
set up comparisons using the relational operators. The
format of an IF statement requires the key word THEN to
follow  a  relational  comparison.  If  the  computer
determines that the comparison is true, it will perform
whatever instructions follow the THEN.

For example:

10 IF X > 5 THEN PRINT X

In the above program line, the computer will PRINT the
value of X if it determines that X is greater than 5.
If the computer concludes that the value of X is not
greater than 5, the computer will drop down to the next
program line. If none exits, it will do nothing else.

10 If Y < 4 THEN GOTO 66

If the computer, in the example above, finds the value
of Y to be less than 4, it will proceed to line 60. If
the value of Y is not less than 4, the computer will
drop down to the next consecutive program line.

Please note: Do not divide the two-character relational
operators (<=, >=, and <>) with a space. If you do, the
computer will list out an error.

Instruction:
------------
You type:
---------

10 LET F = 1
20 LET H = 6
30 IF F + H < 5 THEN PRINT "SUM IS LESS THAN 5"
40 IF F + H > 5 THEN PRINT "SUM IS GREATER THAN 5"
RUN

How does the computer respond? What does it PRINT? In
the  above  program  "SUM IS LESS THAN 5" will  never be

--39--



PRINTed. Given the values of the variables F and H, the
computer will always determine that F plus H is greater
than 5. Now, try other values for F and H for example,
(Try F•= 1 and H = 1) by typing in a new line 10 and/or
line 20.

You will recall that in Chapter VI, we discussed the
GOTO  statement  and  the  loop.  We  can  now  use  these
concepts along with our relational operators to produce
for the first time an example with all the classical
elements of a true computer program.

Instruction:
------------
You type:
---------

NEW
10 LET G•= 1
20 PRINT "HI, MOM"
30 LET G = G + 1
40 IF G < 8 THEN GOTO 20
50 PRINT "END OF PROGRAM"
RUN

Notice  that  the  computer  has  PRINTed  "HI,HOM"  seven
times.

What is the process the computer follows? First, the
computer reads line 1" and initializes the value of the
variable G at 1. Then, the computer executes line 20.
It PRINTs "HI, MOM". When the computer reaches line 30,
it changes the value of the variable G by adding 1 to
the  old  value  of  G  as  we  did  in  Chapter  VI.  The
computer then replaces the old value of G with its new
value, 2. Each time the computer reaches line 40 it
"tests" or "decides" if the value of G is less than 8.
If the value of G is less than 8 the computer determines
that the rest of the statement must be performed. It
therefore obeys the command and loops back to line 20.
On the other hand, if the value of G is not less than 8,
the  computer  will  determine  that  the  rest  of  the
statement  should  not  be  executed.  Anytime  an  IF
statement determines that the following comparison is
false, it causes control to pass to the next succeeding
line number. Therefore, in this case, control passes to
line 50.

Also, it is important to understand that if you wanted
"HI,MOM"  PRINTed  exactly  8  times,  you  would  have  to
change line 40 to read:

        40 IF G <= 8 THEN GOTO 20

                   or

        40 IF G <> 8 THEN GOTO 20

                   or

        40 IF G < 9 THEN GOTO 20

Change line 40 to be the same as one of the lines 40
above. (Remember:  just retype  the line  including the
line number. The computer will automatically disregard
the previous line 40.) RUN the resultant program and
verify that the loop now executes exactly 8 times.

Try changing line 20 to:

20 PRINT "HI, MOM", G
RUN

Remember, by using the comma, we force the computer to
display HI, MOM and the value of G on the same line one
column apart.

Now change line 30 to:

30 LET G = G + 2
RUN

How does this affect the program? How many times does
the computer display HI, MOM and the value of G? What
value does it display?

Change line 10 to say:

10 LET G = 100

What happens to the program? Why is HI, MOM and the
value of G displayed only once? Because we initialized
the value of the variable G at 100, the value is already
greater than the test value in line 40.

Exercises:
----------

1) Write a program which will count to 100 by fives and
display the results.

2) Write a program which will PRINT this message 10
times "COUNTING BY COMPUTER IS EASY."

3) Write a program which will assign a value to each of
two variables. It will then tell the user if the
numbers are equal and if not, which is the larger
number.

CHAPTER VIII
------------

                       INPUT:
                         TALKING BACK TO THE COMPUTER (BASIC)

Glossary:
---------

INPUT        A statement which allows data to be entered
Statement    from the keyboard into the program without
---------    changing the BASIC program.

In the programs we have discussed up to this point, the
computer has done all of the work while a program ran.
An INPUT statement, though, causes the computer to stop
part way through the program. It waits while the user
types some data into the machine. When the user has
finished, the  computer then  continues to  execute the
program. In addition to stopping the program, the INPUT
statement also acts in a manner similar to that of a LET
statement. The computer sees the INPUT statement as an
assignment to a variable. While the computer waits for
the user to type in data, it is, in essence, waiting for
a value to be assigned to the variable.

When the computer executes an INPUT statement, it PRINTs
or displays a question mark. The question mark acts as
a cue or a prompt. The purpose of this is to remind the
user that it is his/her turn to do something.

Software products almost always have INPUT statements.
These allow users to balance checkbooks, do financial
planning  or  play  games  by  changing  the  numbers  upon
which the program acts. Again, the important thing to
remember is that nothing will happen after the execution
of an INPUT statement until the computer receives an
acceptable answer.

Instruction:
------------
You type:
---------

10 PRINT "PICK A NUMBER"
20 INPUT A
30 PRINT "PICK A NUMBER"
40 INPUT B
50 PRINT A+B,A-B,A*B,A/B

Let's analyze this program:

Line 10 is a PRINT statement which tells the user what
he/she should type into the computer.

Line 20 is an INPUT statement which forces the computer
to display a question mark and then wait until the user
provides  some  data.  When  the  user  types  a  number
(followed, of  course, by  (RETURN}), BASIC  places the
value of the number into the variable named A.

Line 30 is a repetition of line 10.

Line  40  is  a  repetition  of  line  20  except  that  the
variable used is named B.

Line 50 is a PRINT statement which tells the computer to
display the sum, difference, product and quotient of the
two numbers.

      -----------------------------------------
      |   NOTE:  In  the description above,   |
      |   we use the word  "user"  to  mean   |
      |   statement.  It is true that there   |
      |   the  person  who  is  RUNning the   |
      |   program.  As the programmer,  you   |
      |   are or can be distinct  from  the   |
      |   "user".  Try   it!   Let  someone   |
      |   else  RUN   this   program.   The   |
      |   difference between this  and pre-   |
      |   vious  programs  is  that in this   |
      |   example   the  program  interacts   |
      |   dynamically with the user.          |
      -----------------------------------------

Instruction:
------------
You type:
---------

RUN

Try this program several times with different answers.
Also, try purposely hitting just {RETURN} in response to
the prompts. Did you find any problems? Notice how the
computer  must  stop  each  time  it  reaches  an  INPUT
statement. Another way to write the program would be.

Instruction:
------------
You type:
---------

NEW
10 PRINT "PICK TWO NUMBERS"
20 INPUT A, B
30 PRINT A+B, A-B, A*B, A/B
RUN

When you RUN this program, the computer will display the
PRINT statement. On the next line, a question mark will
be shown. Nothing will happen until the user enters two
numbers  into  the  computer.  This  can  be  accomplished
either by typing two numbers separated by a comma or by
typing one number and {RETURN], and another number and
{RETURN}. The computer, in the latter case, will not
continue with the program until two numbers have been
entered.  Also  the  computer  will  display  another
question mark to remind the user that it is waiting for
another number.

There are some problems that you might encounter. If
the user enters two numbers without a comma to separate
them, the computer will display "ERROR B INPUT". This
simply  means  the  computer  has  received  inappropriate
data. Remember, computers are not as smart as people.
Information entered  may not  vary from  the prescribed
format. A missing comma may be no big deal to a person;
but, to the computer, it is a huge deal.

Also the computer will not allow the user to put a space
before  the  comma.  Spaces  before  each  number  are
permitted but not before the comma.

       Acceptable                       Unacceptable
       ----------                       ------------
       25,5                             25       5
          25,       5                   25   ,   5

For these and other reasons, it is generally advisable
to use only one variable per INPUT statement.

INPUT statements are often used in educational programs.
By  using  INPUT  statements  one  could  test  a  students
knowledge of mathematics. For example, if you wanted to
test a student's ability to determine the area of a
rectangle, you might use a program like this:

Instruction:
------------
You type:
---------

10 PRINT "WHAT IS THE LENGTH"
20 INPUT L
30 PRINT "WHAT IS THE WIDTH"
40 INPUT W
50 PRINT "WHAT IS THE AREA"
60 INPUT A
70 IF A = L*W THEN PRINT "THAT'S CORRECT"
80 IF A <> L*W THEN PRINT "SORRY, THAT'S NOT RIGHT"

If the student knows the formula for calculating the
area  of  a  rectangle,  this  small  program  also  will
provide  an  adequate  test  of  his/her  multiplication
skills.

Exercises:
----------

1) Write  a  program  that  will  compute  the  area  of  a
triangle. The formula for determining the area of a
triangle is BASE*HEIGHT/2.

2) Write a program that will allow 3 test scores to be
entered and which will then PRINT the average of
those scores.

3) Write a program which will display all whole numbers
and their squares, starting at one and ending at a
number determined by a user's INPUT.

CHAPTER IX
----------

             LOGICAL OPERATORS:
                    DOES THIS MAKE SENSE

Glossary:
---------

Logical Operators
-----------------

AND          Is used to connect two conditions and to
---          determine if both conditions are met.

OR           Is used to connect alternative conditions
–-           and determine if either one condition or
             the other is met.

In addition to the relational operators, discussed in
Chapter VII, BASIC also uses logical operators. These
operators  include  AND  and  OR.  The  logical  operators
allow the computer to make more complex decisions than
were  possible  with  just  the  relational  operators.
Almost always logical operators are used in conjunction
with relational operators.

Examples:
---------

     60 IF X=9 AND Y>15 THEN PRINT X*Y
     90 IF MONTH<12 OR DATE>26 THEN GOTO 10

Although AND and OR are similar, the computer reacts to
them in very different ways.

Using the AND operator, the chart below shows that the
only time a statement is true is if both conditions are
true:

       True AND True = True

       True AND False = False

       False AND True = False

       False AND False = False

You read this chart to mean (for example) "When the
first comparison is TRUE AND the second comparison is
TRUE the entire condition is TRUE." On the other hand,
the OR chart produces more true statements:

       True OR True = True

       True OR False = True

       False OR True True

       False OR False = False

For a real life example using logical operators such as
AND and OR, imagine that you and a friend are shopping
for  tonight's  dessert.  You  will  buy  ice  cream  for
dessert if AND only if both you AND your friend choose
ice cream. On the other hand, if your allowances are a
bit more generous, perhaps you could buy ice cream if
either you OR your friend choose ice cream for dessert.

With the logical operator AND, you both must want the
ice  cream.  Two  conditions  must  be  true.  Using  the
logical operator OR, only one condition must be true -
either you OR your friend must want ice cream.

In a computer statement using the logical operator AND,
if  any  part  of  the  statement  is  false,  the  whole
statement is false. The logical operator AND forces the
computer to make a decision.

Remember, when using the IF statement, if the statement
is  true,  the  computer  will  execute  one  set  of
instructions  (those  following  the  THEN).  If  the
statement  is  false,  the  computer  will  skip  those
instructions  and  will  continue  to  execute  the
instructions on the next line.

Examples:
---------

IF 4<5 AND 10>6 THEN PRINT "BOOK"

Whenever the computer reaches the above line, it will
ascertain that 4 is less than 5 AND 10 is greater than
6. Since both relational conditions of the IF statement
are true, the computer will PRINT the string "BOOK".

IF X=C OR D=T THEN PRINT X-T

In the above example, the computer must decide if X is
equal to C. It must also determine if D is equal to T.
If  either  is  true,  the  .computer  will  display  the
difference between  X and  T. One  relational condition
(X=C) or the other relational condition (D=T) must be
true. If both are true, the computer will display the
difference  between  X  and  T.  If  both  relational
conditions are false, the computer will simply drop down
to the next consecutive program line.

IF 80<100 AND Z<A THEN GOTO 40

Again, the computer will ascertain if each relational
condition is true. If both and only if both are true,
the  computer  will  proceed  to  line  40.  If  either
relational  condition  is  not  true,  the  computer  will
proceed  to  the  next  consecutive  program  line.  (Note
that, since 80 is always less than 100, this conditional
test actually depends only on the truth of Z<A.)

IF l00=G OR 15>5 THEN GOTO 50

In the above example, the computer will decide if 100 is
equal to the value of the variable G. If it is or if 15
is greater than 5, the computer will proceed to line 50.
If both relational conditions are false, the computer
will drop down to the next consecutive program line.
(Again,  since  15  is  always  >  5,  the  conditional  is
always TRUE, so the value of G really doesn't matter.)

        ----------------------------------------
        |   NOTE:   Since   you,   as    the   |
        |   programmer, can always know then   |
        |   truth  value  of comparisons  of   |
        |   constants,  you would seldom (if   |
        |   ever) put such comparisons in  a   |
        |   program.  They are included here   |
        |   for illustration purposes only.    |
        ----------------------------------------

Instruction:
------------
You type:
---------

10 PRINT "WHAT WAS YESTERDAY'S TEMPERATURE";
20 INPUT YESTEMP
30 PRINT "WHAT IS TODAY'S TEMP";
40 INPUT TODTEMP
50 IF YESTEMP > 80 AND TODTEMP > 80 THEN PRINT "WE HAVE
A HEAT WAVE"

There are some important concepts in the above program.
First,  notice  the  semicolon  following  our  PRINT
statements. Remember a semicolon suppresses the return
to the new line and an INPUT statement automatically
produces a question mark. By using the semicolon in our
PRINT  statement  prior  to  the  INPUT  statement,  the
question mark directly follows our question. This is no
big deal. If you don't use the semicolon, it will not
alter  the  function  of  the  program.  However,  the
semicolon  does  make  the  screen  look  nicer  when  the
program runs.

In  line  50  we  have  used  our  logical  operator  AND.
Notice the computer is forced into a decision. If the
computer  ascertains  that  the  statement  is  true,  it
continues to execute the portion of the line following
THEN. On the other hand, if the computer finds that the
statement  is  false,  it  attempts  to  execute  the  next
line.  Since  there  is  no  following  line,  the  program
ends.

RUN this program and answer the question to see what
happens.  Try  it  several  times  with  different
temperatures.

Instruction:
------------
You type:
---------

50 IF YESTEMP < 60 OR TODTEMP < 60 THEN PRINT
"MY, YOU SURE HAVE SOME COLD DAYS".

(Remember, we have changed line se by just typing se and
the new line. The computer will automatically disregard
the previous line 50).

Again, we have forced the computer to make a decision.
If either condition in line 50 is true, the computer
will continue to execute the line. If neither is true,
the computer stops.

Again, RUN  the program  several times,  giving various
answers to the questions.

Exercises:
----------

1) Write a program which will compare the scores of two
bowling games for you and your opponent.

2) Write  a  program  which  will  PRINT  a  message  if  a
number INPUT by a user is between Ie and 2e.

3) Write a program which allows the user to INPUT three
numbers. Display the numbers only if they were typed
by the user in numerical order, from smallest to
largest.

CHAPTER X
---------

                          RANDOM:
                             I WON WHAT?

Glossary:
---------

Random       The  process  by  which  an  item  is  chosen
------       Selection without a pattern or a definite
             aim.

RANDOM      A  function  which  asks  the  computer  to
------      automatically  select  a  random  number  from
            between specified boundaries.

Integer     A whole number: a number with no fractional
-------     or decimal part.

You have probably witnessed or taken part in a random
drawing. People buy tickets to win a prize. One part
of the ticket is kept by the buyer: the other part is
thrown  in  a  hopper  or  a  hat.  One  winning  ticket  is
selected without definite aim. All tickets have an equal
chance to be chosen. This is a random selection.

Another type of random selection involves board games.
These games usually come with a spinner, die, or dice.
Although the number of choices is limited, the actual
number is selected by chance. Each number has an equal
opportunity to be selected.

Using  BASIC,  a  computer  is  capable  of  making  random
selections.  Because  computers  are  often  required  to
select  a  random  number,  the  function  for  choosing  a
random number is built into the computer language. You
could write your own program that would select a number
at random, but it is easier to use the random function
which is built into the language.

In BASIC XL, the random function is written RANDOM( , ).
The blank spaces indicate numbers to be supplied by the
programmer. The first number determines the lower limit
of  the  range  of  numbers  to  be  selected.  The  second
number  determines  the  upper  limit  of  the  range  of
numbers to be selected. The two numbers are separated
by a comma. The RANDOM function always selects integers
(whole numbers).

Examples:
---------

RANDOM(1,6)  selects  integers  between  1  and  6
inclusively. This is identical to throwing a single die
in a dice game.

RANDOM(l,100)  selects  integers  between  1  and  100
inclusively.

RANDOM(5,50)  selects  integers  between  5  and  50
inclusively.

RANDOM(100,1000) selects integers between 100 and 1000
inclusively.

        -----------------------------------------
        | Since  BASIC  XL  considers  the  left |
        | parenthesis to be part  of the name of |
        | the function RANDOM(', there can be no l
        | space after the 'M' when  the  name is |
        | typed in. This rule holds true for a11 |
        | BASIC XL functions.                    |
        ------------------------------------------

There is one major difference in the way the computer
chooses a random number and our random prize drawing
example. In the random prize drawing, if more than one
number is to be selected, the first number drawn is put
aside. The remaining numbers now have a better chance
of being selected. When the computer selects a random
number, it tosses the number back into the "hopper".
Each  time  a  number  is  selected,  it  has  the  same
probability of being selected the next time. If, for
example, you wanted the computer to choose 15 numbers
between  1  and  100,  it  is  possible  that  one  or  more
numbers might be repeated. It is not an easy task to
keep the computer from rese1ecting a number, and the
topic will not be discussed here.

Selecting a random number is the basis for a simple
guessing  game.  As  you  learn  more  about  computer
programming, you will be able to use variations of this
game to form a more sophisticated program.

Also please note: to select a number with zero as the
lower limit, it is not necessary to write the zero in
the RANDOM function. When only one number is present,
the  computer  assumes  the  lower  limit  to  be  zero.
However, it then assumes that the upper limit is ONE
LESS  than  the  number  given.  (There  is  a  historical
reason for this: Apple II Integer BASIC functions this
way.)

RANDOM(100)  selects  integers  between  0  and  99
inclusively.

RANDOM(50)  selects  integers  between  0  and  49
inclusively. The RANDOM function tells the computer the
range of numbers from which it is to make its selection.
You must tell the computer how many numbers to select.

Instruction:
------------
You type:
---------

           10 LET I = 1
           20 PRINT RANDOM(1,100)
           30 LET I =I + 1
           40 IF I < 7 THEN GOTO 20

The above program will select 6 random numbers between 1
and 100. Let's analyze each statement line. First, we
initialized our variable I in line 10. In line 20 we
tell the computer to PRINT a random number. Next, we
use our non-algebraic equation as a counter. Each time
the computer executes line 39, the value of I increases
by  one.  Finally,  we  force  the  computer  to  make  a
decision. The computer must decide if the value of I is
less than seven. If it is, the computer returns to line
29 and executes that line and the succeeding lines. On
the other hand, if the computer determines that I is
equal or greater than 7, it stops.

Instruction:
------------
You type:
---------

          10 LET NUMBER = RANDOM(1,100)
          20 PRINT "PICK A NUMBER BETWEEN 1 and 100"
          30 INPUT GUESS
          40 PRINT GUESS , NUMBER

The above program represents a game where the computer
selects  a  number,  and  the  user  tries  to  guess  the
number. Notice the value of NUMBER is chosen at random
by the computer. The number 1 is the lower limit and
100 is the upper limit from which the computer might
make  its  selection.  In  line  20,  the  PRINT  statement
indicates which kind of data the user should enter into
the  program.  The  INPUT  statement  makes  the  computer
wait until the value of the numeric variable has been
entered. Notice GUESS is a numeric variable. The user
may chose any number between 1 and 100. Thus, any value
between 1 and 100 may be assigned to our variable GUESS.
In line 40, the PRINT statement tells the computer to
display the GUESS and the NUMBER it chose. The comma
suppresses the carriage return so that both the number
and GUESS are displayed on the same line. Again, this
line shows the two variables: the NUMBER variable whose
value  was  determined  by  the  computer  and  the  GUESS
variable whose value was determined by the user.

Exercises:
----------

1) Write a program to generate 10 random numbers in the
range 1 to 50.

2) Modify  the  above  program  so  that  the  range  of
numbers is 0 to 999.

3) Modify  our  guessing  game  example  so  that  the
computer picks a random number between 1 and 50, and
you try to guess the number. Remember to use INPUT.

4) Modify the above program so that the computer PRINTs
hints to the user to trap the number. Hints might
include "TOO SMALL" and "TOO LARGE". Allow the user
to  keep  on  guessing  until  he  gets  the  correct
answer.

CHAPTER XI
----------

                 THE PROGRAM RECORDER:
                     HITS ON TAPE

        --------------------------------------
        |  The following  is an explanation  |
        |  of how to use the  Atari program  |
        |  recorder.   If  you  own  one of  |
        |  these devices, this chapter will  |
        |  explain  how  to  preserve  your  |
        |  programs  on  tape  and  how  to  |
        |  retrieve them. If you do not own  |
        |  a recorder, the information that  |
        |  follows   should   enhance  your  |
        |  understanding of the product and  |
        |  also help you to decide  whether  |
        |  or  not  to  purchase a  program  |
        |  recorder.                         |
        --------------------------------------

Glossary:
---------

Program      Is a  cassette tape  recorder that  can be
Recorder:    used to transfer programs to and from the
---------    Atari Home Computer.

CLOAD        A command which is used to enter a program
-----        from the program recorder to the computer's
             memory  and  is  used  in  conjunction  with
             CSAVE.

CSAVE        Is  a  command  which  is  used  to  store  a
-----        computer program from the computer's memory
             to  the  program  recorder  and  is  used  in
             conjunction with CLOAD.

LIST "C:"    Is a command which directs the computer to
---------    store programs or specified program lines
             on  a  cassette  tape  and  is  used  in
             conjunction with ENTER "C:".

ENTER "C:"   Is a command which directs the computer to
----------   enter a program from the program recorder
             to  the  computer's  memory.  It  is  always
             used in conjunction with LIST "C:".

By  now  you  are  getting  some  ideas  about  computer
programming. Perhaps you are beginning to see how this
new skill will make life simpler, or perhaps you are

becoming aware of how the computer can be used in your
home, office, or school.

As discussed earlier, once you turn off your computer,
all the programs you were working on are erased from the
computer's memory. This is also true if you type "NEW"
into the computer (or open the cartridge door on the
Atari 4ee or ae9). However, there are times when you
want to save a program. You could use paper and pencil
and write down your program. Then, each time you want
to  use it, you would have to re-enter the program into
the computer's memory. This is very time consuming and
also very frustrating, especially if it is a program you
often use.

Fortunately, there is an easy way to save your programs.
You  can  purchase  an  ATARI  Program  Recorder.  This
recorder is very similar to a portable tape recorder,
but it is designed specifically to work with an ATARI
computer. A 30 minute cassette will conveniently hold
two of your computer programs, one on each side. Also,
if you own a program recorder, you can buy many program
tapes.  These  tapes  contain  various  computer  programs
that will help with family budgeting, learning a foreign
language, or games, just to name a few.

Be careful when handling the cassettes. They are easily
damaged, particularly if you touch the tape itself. Be
sure to store the cassettes in their cases when not in
use.  Do  not  store  the  tapes  in  hot  areas,  direct
sunlight, or near magnetic fields, such as those found
near motors, magnets, or airport security detectors.

It is a good idea to label every cassette with the names
of the programs it contains. This will make it easier
for you to locate a particular program when you want it.
Notice, each cassette has two notches in the rear edge.
The Program Recorder will not record on a cassette tape
which  has  the  holes  exposed.  After  recording  one
program or two programs on a tape, you can protect the
program(s) from being erased or taped over by punching
out the square of plastic and exposing the holes.

That should be enough background information to get you
thinking about purchasing a Program Recorder. Now the
important  stuff.  You've  just  finished  writing  a
brilliant program, and you want to save it. You may type
one of two statements in order to save your program.
One  statement  is  CSAVE;  this  can  only  be  used  for
storing programs on a cassette. The other statement is
LIST"C:". (A variation of this statement can be used
with other devices in addition to the program recorder.)

CSAVE  will  always  save  the  entire  program  from  the
computer's memory. With LIST"C:", you have two choices.
You can save the entire program or, if you wish, you can
specify the first through last lines to be saved. For
example,

 LIST "C:",200,1500

would cause all program lines between 200 and 1500 to be
listed on a cassette.

Whichever statement  you use,  CSAVE or  LIST"C:", each
causes the computer to react in the following manner.
First,  the  computer  will  beep  twice.  This  is  your
signal to put the cassette into the program recorder and
to move the tape using "Rewind" or "Fast Forward" to the
point at which you want the recording to begin. Press
the  {Record}  and  {Play}  levers.  Since  the  computer
cannot determine when you are finished setting up the
tape, you must signal it by pressing the {RETURN} key.
Once you do that, the tape will start moving. If you
turn up the volume on your monitor, you will hear the
recording  taking  place.  When  the  recording  is
completed, the tape will stop moving and you can press
the {Stop} lever.

If you have saved your program on cassette by using a
CSAVE statement, then you must use the CLOAD statement
to get your program from cassette into the computer's
memory. If you used LIST"C:" to store your program on
cassette, then you must use ENTER"C:" to load programs
back into the computer.

CLOAD will erase the program currently in memory before
loading a new one. On the other hand, the ENTER"C:"
statement  will  merge  the  program  it  loads  with  the
program  in  memory.  However,  if  the  incoming  line
numbers are the same as the existing ones, the incoming
lines  will  replace  the  existing  ones.  To  avoid  the
merging, type "NEW" before using the "ENTER" statement.

When you use CLOAD or ENTER"C", the computer will beep
once. Again, this is your signal to get the cassette
into the program recorder and to adjust the tape to the
point  at  which  the  program  begins.  Press  the  {Play}
lever on the program recorder and then the {RETURN} key
on the computer's keyboard, The tape will begin moving.
When the tape stops, press the program recorder's {Stop}
lever.

After your program is loaded into the computer's memory,
simply  type  RUN  and  press  the  {RETURN}  key.  This
executes your program.

In order to see one advantage of using the LIST"C:", try
this exercise:

Instruction:
------------
You type:
---------

100 PRINT "THIS IS LINE 100.
110 PRINT "IN ORIGINAL PROGRAM"
200 PRINT "THIS IS LINE 200"
2l0 PRINT "IN ORIGINAL PROGRAM"
300 PRINT "THIS IS LINE 300"
3l0 PRINT "IN ORIGINAL PROGRAM"

Now RUN this program to see what it does.

Then type:
----------

LIST"C:",200,2l0

When you hear the two beep signal, place the cassette
into  the  Program  Recorder.  Push  the  Play  and  Record
lever.  Press  the  {RETURN}  key.  When  the  tape  stops
moving, press the {Stop} lever on the Program Recorder.

Now type:
---------
NEW
100 PRINT "THIS IS A NEW LINE 100"
200 PRINT "THIS IS A NEW LINE 200"
300 PRINT "THIS IS A NEW LINE 300"

Now RUN the program to see what is does. Rewind the
cassette  tape  and  then  type  "ENTER  "C:".  After  the
beep, press the {Play} lever on the recorder and the
{RETURN}  key  on  the  keyboard.  When  the  tape  stops,
press the {Stop} lever.

To see the results, type LIST and RUN. Notice how the
programs have merged. This may not seem important to
you now, but as your programming ability increases, this
will become a handy procedure. It allows you to alter
parts of programs or to combine programs easily. Again,
this  can  save  you  valuable  time  and  expand  your
programming horizons.

Exercise:
---------

1) Practice using the program recorder by copying any
of the programs previously presented in this book.

CHAPTER XII
-----------

THE DISK DRIVE:
                          BEING FLOPPY ISN'T SLOPPY

        ---------------------------------------
        |  The information contained in this  |
        |  chapter  is  intended to give the  |
        |  reader  general  information   on  l
        |  disk  drives  and their uses.  If  |
        |  you own  one  of  these  devices,  |
        |  this  chapter should enhance your  |
        |  knowledge  of  how  to  use   the  |
        |  drive.   If you do not own one of  |
        |  these devices, the information in  |
        |  this  chapter   will   help   you  |
        |  determine if you should or should  |
        |  not   purchase   this  additional  |
        |  piece of equipment.                |
        ---------------------------------------

Glossary:
---------

Disk         Is a device which connects to the Atari Home
Drive        computer and reads or writes information on
-----        a diskette.

Diskette     Is a  vinyl "record"  enclosed in  a stiff
--------     plastic envelope.

File Name    Consists of up to eight characters and is
---------    used to distinguish files on a diskette.

File Name    Is a suffix or an addition to a file name
Extension    and consists of a period and one, two, or
---------    three characters.

Boot         Is the method for starting the disk drive
----         and loading its operating system into the
             computer's memory.

DOS          An acronym for Disk Operating System. Also,
---          a command from BASIC XL which will transfer
             control  to  DOS  (not  discussed  in  this
             book).

System       Any diskette which contains a DISK OPER-
Diskette     ATING SYSTEM (DOS) on the diskette.
--------

DIR          Is a command which causes the computer to
---          list all the files currently located on a
             particular diskette.

LOAD         Is  a  command  which  causes  a  program
----         previously  stored  on  a  diskette  to  be
             entered  into  the  computer's  memory.  This
             command is used in conjunction with SAVE
             "D: FILENAME. FNE".

SAVE         Is a command which causes a program in the
----         computer's  memory  to  be  stored  on  a
             diskette.  This  command  is  used  in
             conjunction with LOAD "D:FILENAME.FNE".

ENTER        Is  a  command  which  causes  a  program
-----        previously  stored  on  a  diskette  to  be
             entered  into  the  computer's  memory.  This
             command is used in conjunction with LIST
             "D: FILENAME. FNE".

LIST         Is a command which directs the computer to
----         store programs or specified program lines
             on  a  diskette.  This  command  is  used  in
             conjunction with ENTER "D:FILENAME.FNE".

In the previous chapter, we discussed the advantage of
owning  an  Atari  Program  Recorder.  If  you  enjoy
programming or if you are planning on using a great deal
of  the  commercially  produced  software  programs,  you
should consider  purchasing a  disk drive.  Although it
initially costs more than the program recorder, the disk
drive  substantially  increases  the  usefulness  of  your
Atari Home Computer.

A disk drive is more effective as a program storage
device than a program recorder. A single diskette could
hold  as  many  as  64  programs  while  a  cassette
conveniently  holds  only  two  program.  It  operates
quicker  and  permits  almost  instantaneous  access  to
information.  The  disk  drive  also  is  more  reliable.
With a disk drive, you have more choices of prepared
software. Also, the diskette used with the disk drive
keeps track of all the information stored on it. Unlike
the program recorder, you do not have to remember where
a program is stored. The diskette (or, more properly,
the Disk Operating System) remembers for you. There are
other major advantages to disk drives versus cassettes,
particularly as the programs written or used become more
sophisticated.

Before  we  get  ahead  of  ourselves,  let's  begin  by
( explaining the physical aspects of the disk drive.

The  disk  drive  is  a  rectangular  box  which  connects
directly into the Atari Home Computer. (In the Atari
l450 XLD a disk drive is built into the system.) Imagine
a record player which only operates when the lid is
shut.  Essentially  that  is  how  the  disk  drive  works.
While a record player uses a needle to produce sound
from a record, the disk drive uses a magnetic head that
can  read  or  write  information  on  a  special  "vinyl
record" called a diskette.

If you examine a diskette carefully, you will notice
that  there  are  openings  in  the  plastic  envelope
surrounding  the  "vinyl  record",  These  openings  allow
the  disk  drive  to  read  information  from  or  write
information on the diskette without having to remove the
protective plastic cover.

On the upper right side of the diskette, there is a
notch. This notch works in the same way as the plastic
squares on the rear edge of the cassette tape. If the
notch is present, then information can be recorded on
the diskette. If it is not present, the disk drive will
not write on that particular diskette. Some diskettes
contain no notch; thus, they are permanently protected
from accidental writing. If you store information on a
diskette and you want to be sure it is protected, you
can cover the notch with a special label called a "write
protect tab" or with a piece of opaque tape.

Although a diskette is small, 5 1/4 inches in diameter,
it can store a great deal of information. A diskette is
often  compared  to  a  file  drawer.  The  diskette  is
capable of holding many files in the same way that a
file drawer can hold many file folders.

In order to distinguish each file from the other files
stored on a diskette, a name is assigned to each file.
A name consists of a maximum of eight characters. The
first character must be a capital letter. However, the
name itself can contain a combination of capital letters
or  capital  letters  and  numbers.  No  blank  spaces,
punctuation marks, or special characters are permitted
in file names.

To further distinguish one file from another, you might
include a file name extension. A file name extension
consists  of  one,  two,  or  three  characters  and  may
contain  any  combinations  of  letters  and  numbers.  To
specify a file name extension, simply add a period to

the end of the file name and then add the extension.
For example, BUDGET.88 might be the name of your family
budget program for 1988.

A reserved section of the disk, called the directory,
remembers these names and file name extensions for you
(along with their locations and sizes). When you use
the name of a file that you think is already on the disk
(as in LOAD or ENTER), the directory for your file and
"connects" BASIC and your program with it if it exits.
If it doesn't exist, DOS indicates an error. When you
use the name of a new file (as in SAVE), DOS creates an
entry in the directory for you.

      --------------------------------------------
      | NOTE: If you have Atari DOS 2.0,  DOS XL |
      | version 2, or  OS/A+ version 2, you  may |
      | ignore this  note.   If you  have such a l
      | DOS,  you should refer  to its reference |
      | manual    for   information    regarding |
      | acceptable file names.   Some Disk Oper- |
      | ating Systems  currently  available  for |
      | the  Atari allow longer and more complex |
      | filenames.  OS/A+   version   4.1,   for |
      | example,  allows  30-character names and |
      | allows almost any character in a name.   |
      --------------------------------------------

In  order  to  use  a  disk  drive,  be  sure  that  it  is
properly connected to your Atari Home Computer. If you
have  questions  about  this,  check  your  Disk  Drive
Reference Manual.

Next, if your computer is on, turn it off. Turn on the
disk  drive  and  insert  a  system  diskette.  Follow  the
procedure for properly inserting the diskette according
to your disk drive manual. Be sure that your fingers
touch  only  the  protective,  plastic  envelope.  The
diskette should slide into the drive. Gently close the
disk  drive  door.  If  there  is  any  resistance  while
inserting the diskette or closing the disk drive door,
stop, remove the diskette, and try again. This is an
easy procedure and does not require brute strength. If
you use force, you could ruin the diskette. Turn the
computer on; the disk drive will make a whirring noise.
This means the disk operating system is loading from the
diskette into the computer1s memory. This procedure is
known as "booting". When the READY message appears, the
boot procedure is completed.

If you do not use the proper diskette, a BOOT ERROR will
result. A proper diskette is one that contains a disk
operating system, often referred to as DOS.

       -----------------------------------------
       | Some DOS disks (such as OS/A+ version |
       | 2.1) do not  automatically  enter the |
       | cartridge  when the  power is  turned |
       | on.  The user must type  some command |
       | ('CAR' in the case of OS/A+) to enter |
       | the BASIC XL cartridge.               |
       -----------------------------------------

Once you have successfully booted a diskette, you will
want to know what programs are located on it. To get
the  "directory"  or  listing  of  the  files,  type  DIR
{RETURN} into the computer. This will list the diskette
directory on the screen.

If you have digested all of the above information, you
are now ready to use your diskette. Choose the program,
from the directory that you would like to run. Type
LOAD "<name. ext>" • When the computer has loaded the
program from the diskette to its memory, it will say
READY. Next, type RUN. The computer will execute the
program.

If you have created a program and you would like to
store it on your diskette, first choose one with DOS
(DISK OPERATING SYSTEM) already on it. Then, type SAVE
"D:<name.ext>". This will store an entire program from
the computer's memory onto the diskette.

Remember, if you use the command SAVE to store a program
on a diskette, then you must uses the command LOAD to
retrieve the program from the diskette and record it in
the computer's memory. SAVE and LOAD are used to store
and retrieve for an entire program.

If you want to keep part of a program, then you may use
the  command  LIST"D:<name.ext>",<line  range>.  This
command also allows you to preserve a particular program
line or range of files. For example:

        LIST"D:INCOME.87",310

will record on a diskette only line 310 (if it exists)
of  the  program  currently'  in  memory.  It  gives  the
filename "INCOME.S?" to the file containing that single
line.

If, you want to keep only lines 310 to 450, you would
use the following command:

        LIST"D: INCOME.S?", 310, 450

This command will preserve only lines 310 through 450.
If there are additional lines in the program, they will
not be preserved on the diskette. To keep the entire
program, simply use LIST"D:INCOME.87" without any line
numbers.

Finally,  the  commands  LIST  and  ENTER  must  be  used
together. One advantage of using LIST and ENTER over
SAVE and LOAD is that the LIST and ENTER method allow
you  to  merge  programs.  LOAD  will  erase  the  program
currently in the computer's memory before loading a new
one.  On  the  other  hand,  the  ENTER:"D:FILENAME.FNE"
statement  will  merge  the  program  it  loads  with  the
program in memory. As with keyboard programming, if the
incoming line numbers are the same as the existing ones,
the incoming lines will replace the existing ones.

In  order  to  see  the  advantage  of  using  LIST
"D:FILENAME.FNE", line, try this exercise:

        Type:
        -----

        100 PRINT "THIS IS LINE 100"
        110 PRINT "IN ORIGINAL PROGRAM"
        200 PRINT "THIS IS LINE 200"
        210 PRINT "IN ORIGINAL PROGRAM"
        300 PRINT "THIS IS LINE 300"
        310 PRINT "IN ORIGINAL PROGRAM"

Now RUN this program to see what it does. Then, type
LIST "D:SILLY.PRG",200,210.

        Now Type:
        ---------

        NEW
        l00 PRINT "THIS IS A NEW LINE l00"
        200 PRINT "THIS IS A NEW LINE 200"
        300 PRINT "THIS IS A NEW LINE 300"

Now RUN this program to see what it does.

        Type:
        -----

        ENTER "D:SILLY.PRG"

To see the results, type LIST and RUN. Notice how the
programs have merged. This may not seem important to
you  now.  but  as  your  programming  ability  increases,
this will become a handy procedure. It allows you to
alter parts of programs or to easily combine programs.
Again, this can save you valuable time.

COMMENTARY: One use for the file name extension is as an
indicator of which command was used to store a program.
Once you see the program name in the directory, you will
know which command to use to retrieve the program into
the computers memory.

Here is a sample directory:

* DOS     SYS 004
* INIT    COM 006
* RS232   COM 001
* COPY    COM 075
  DEMO    LIS 014
  NEWDEMO LIS 007
  MOVE1   SAV 004
* MOVE2   SAV 004
592 FREE SECTORS

The first file shown is DOS.SYS or the Disk Operating
System. The next several files are for utility programs
which were previously stored on the diskette and permits
access to the diskette in different ways. Also, several
programs  are  preceded  by  an  asterisk.  The  asterisk
indicates that that program is protected and cannot be
renamed or erased or written to until it is unprotected.
If you desire more information, please consult the OS/A+
or DOS XL Reference Manual.

The other programs stored on the diskette were those
used as examples in this book. Notice the file name
extensions. Some  programs use  LIS for  programs which
had been LISTed to the diskette, and others use SAV for
programs previously SAVEd to the diskette.

In the process of placing a program in the computer's
memory from the diskette, it is important to follow the
correct procedure. That is, if the user employs ENTER
with a program previously stored with SAVE or LOAD with
a program previously with LIST, an ERROR will be the
result.

Exercises:
----------

1) Type in a program. SAVE it to a disk file with a
name of your choice. Type NEW and then LOAD your
program and RUN it. Did it work?

2) Try to LOAD a LISTed file. What ERROR did you get?

3) Try to ENTER a SAVEd file. What ERROR did you get?

CHAPTER XIII
-------------

                THE PRINTERS:
                      HARDCOPY ISN'T HARD

   ---------------------------------------------------
   |  As  with  the  previous  two  chapters,  this  |
   |  chapter is intended to give  the  reader some  |
   |  general information  concerning printers.  If  |
   |  you already own  one of  these  devices,  the  |
   |  information  which  follows will enhance your  |
   |  understanding. If you do not own one of these  |
   |  devices,  the information  contained  in this  |
   |  chapter should help you determine  whether or  |
   |  not to purchase a printer.                     |
   ---------------------------------------------------

Glossary:
---------

Software     The programs or instructions which make the
--------     computer perform specified tasks.

Hardware     The  equipment  which  makes  up  a  computer
--------     system.  Hardware  includes  the  computer,
             monitor, disk drive, program recorder and
             printer.

Printer      A device which produces on paper characters
-------      previously stored in the computer's memory.
             Printers are also known as line printers.

Hardcopy     Is the same as a paper copy.

LIST"P:"     A statement used for obtaining a hardcopy
--------     listing of a program previously stored in
             the computers memory.

LPRINT       A statement which causes the computer to
------       produce output on paper.

Output       Many  programs  utilize  BASIC  statements
------       which cause information to appear on the
             screen  or  printer.  Such  information  is
             referred to as output because the computer
             puts the information out on the screen or
             on the printer.

Throughout this book we have made a concerted effort in
our  explanations  not  to  use  jargon  or  computerese.
Perhaps a few buzz words won't hurt now.

Often  computer  discussions  will  include  the  terms
"software" and "hardware". Software refers to computer
programs. If you buy a computer game or any program,
you are buying software. Hardware, on the other hand,
is the equipment and includes the computer itself, the
monitor or television, the disk drive, program recorder,
and the printer.

A  printer  is  a  device  which  produces,  on  paper,
characters previously stored in the computer's memory.
Printers  are  manufactured  by  ATARI  and  many  other
companies. Some are designed to operate with your Atari
Home  Computer.  Before  you  purchase  a  printer,  check
with the dealer to be certain that the printer will work
with your computer.

Why do you need a printer? Truthfully, you may not. A
printer produces a paper copy of your program or the
information produced by your program. A paper copy in
computer jargon is called a hardcopy. What you see on
your monitor or television is softcopy. Hardcopy can be
held in your hand: softcopy cannot be held.

Hardcopy is desirable if you want to produce business or
financial reports. Also, hardcopy is essential if you
use your computer to produce letters. For a programmer,
hardcopy provides a means for reviewing a program whose
length exceeds one screen size.

Printers come in a variety of sizes. The amount you pay
for a printer will be determined by its speed and the
quality of the print. Typewriter quality print is the
most  expensive  kind  of  print.  Other  printers  are
capable of  producing graphic  displays. Some  are even
able to reproduce color.

If you decide to buy a printer, spend some time looking
at  the  various  models  available.  Then  determine  the
features you want and make your selection.

In order to use a printer there are two commands you
need to know. LIST"P:" will cause the computer to PRINT
your  entire  program.  If  you  want  only  one  line  to
appear on paper add the line number to the LIST"P:"
statement. Alternatively,. if you include two numbers
separated by a comma following the LIST"P:" statement,
the computer will PRINT those lines inclusively.

Examples:
---------

LIST"P:"       Will list your entire program.

--67--



LIST"P:",15     Will list only line 15.

LIST"P:",30,56  Will list lines 30 through 56
                inclusively.

The other command used in conjunction with a printer is
LPRINT.  LPRINT  (Line  PRINT)  causes  the  computer  to
PRINT data or output on paper rather than on the screen.
If  you  want  to  see  results  on  paper  instead  of  the
screen use LPRINT.

Example:
--------

l0 LET P=l
20 LET X = RANDOM(l,l00)
30 PRINT X
40 LET P=P+l
50 IF P<ll THEN GOTO 20

The above program will produce 10 random numbers between
1 and l00 on your monitor screen. By changing line 30
to:

30 LPRINT X

the same results will be produced on paper.

You  must  have  a  printer  and  you  must  turn  on  your
printer  in  order  to  use  the  LPRINT  and  LIST"P:"
statements. If you try to use these statements without
a printer, an ERROR 138 will be the usual result.

    -----------------------------------------------
    | The command  LPRINT  forces the computer to |
    | the  beginning of a new line.  Using ; or , |
    | as we did  with  PRINT on the end of a line |
    | may not produce  the results you desire. If |
    | you need to produce  more  complex  printed |
    | output,  you  may  have  to  avoid  LPRINT. |
    | Please refer to the OPEN and PRINT sections |
    | of your BASIC XL Reference Manual.          |
    -----------------------------------------------

Exercises:
----------

1) Use LPRINT statements to write on your printer a
note to a friend.

2) List your program to the printer.

CHAPTER XIV
-----------

                 GRAPHICS PART I:
                      I GET THE PICTURE

Glossary:
---------

Graphics        A state in which the computer responds
--------        to Mode instructions for the purposes of
                drawing  pictures,  designs,  graphs,  or
                variations of the standard characters.

Pixels          Shaded blocks of colors used in GRAPHICS
------          modes  to  create  pictures,  designs  and
                graphs.

Graphics        The  large  area  of  the  monitor  or
Window          television  screen  in  which  graphics,
--------        words,  designs,  or  pictures  can  be
                displayed.

Text            An area at the bottom of the monitor or
Window          televison screen which contains enough
------          space for four lines of text.

COLOR           A  statement  which  selects  one  of  the
-----           available colors which will be used with
                subsequent PLOT and ORAWTO statements.

PLOT            A statement which illuminates a single
----            point on the screen.

DRAWTO          A statement which causes a line to be
------          drawn  from  a  point  (the  last  plotted
                point) to a specified location.

One of the most exciting features of your ATARI Home
Computer  is  its  graphics  capabilities.  Using  the
various  graphics  modes  can  enhance  any  program  you
write.  Your  ATARI  is  capable  of  being  used  in  many
different  graphic  modes••  Some  graphics  require  more
than  a  beginner's  understanding  of  programming  and
computer design. This chapter is intended to give you an
introduction into the world of graphics and for that
reason we will be discussing only modes 0 and 7, the
most commonly used modes.

To enter the graphics mode, type GRAPHICS and the number
of the mode you desire. In this chapter, as noted, we
will only use GRAPHICS 0 and GRAPHICS 7.

To better  conceptualize the  graphics modes,  think of
your screen as a piece of graph paper. In mode 0, the
screen is equal to 40 columns X 24 rows. In mode 7 the
full screen is equal to 160 rows X 96 columns. (Rows
run horizontally and columns run vertically.)

GRAPHICS 0 is the mode we have been using throughout
this book. Whenever you turn on your ATARI, the screen
mode is automatically set to GRAPHICS 0. In some ways,
it might be easier to think of GRAPHICS 0 as turning OFF
graphics. In GRAPHICS 0 you cannot PLOT and DRAWTO as
you can in other modes.

Also,  in  GRAPHICS  0  what  appears  on  the  screen  are
characters.  The  numbers,  letters  and  special  symbols
that are on the ATARI keyboard are also displayed on the
screen.

In graphics mode 7, you have more choices. First, there
are more columns and rows than were found in mode 0.
Unlike  GRAPHICS  8,  though,  GRAPHICS  7  uses  pixels
instead  of  characters.  Pixels  are  shaded  blocks  of
color. Remember our comparison of the monitor screen to
a piece of graph paper. A pixel represents one square
which is or can be filled with color.

Instruction:
------------
You type:
---------

        GRAPHICS 7

What changes took place? Notice the physical changes in
the  screen.  First,  the  screen  is  split.  There  is  a
graphics window which is black and a text window which
is blue. (If you have a black and white monitor or TV,
obviously you will only see shades of gray. In most
cases you will be able to distinguish the various colors
anyway.) The graphics window is the large area of the
monitor or television screen in which graphics words,
designs or pictures can be displayed. The text window is
the area at the bottom of the monitor or television
screen which contains enough room for four lines of text
or program statements to be displayed.

The pictures or designs you. create will appear in the
larger, graphics screen, while your program statements,
prompts,  messages,  etc.,  appear  in  the  text  window.
Note that displays in the text window appear to be the
same as those in GRAPHICS 0. This is proper, since the
text window is actually and simply a small GRAPHICS 0
screen.

Also in mode 7, you have a choice of colors. The COLOR
statement  allows  you  to  select  one  of  the  available
colors and to draw with that color. In graphics mode 7,
there are four available colors: black, orange, green,
and blue. The black is used for background: but, along
with the other colors, it may also be used by PLOT and
DRAWTO.

The PLOT statement enables you to tell the computer a
particular point position you desire. Again, think of
your screen as a graph. The first number after PLOT
tells the computer the column desired: the second number
indicates  the  desired  row.  The  screen  positions  are
numbered  from  the  upper-left  corner  of  the  screen
starting  with  the  number  zero  (0).  Numbers  in  PLOT
statement are separated by a comma.

                      columns

         0 1 2 3 4 5 6 7 8 9
        +-+-+-+-+-+-+-+-+-+-+-+-+-+-+-+-
      0 |X| | | | |X| | | | | | | | | |
        +-+-+-+-+-+-+-+-+-+-+-+-+-+-+-+-
      1 | | | | | | | | | | | | | | | |
        +-+-+-+-+-+-+-+-+-+-+-+-+-+-+-+-
      2 | | |X| | | | | | | | | | | | |
r       +-+-+-+-+-+-+-+-+-+-+-+-+-+-+-+-
o     3 | | | | | | | | | | | | | |
w       +-+-+-+-+-+-+-+-+-+-+-+-+-
s     4 | | | | | | | | | | | | | 
        +-+-+-+-+-+-+-+-+-+-+-+-
      5 |X| | | | | | | | | |
        +-+-+-+-+-+-+-+-+-+-
      6 | | | | | | | | | |

The points shown in the above examples would be entered
into the computer as follows:

                GRAPHICS 7
                COLOR 1
                PLOT 0,0
                PLOT 5,0
                PLOT 2,2
                PLOT 0,5

Another  statement  used  in  most  graphic  modes  is  the
DRAWTO.  The  DRAWTO  statement  tells  the  computer  the
position at which to END a line. That means, the PLOT
statement tells the computer to start a line at the
position  given,  and  the  DRAWTO  statement  tells  the
computer to end the line at the position given. Just
like the PLOT statement, the DRAWTO statement also uses
two numbers and a comma.

                PLOT 0,0
                DRAWTO 15,25

Here are some examples for you to try.

Instruction:
------------
You type:
---------

{SYSTEM RESET]
GRAPHICS 7
COLOR 1
PLOT 5,5
DRAWTO 96,64
PLOT 150,9
DRAWTO 96,64

What  did  you  see?  You  should  see  a  large,  somewhat
misshapen "V" drawn in orange.

{SYSTEM RESET}
GRAPHICS 7
COLOR 1
PLOT 10,10
DRAWTO 159,10
COLOR 2
PLOT 10,30
COLOR 3
PLOT 10,50
DRAWTO 159,50
COLOR 0
PLOT 10,70
DRAWTO 159,70

Did you notice that nothing appeared to happen after you
performed the last set of instructions? That is due to
the fact that the fourth color, COLOR 9, is black. Now
add this:

PLOT 80,0
DRAWTO 80,95

Now you can see COLOR 0 as it cuts through the other
colors.

While working in the various graphics modes, you may
encounter  an  ERROR  141.  This  simply  means  you  have
PLOTted or drawn to a point beyond the range of the mode
in  which  you  are  working.  In  GRAPHICS  7,  do  not  go
beyond 159 in the column position, nor should you go
beyond 95 in the row position. Otherwise you will end up
with an ERROR 141.

CAUTION: row positions 80 through 95 are NOT displayed
in  GRAPHICS  7,  but  use  of  these  positions  does  NOT
generate an error.

Let's try to write a program using graphic mode 7. What
we  want  to  do  is  to  create  a  bar  graph  that  will
indicate  a  person's  physical,  mental  and  emotional
well-being.  We  will  need  to  use  GRAPHICS  7,  COLOR,
PRINT,  INPUT,  and  other  statements  and  commands
discussed earlier in this book.

First, since we know we want to make a bar graph, we
will  need  to  get  into  graphics  mode  7.  Begin  by
pressing {SYSTEM RESET}.

Instruction:
------------
You type:
---------

NEW
100 GRAPHICS 7

Next, we want to formulate a PRINT statement that will
indicate the intent of our program. You might choose
something like this:

You type:
---------

110 PRINT "ON A SCALE OF 1 TO 100, RATE YOUR"
120 PRINT "PHYSICAL, MENTAL, AND EMOTIONAL STATES"
130 PRINT "FOR TODAY."

SPECIAL NOTE: The reason we have chosen to write our
PRINT statement on three separate lines is for a better
format.  All  of  our  information  would  fit  on  three
display  lines,  our  legal  limit  for  a  program  line.
However, when the computer executed our program, some of
the words would be divided. These divisions would not
necessarily  follow  the  syntax  rules  of  English.  To
avoid user confusion, we divided our PRINT statement so
that no words would be divided in a confusing manner.

Now we are going to put in a pair of statements that
will allow some special features to be added to the
program later.

Instruction:
------------
You type:
---------

200 LET DAYSTART = 10
210 LET DAY = 1

And  then  some  statements  which  allow  the  program  to
interact with the user.

300 PRINT "PHYSICAL RATING, DAY"; DAY;
310 INPUT PHYSICAL
320 PRINT "EMOTIONAL RATING, DAY"; DAY;
330 INPUT EMOTIONAL
340 PRINT "MENTAL RATING, DAY"; DAY;
350 INPUT MENTAL

In statements 300 through 350, we have told the computer
to PRINT messages and to wait for responses. Statements
400 through 480 will indicate COLOR, PLOTs, and DRAWTOs.
These will allow us to form our bar graph.

Instruction:
------------
You type:
---------

400 COLOR 1
410 PLOT 0, DAYSTART
420 DRAWTO PHYSICAL, DAYSTART

These statements will draw an orange horizontal line (or
"bar", hence "bar graph) on the screen. Since DAYSTART
contains 10, the line will be drawn 10 units down from
the top and will start at the left hand edge (PLOT 0,
DAYSTART)  and  go  right  to  the  position  specified  by
PHYSICAL (DRAWTO PHYSICAL, DAYSTART).

Now we will repeat similar statements for the EMOTIONAL
and MENTAL portion of the graph.

Instruction:
------------
You type:
---------

430 COLOR 2
440 PLOT 0, DAYSTART + 2

This + 2 is to allow for separation between the lines of
the graph.

Instruction:
------------
You type:
---------

450 DRAWTO EMOTIONAL, DAYSTART + 2
460 COLOR 3
470 PLOT 0, DAYSTART + 4
480 DRAWTO MENTAL, DAYSTART + 4
RUN

Answer the questions as they appear on the screen. Try
this program several times varying your answers.

If you wanted to chart a person's physical, emotional
and mental states for the entire week, what lines would
you need to add? There are many attributes which make
up a good programmer. One of those necessary qualities
is insight.  The above  program is  relatively straight
forward.  However,  in  order  to  expand  the  program,
insight  is  necessary.  You  will  recall  that  in  the
beginning  of  the  program  (lines  200  and  2l0),  we
indicated that these lines would allow features to be
added. Now that we want to include the entire week on
our bar graph, DAY and DAYSTART are very necessary.

Instruction:
------------
You type:
---------

500 LET DAYSTART = DAYSTART + 10
510 LET DAY = DAY +1
520 IF DAY <= 7 THEN GOTO 300

In line 500 we have added a non-algebraic expression
which will increase the spaces on the graph. This is
necessary  so  that  the  line  of  the  graph  will  be
distinct.  Line  519  is  another  non-algebraic  equation
which permits the number of days to increase by one each
time  line  5le  is  executed.  Again,  this  permits  a
separation of groups of lines so that each day's rating
will be distinguishable from the other days. Finally,
the  conditional  statement  in  line  52e  forces  the
repetition of lines 399 to 519 until 7 days have been
completed. When DAY is equal to 7, the program will
stop.

Exercises:
----------

1) Write a program which will draw random lines on the
screen, changing colors randomly.

2) Write a program which will draw random rectangles on
the screen, changing colors randomly.

3) Write a program which Will draw concentric boxes on
the screen, changing colors randomly.

CHAPTER XV
----------

             EDITING FEATURES:
                   THE SCREENING PROCESS

Glossary:
---------

{DELETE}     This key will "erase" any letter you type
--------     by accident: it moves the cursor one space
             to  the  left  each  time  you  hit  the  key,
             removing the letter, symbol, or number it
             replaces.

{BREAK}      A key which when pressed stops whatever the
-------      computer was doing.

{CTRL}       This is the control key and when it is used
------       in conjunction with various other keys it
             allows the user to edit program lines.

{←}          A key which when used in conjunction with
---          the control key will move the cursor one
             character to the left without erasing the
             characters over which it passes.

{→}          A key which when used in conjunction with
---          the control key will move the cursor one
             character to the right without erasing the
             characters over which it passes.

{INSERT}     A key which when used in conjunction with
--------     the control key will add a space to any
             part of a display line and at the same time
             will move the rest of the line one space to
             the right.

{↑}          A key which when used in conjunction with
---          the control key will move the cursor up one
             display  line  without  erasing  characters
             over which it passes.

{↓}          A key which when used in conjunction with
---          the control key will move the cursor down
             one  display  line  without  erasing  the
             characters over which it passes.

{TAB}        Moves the cursor one tabular position to
-----        the  right.  The  number  of  spaces  between
             tabular positions is usually eight.

{SPACE BAR}  Moves the cursor one space to the right and
-----------  replaces the character over which it passes
             with a blank space.

{CTRL} {l}   Allows the user to freeze a listing of a
----------   program.

{CTRL} {2}   Produces a bell sound.
----------

{ESC}        Allows screen controls to take place when a
-----        program is running.

{CAPS/LOWR}  A key which allows the user to use lower as
-----------  well as upper case letters.

----------------------------------
| Please note:  {CTRL}  and  any |
| other key means to  hold  down |
| the (CTRL} key and at the same |
| time press the other key.  For |
| example, in the glossary above |
| {CTRL} {1}  means press {CRTL} |
| and at  the  same  time  press |
| {l} key.                       |
----------------------------------

Throughout  this  book  we  have  discussed  many  of  the
special features of your Atari Home Computer. As you
have learned to program BASIC, you have made mistakes.
Mistakes are part of the learning process: and as your
computer programming expertise continues to grow, your
frustration at simple errors will also increase. This
chapter  is  intended  to  explain  all  of  the  editing
features of your Atari Home Computer and, in doing so,
lessen those frustrations caused by simple typographical
errors.

This chapter is divided into three sections. The first
section discusses changes that can be made before the
{RETURN} has been pressed. The second section explains
the changes that can be made after the {RETURN} has been
pressed.  The  final  part  of  this  chapter  describes
ancillary features.

Some keys which make it possible to correct errors have
already been discussed. For the sake of completeness, we
will review them here.

--77--



BEFORE {RETURN} IS PRESSED
--------------------------

The  {BACK  SPACE}  key  will  "erase"  any  character  you
type. It moves the cursor one space to the left each
time you hit the key. If you hold down this key or any
of  the  other  keys,  the  Atari  Home  Computer  will
automatically repeat the key function until you release
the  key.  If  you  notice  a  mistake  in  a  line,  simply
press the {BACK SPACE}. Back up to the mistake, and
then correctly retype the line.

Instruction:
------------
You type:
---------

10 PINT "HELLO"

You know that the computer will not accept PINT. Press
the {BACK SPACE} 11 times. Change PINT to PRINT and
correctly retype the rest of the line. If that is too
much work, you can simply press the {BREAK} key and the
entire line will be eliminated.

Another way to change the line is by using the {CTRL}
and {←} keys. Located below the {CLEAR} key and the
(DELETE BACK SPACE) key are four keys which have black
arrows  in  white  squares.  These  keys  when  pressed
along with the {CONTROL} key will move the cursor one
space in the direction that the arrow is pointing. As
the cursor moves it does not erase the characters over
which it passes.

Type  the  incorrect  line  again.  Press  {CTRL}  and  {←}
11 times. The cursor should be on the letter I. Then
press  {CTRL}  and  {INSERT}  once.  When  used  in
conjunction with the {CTRL} key, the {INSERT} key will
add a space to any part of a display line and at the
same time will move the rest of the line one space to
the right. Type in R.

The changes you have made are on the display line only.
If  you  want  the  changes  also  to  be  made  in  the
computer's memory you must press the {RETURN} key. Be
sure the line is typed correctly. Press {RETURN} and
then type RUN.

Instruction:
------------
You type:
---------

10 PRINT "SAMM IS SXTY AND STLL SEXY"

To change this line, first press the {CTRL} and the
{←}  key  twenty-two  times.  Each  time  these  keys  are
pressed, the cursor simply moves one space to the left
and passes over, without erasing, each character.

The  cursor  should  be  on  the  second  M.  By  pressing
{CTRL}  and  the  (DELETE}  at  the  same  time,  all  the
characters on the right side of the cursor are moved one
space to the left.

Using the {CTRL} and the {→} keys, move the cursor 5
spaces to the right. The cursor is now on the X. To
add one space before the X, press {CTRL} and {INSERT}
once, and simply type in the missing I.

Press {CTRL} and {→} 8 times, placing the cursor on
the first L. Once again press {CTRL} and {INSERT} and
add the missing I. Remember to press {RETURN} once the
corrections are made to your satisfaction. Although the
changes are present on the display screen, they are not
recorded in the computer's memory until the {RETURN} key
is pressed.

Sometimes a computer line will exceed a display line.
Editing changes may be made before the {RETURN} key is
pressed.

Instruction:
------------
You type:
---------

10 "SAM IS SIXTY AND STILL SEXY. HIS TNNIS SERVE TINKS.
SAMM DON'T CARE. HE STILL LUVES THE GAME."

In  the  above  example,  the  words  may  break  in
inappropriate  places.  Please  just  ignore  the  English
syntactical  errors  and  concentrate  on  the  editing
features.

There is no correct order in which to make changes. If
you decide to complete the changes in another way, that
is  fine.  There  are  several  different  options.  The
following is but one of the possible methods. Begin by
pressing {CTRL} and {<--} 15 times. The cursor should
be on the U in the word LOVES. To change the U to an O,
simply type O. The O will replace the U.

Using {CTRL} and {A} move the cursor up one display
line. Like the other arrow keys, the {A} when used in
conjunction with the {CTRL} key will move the cursor up
one display  line without  erasing the  characters over
which it passes.

Next move the cursor one space to the right by using the
{CTRL}  and  {→}.  The  cursor  should  now  be  on  the
letter T in the word TINKS. Press {CTRL} and {INSERT}
once and add the S before TINKS. Move the cursor 12
spaces to the left by using the {CTRL} and {←}. The
cursor should be on the first N in TNNIS. Then press
{CTRL} and {INSERT} once and add the E.

Next, move the cursor 23 spaces to the right by using
the {CTRL} and {→} keys. The cursor should be on the
second M in SAMM.

Another method you might try here is to use the {TAB}
key.  The  {TAB}  key  moves  the  cursor  one  tabular
position  to  the  right.  The  number  of  spaces  between
tabular positions is usually eight. Press the {TAB} key
3 times. You may have to adjust the cursor from there.

The cursor should now be located on the N in DON'T.
Press {CTRL} and {INSERT} twice and type in E and S.

Have  all  the  corrections  been  made?  If  so,  press
{RETURN}. Notice where the cursor is now located. It
is at the beginning of the next computer line; not the
next display line. Remember, pressing {RETURN} stores
our PRINT  statement into  the computer's  memory. Even
though we have used almost three full display lines, we
have used only one computer line. The cursor indicates
that the computer is waiting for our next program line
or  command.  Type  RUN.  The  corrections  have  been
recorded.

Notice that in our PRINT statement all the words are
syntactically correct, but not when we RUN the program.
Now  let's  make  the  changes  so  our  program  will  RUN
properly.

AFTER YOU HAVE PRESSED RETURN
-----------------------------

When  making  changes  after  the  {RETURN}  key  has  been
pressed,  it  is  necessary  to  go  back  to  the  program
line - not the area of the monitor where the computer
displayed our program.

Press the {CTRL} and {↑} 8 times. Press the {CTRL} and
{→} 3 times. The cursor. should be on the blank space
between  HIS  and  TENNIS.  Press  {CTRL}  and  {INSERT}  4
times. This will add 4 spaces between HIS and TENNIS.
This should take care of the irregular break in the word
TENNIS.

Now using the {CTRL} and {↓}, move the cursor down one
line. The {CTRL} and {↓} function in the same manner as
the other arrow keys. The cursor moves down one display
line without erasing any characters each time {CTRL} and
{↓} are pressed simultaneously. The cursor should be
located on the blank space before HE.

Press  {CTRL}  and  {→}  7  times,  and  press  {CTRL}  and
{INSERT} once. Now press {RETURN} again. Remember for
changes to take place in the computer's memory as well
as on the display screen, you must press (RETURN]. Since
there  is  so  much  information  on  the  screen,  press
{SHIFT} and {CLEAR}. This will remove all material from
the  display  screen  without  erasing  the  computer's
memory.

Instruction:
------------
You type:
---------

LIST
RUN

Notice anything strange? Yes. There is a period at the
beginning of the line. Not exactly where it should be.
Again, go back to the PRINT statement. Change the line
by adding four spaces before the word CARE. You may do
this using the {CTRL} key and the various arrow keys to
move  the  cursor.  Then  as  we  have  done  in  previous
examples, you may use the {CTRL} and {INSERT} keys to
add the necessary spaces.

Again  press  {SHIFT}  and  {CLEAR}  to  remove  all  the
material that is currently on the screen. Type:

LIST
RUN

What happened this time when you ran the program? Look
at the word "GAME". Why is there only a G and not the
whole word GAME? While we were adding spaces to correct
the  syntactical  errors,  we  added  more  spaces  to  our
computer line. Blank spaces are counted by the computer
in  the  same  manner  that  it  counts  any  character  -
letter, number, punctuation mark or special symbol. When
we added extra spaces we exceeded our legal line limit
of 114 characters. The computer automatically cuts off
any characters after the legal limit.

ADDITIONAL FEATURES
-------------------

As you probably have noticed, the {CONTROL} key used in
conjunction  with  other  keys  gives  the  Atari  Home
Computer  many  additional  editing  features.  Some  have
already been mentioned, And although several others do
exist, we will only discuss two more {CONTROL} features
here. If you have written a very long program and want
to check various parts of the listing, you can stop the
listing  by  pressing  the  {BREAK}  key.  Although
effective,  this  method  is  somewhat  hit  and  miss.  By
pressing {CTRL} key and {1} you can freeze the listing
of a program. When you type {CTRL} and (I) again, the
listing  will  continue.  {CTRL}  and  (I)  allows  you  to
review a program listing at your own pace.

Another {CTRL} feature is used in conjunction with {2}.
This  produces  a  bell  sound.  Used  within  a  PRINT
statement, {CTRL} and {2} can act as an alarm to let you
know  when  the  computer  is  PRINTing  something  in
particular. It is also an easier way to produce a sound
than going through the sound registers.

In order to use the {CTRL} {2} within a program, you
must also use the {ESC} key. {ESC} stands for escape.
This term dates back to the time when teletypes were
commonly used as computer terminals. Although that is
not the case today, the name has stuck just like the
term PRINT has stuck. It is located in the top row to
the left of the {1} key. The escape key, unlike the
{CTRL} and {SHIFT} keys, is pressed and released before
another key is pressed.

Essentially, the {ESC} key works in a similar manner to
program  line  numbers.  The  line  numbers  defer  the
operation of statements until the program is RUN. The
{ESC} key defers the operation of the editing keys until
a program is RUN. There is one exception: {ESC} {CTRL}
{1} cannot be edited into a program.

Instruction:
------------
You type:
---------

10 LET I=1
20 IF LET=I+1
30 IF I < 100 THEN GOTO 20
40 PRINT "{ESC} {CTRL} {2} I'M FINISHED COUNTING"

Remember!  {CTRL}  {2}  means  hold  down  {CTRL}  while
typing {2}.

RUN

Be sure to watch the screen, and listen. When did the
bell ring?

Try another program using the {ESC} key.

Instruction:
------------
You type:
---------

10 PRINT "{ESC} {CLEAR}"
20 PRINT "AHA! THE SCREEN IS CLEARED!"
30 PRINT "GREAT {ESC} {←} {ESC} {←} {ESC} {←}
{ESC} {←} {ESC} {←} {ESC} {↓}

In the beginning of this book, we told you that your
Atari  Home  Computer  normally  displays  UPPER-CASE
letters. The computer will accept instructions written
in lower-case letters; and there may be times when, for
appearance sake or for clarity, you would like to use
lower-case letters.

Find  the  {CAPS/LOWR}  key  located  directly  below  the
{RETURN} key. If you press this key once, you will be
able  to  type  lower-case  letters,  numbers,  some
punctuation marks and the arithmetic operation symbols.
If you now press either {SHIFT} key and at the same time
another  key,  you  will  be  able  to  produce  upper-case
letters (capital letters) of any of the characters shown
on the upper half of the keytop on the keyboard. If you
want to return to upper-case letters, press the {SHIFT}
and {CAPS/LOWR} keys at the same time.

A special feature of your Atari Home Computer keyboard
is a special set of graphic characters that appear only
when  the  {CTRL}  key  is  pressed  at  the  same  time  as
another key. Using these graphic keys, you can create
interesting picture graphics or designs.

This mode will work either with your BASIC cartridge or
without it. If you want to stay in this graphic drawing
mode, press the {CTRL} key and the {CAPS/LOWR} key at
the same time. This procedure will lock the keyboard
into the "graphic drawing mode". In order to return the
keyboard back to normal, press one of the {SHIFT} keys
at the same time as you press the {CAPS/LOWR} key.

You may have noticed on your computer keyboard one key
which has the Atari logo on it. This key switches the
video mode from the normal display--blue background with
white lettering--to inverse video. In inverse video the
characters are blue with a white background.

As  you  become  familiar  with  these  various  editing
features, you will find them very useful. As with other
facets  of  the  computer,  you  will  probably  make  some
mistakes, erasing or deleting lines which you wanted to
save. Like anything else, practice will increase your
skill and lower your frustration level.

CHAPTER XVI
-----------

                       IF REVISITED:
                            THEN WE CAN DO ANYTHING

Glossary:
---------

IF...THEN     A conditional statement which causes the
---------     computer to make a decision.

END           A statement which tells the computer to
---           suspend execution of the program.

 :            A punctuation mark used in BASIC XL which
---           allows  the  programmer  to  use  multiple
              statements on a single program line.

When we discussed IF...THEN in Chapter VII we may have
given you the impression that an IF...THEN statement was
always  followed  by  a  GOTO  statement  or  a  PRINT
statement. Not so. Any statement may follow an IF...THEN
statement.

Examples:
---------

10 IF A = 100 THEN PRINT "TERRIFIC"
10 IF TREES = OAK THEN GOTO 100
10 IF BONUS = 1500 THEN LET SALARY = 25000
10 IF A > B THEN INPUT D

In each of the program lines above, the IF part begins a
conditional statement. The computer must decide whether
the statement is true or not. Assuming the statement is
true, the computer does whatever is specified in the
rest of the program line.

It may PRINT or GOTO. It may also initialize a variable
or change the value of a variable. Or, the computer may
wait for data to be entered from the keyboard. BASIC XL
also  permits  the  programmer  to  use  an  IF...THEN
statement followed by just a line number.

Instruction:
------------
You type:
---------

10 PRINT "WHAT WAS YOUR MOST RECENT GRADE POINT AVERAGE"
20 INPUT AVERAGE
30 IF AVERAGE < 3 THEN 60
40 PRINT "I'M PROUD OF YOU"
50 END
60 PRINT "YOU BETTER STUDY"

In the above program our first PRINT statement indicates
the data desired and our INPUT statement allows the user
to  enter  that  data  into  the  computer.  We  use  an
IF...THEN statement in line 30 in conjunction with only
a line number. It is as though the GOTO is implied. The
computer  executes  THEN  GOTO  and  THEN  line  number  in
the same manner.

For the first time in the above program we have used an
END statement. END tells the computer to discontinue the
execution of a program and to return to direct mode.
Often  an  END  statement  is  the  last  statement  in  a
program. BASIC XL does not require an END statement.
However,  an  END  statement  is  necessary  to  our  above
program.

What would happen without the END statement in line 50?
If the user's grade point average was greater than 3,
the computer would PRINT lines 40 and 60 which you may
or may not desire. With our END statement, a user with
a grade point average equal or greater than 3 would
receive the message "I'M PROUD OF YOU". While the user
with a grade point average of less than 3 would receive
the message "YOU BETTER STUDY". The END statement will
be used again in our next chapter. Just remember END
discontinues the execution of a program and places the
computer in direct mode.

Sometimes you may want to join two or more statements
into  one  program  line.  By  using  a  colon,  you  as  a
programmer may connect multiple statements on the same
program line.

Examples:
---------

40 IF X <> A THEN PRINT "TRY AGAIN": GOTO 10
10 IF G <= R AND B <> K THEN LET X=50: PRINT
   "THAT IS INCORRECT"
30 PRINT: PRINT: PRINT "LOTS OF PRINTS ON ONE LINE": PRINT:
   PRINT

Remember,  it  is  perfectly  legal  and  syntactically
correct to add a colon in order to put more than one
instruction  on  a  program  line.  However,  in  joining
statements you may not exceed the single program line
limit of 114 characters. Using a colon may enable you to
better  organize  your  programs  and  to  save  time  and
space.

Throughout this book whenever we have named a variable,
we have used a LET statement. We did this to remind you
that  we  were  dealing  with  a  variable  and  not  a
mathematical expression. Although it is good form and

acts as a reminder, the LET is usually not necessary.
In BASIC XL the LET is implied whenever a variable is
named.

Of course, there is an exception. Should you want to
use a statement word as the name of your variable, you
must use the LET. For example:

10 LET LET = 15
20 LET PRINT = 235
30 PRINT PRINT, LET

Also,  if  your  variable  name  BEGINS  with  a  statement
word, you MUST use LET.

30 LET COLOR0 = 3
40 COLOR = COLOR0
50 LET LETTER = 26

Exercises:
----------

1) Reggie Smith's salary is based on sales. He is paid
$1000 unless his sales are over $20,00"', in which
case he is paid $2"'''''''. Write a program which
will PRINT his salary.

2) Modify the above program to include deductions for
federal taxes and Social Security. Use a colon to
join computer statements on one line. (Use arbitrary
values for the taxes, if you like.)

CHAPTER XVII
------------

                       SUBROUTINES:
                          CALLING FOR HELP

Glossary:
---------

Subroutine     A  statement  or  a  group  of  statements
----------     within  a  computer  program,  yet
               distinguishable  from  the  rest  of  the
               program,  which  performs  a  separate  and
               complete function.

GOSUB          A statement that tells the computer to
-----          execute a subroutine. This statement is
               always followed by a line number and is
               always paired with a RETURN command.

RETURN         A statement which ends a subroutine and
------         tells the computer to go back to the next
               command in the main body of the program.

As your ability to program increases, the length or your
programs will also increase. Long programs can become
cumbersome. They are difficult to read and their logic
may be to hard for a beginner to decipher. In analyzing
a  long  program,  it  is  helpful  to  break  it  into
functional parts. You may find some repetition in the
program. Perhaps a group of commands have been repeated
in several places.

Instruction:
------------
You type:
---------

10 LET SCORE = 0
20 PRINT "ONE CUP EQUALS 1) 8 OZS. 2) 16 OZS. 3) 32 OZS."
30 LET CORRECTANSR = 1
40 INPUT ANSWER
50 IF ANSWER <> CORRECTANSR THEN PRINT "THE CORRECT
ANSWER IS"; CORRECTANSR
60 IF ANSWER = CORRECTANSR THEN PRINT "THAT'S RIGHT";
LET SCORE = SCORE + 1
70 PRINT "ONE QUART EQUALS 1) 18 OZS. 2) 24 OZS. 3)32 OZS"
80 LET CORRECTANSR = 3
90 INPUT ANSWER
100 IF ANSWER <> CORRECT ANSWER THEN PRINT "THE CORRECT
ANSWER IS"; CORRECTANSR
110 IF ANSWER = CORRECTANSR THEN PRINT "THATS RIGHT";LET
 SCORE = SCORE + 1
120 PRINT "ONE GALLON EQUALS 1) 64 OZS. 2)128 OZS. 3)200 OZS"
130 LET CORRECTANSR = 2

140 INPUT ANSWER
150 IF ANSWER <> CORRECTANSR THEN PRINT "THE CORRECT
ANSWER IS"; CORRECTANSR
160 IF ANSWER = CORRECTANSR THEN PRINT "THAT'S RIGHT":
LET SCORE = SCORE + 1
290 PRINT "THE TEST IS OVER. YOUR SCORE IS": SCORE

---------------------------------------
|  By now,  it would be nice  if you  |
|  could read  a program such as the  |
|  above and understand most of what  |
|  it  does.  If you can't  do  this  |
|  yet, do not worry about it.  Type  |
|  the program  in and  RUN it.  But  |
|  keep trying  to "read"  programs,  |
|  since being  able to  do so  will  |
|  make it easier  to work with both  |
|  this book and computer magazines!  |
---------------------------------------

In  the  program  above,  first  notice  how  many  program
lines  we  used.  Did  you  find  any  repetitions?  Yes,
lines 40, 50, and 60 are the same as lines 90, l00 and
ll0 and the same as lines 140, 150, and 160. Wherever
you find the repetition of program lines, you have the
basis for a subroutine.

A subroutine consists of a group of lines which usually
perform a particular function and are terminated by a
RETURN statement. Most often, a subroutine is used to
execute a specified task by allowing the task to. Be
referenced  from  more  than  one  location  in  the  main
program.

A  subroutine  is  a  statement  or  group  of  statements
within a computer program which performs a separate and
complete function. Subroutines are distinguishable from
the rest of the program. Usually, but not always, a
subroutine is placed at the very end of the program.

A  GOSUB  statement  is  closely  related  to  a  GOTO
statement. Both are followed by line numbers. The one
major  difference  is  that  a  GOSUB  makes  the  computer
remember  where  it  "left  oft"  before  it  goes  to  its
target  line,  which  should  be  a  subroutine.  Also,  a
RETURN  is  always  the  last  executed  statement  of  a
subroutine. The RETURN statement causes the computer to
go back to the main body of the program and execute the
statement following the GOSUB statement.

Although a program may contain numerous GOSUB statements
(including,  usually,  several  GOSUBS  to  the  same
subroutine), the computer always remembers at which line

it left the main program. When the computer encounters
the RETURN, it knows exactly where it left off. It goes
back to its place and continues to execute the next
statement following the GOSUB.

Let's return to our example. We used 17 computer lines
to  give  our  quiz  and  to  keep  the  user's  score.  The
discussion which follows is organized by line numbers to
make it easy for you to refer to the program.

Line 10- The score a person receives when taking a
test is important. Thus, we will have the
computer  keep  track  of  the  score  for  us.
In  this  line,  we  will  initialize  our
variable score at e. We do this because at
the beginning of a test, there is no grade.
Once a question has been answered correctly
or  incorrectly  there  is  a  positive  or
negative score.

---------------------------------------
|  NOTE:  In BASIC,  it is  actually  |
|  often not necessary to initialize  |
|  a variable to 0.  The RUN command  |
|  automatically   initia1izes   all  |
|  variables to e.  All that is nec-  |
|  essary is to name the variable.    |
---------------------------------------

Line 20 PRINT  statement  indicates  our  first
question.

Line 30 The  value  of  our  variable  CORRECTANSR  is
set at 1. We did this because the answer
to the question is 1.

Line 40 The  INPUT  statement  allows  the  user  to
enter data.

Line 50 forces the computer into making a decision.
If  the  answer  is  incorrect,  PRINT  the
correct answer.

Line 60 forces the computer into making a decision.
If  the  answer  is  correct,  PRINT  a  remark
which  praises  the  user  and  increases  the
user's score by 1.

Line 70 PRINT  statement  indicates  our  second
question.

Line 80 The value of our variable CORRECTANSR is set
at 3. Again, we did this because the answer
to the question is 3.

--90--



Line 90 The INPUT statement allows the user to enter
data.

Line 100 forces the computer into making a decision.
If  the  answer  is  incorrect:  PRINT  the
correct answer.

Line 110 forces the computer into making a decision.
If  the  answer  is  correct,  PRINT  a  remark
which  praises  the  user  and  increases  the
user's score by 1.

Line 120 PRINT  statement  indicates  our  third
question.

Line 130 The value of our variable CORRECTANSR is set
at 2. We did this because the answer to the
third question is 2.

Line 140 The INPUT statement allows the user to enter
data.

Line 150 forces the computer into another decision.
If  the  answer  is  incorrect,  PRINT  the
correct answer.

Line 160 forces  the  computer  to  determine  if  the
answer  is  correct.  If  so,  PRINT  a  remark
which  praises  the  user  and  increases  the
user's score by 1.

Line 290 PRINT  statement  indicates  the  end  of  our
program and displays the user's score.

Notice that lines 20 and 30 resemble lines 70 and 80 and
lines  120  and  130.  Although  these  statements  are
similar, they are not exactly the same.

Only lines which are exactly the same can be used for a
subroutine.

But look at lines 90, 100, 110. Aren't they exactly the
same as lines 40, 50, 60, and 140, 150, 160? Yes, they
are. These could be the basis for our subroutine. The
last line of our program is self explanatory.

Let's  rewrite  our  quiz  program  using  a  subroutine.
Lines  10,  20,  30  need  not  be  re-entered.  They  are
correct as is.

Instruction:
------------
You type:
---------

40 GOSUB 400
50 {RETURN}
60 {RETURN}
90 GOSUB 400
100 {RETURN}
110 {RETURN}
140 GOSUB 400
150 {RETURN}
160 {RETURN}
300 END

400 INPUT ANSWER
410 IF ANSWER <> CORRECTANSR THEN PRINT "THE CORRECT
ANSWER IS" ; CORRECTANSR
420 IF ANSWER = CORRECTANSR THEN PRINT "THAT'S RIGHT"
LET SCORE = SCORE + 1
430 RETURN

LIST

The listed program should look like this:

10 LET SCORE = 0
20 PRINT "ONE CUP EQUALS 1) 8 OZS. 2) 16 OZS. 3) 32 OZS."
30 LET CORRECTANSR = 1
40 GOSUB 400

70 PRINT "ONE QUART EQUALS 1) 18 OZS. 2) 24 OZS. 3)32 OZS"
80 LET CORRECTANSR = 3
90 GOSUB 400
120 PRINT "ONE GALLON EQUALS 1) 64 OZS. 2)128 OZS. 3)200 OZS"
130 LET CORRECTANSR = 2
140 GOSUB 400
290 PRINT "THE TEST IS OVER. YOUR SCORE IS": SCORE
300 END
400 INPUT ANSWER
410 IF ANSWER <> CORRECTANSR THEN PRINT "THE CORRECT
ANSWER IS" ; CORRECTANSR
420 IF ANSWER = CORRECTANSR THEN PRINT "THAT'S RIGHT"
LET SCORE = SCORE + 1
430 RETURN

Again, let's analyze each liner but this time let's do
so in the order in which BASIC actually executes the
program:

Line 10 sets the value of the variable score at 0.

Line 20 the  PRINT  statement  indicates  our  first
question.

Line 30 sets the value of the variable CORRECTANSR
at 1.

Line 40 sends the computer to our subroutine.

Line 400 allows the user to enter data.

Line 410 forces the computer into making a decision.
If  the  answer  is  incorrect,  PRINT  the
correct answer.

Line 420 forces the computer into making a decision.
If  the  answer  is  correct,  PRINT  a  remark
which  praises  the  user  and  increases  the
user's score by 1.

Line 430 causes the computer to go back to the main
body of the program and to execute the next
statement  after  the  GOSUB  499.  The  first
time  through  the  subroutine,  the  computer
will go back to line 40.

Line 70 the  PRINT  statement  indicates  our  second
question.

Line 80 sets the value of the variable CORRECTANSR
at 3.

Line 90 sends  the  computer  to  our  subroutine.
Lines 400 through 430 are repeated, exactly
as after line 40.

Line 120 the  PRINT  statement  indicates  our  third
question.

Line 130 sets the value of the variable CORRECTANSR
at 2.

Line 140 sends the computer to our subroutine. Lines
400  through  430  are  repeated,  exactly  as
after line 40.

Line 200 the PRINT statement that indicates the test
is over and the user's score.

Line 300 tells the computer not to go any further.

The numbers may be a bit confusing, but notice how much
easier it is to read and to understand the organization
of our new program. In addition to the subroutine, we
also added an END statement. This was necessary so that

when the computer reached the last program line, it did
not try to execute our subroutine. Remember without the
END statement, the computer would execute each line in
numerical order. Eventually, the computer would reach
line 488 and would wait for the user to INPUT an answer.
Can you see all the problems this would cause? Even if
the user types in an answer, what is the correct answer?
The computer would eventually PRINT ERROR.

A diagram of how the program would RUN might look like
this:

10 20 30 40     70 80 90     120 130 140             300

400 410 4420 430     400 410 420 430     400 410 420 430

Subroutines may be as long or as short as you desire.
They provide an efficient means of organizing a program
that  repeats  a  specified  task,  and  in  general,
subroutines make programs easier to understand.

Exercise:
---------

1) Add two more questions to the above program.

CHAPTER XVIII
-------------

                BETTER LOOPS:
                     WHAT'S THE NEXT STEP

Glossary:
---------

FOR         A  statement  which  indicates  a  sequence
---         that the computer is to complete a number
            of  times.  FOR  is  always  used  in
            conjunction with NEXT.

NEXT          A statement which when used with FOR forms
----          a loop. This loop provides an automatic
              method for making the computer count.

STEP          Is the same as an increment. STEP is used
----          with a  FOR statement  when you  want the
              computer to count by an increment other
              than positive one.

Negative      A number whose value is less than zero. A
Number        negative number is associated with a minus
--------      sign.

In reading this book, you have probably noticed that
there is often more than one method for obtaining a
desired output. At first, we used two variables as a
counter. Later, we introduced a nona1gebraic equation
(LET  X  X+1)  as  a  counter.  In  previous  examples  we
used LET, IF...THEN and GOTO statements to produce a
counter within our program. Two additional statements
which  are  often  used  together  as  a  counter  are
FOR...NEXT.

The statements FOR...NEXT form a loop. FOR...NEXT must
always be used in conjunction with each other. If you
use the NEXT without the FOR, the computer will produce
ERROR 13 - NEXT WITH NO MATCHING FOR. If you use FOR
without a NEXT, the computer will be unable to find the
end  of  the  loop.  The  FOR...NEXT  loop  acts  as  an
automatic counter. The FOR statement is followed by a
numeric variable.

Instruction:
------------
You type:
---------

10 FOR Y = 1 TO 10
20 PRINT Y, "WATCH ME COUNT"
30 NEXT Y

In line 10 above, the FOR is followed by our numeric
variable Y. The one (1) indicates the lower limit of the
variable,  or  the  value  of  Y  when  the  loop  begins.
In other words, the numeric value following the equal
sign initializes the variable.

The number following TO indicates the upper limit of the
value of the variable. When the variable reaches its
upper limit, in this case 10, the loop is completed.

  FOR        Y      =     1        TO        10
             |            |                   |
         NUMERIC     INITIALIZES        THE UPPER LIMIT:
         VARIABLE    THE VARIABLE       THE VALUE OF Y
                                        WHEN THE LOOP IS
                                        COMPLETED.

Between the FOR and the NEXT is a statement. Actually,
any  number  of  lines  and  statements  could  be  placed
between the FOR and the NEXT. These statements indicate
what the computer is to do. In our example, we want the
computer to PRINT the value of Y and the string "WATCH
ME COUNT". The number of times that the statement or
statements  is/are  executed  depends  on  the  number  of
times the FOR statement indicates that the loop is to be
completed. The variable in the NEXT statement must be
the same as the variable in the FOR statement. The NEXT
statement is always the last statement in the loop.

Instruction:
------------
You type:                             Computer Responds:
---------                             ------------------

10 FOR A = 1 TO 3
20 PRINT A, A+A, A*A
30 NEXT A
RUN
                                      1     2    1
                                      2     4    4
                                      3     6    9

In the above program, we have initialized our variable A
and  set  its  upper  limit  at  3.  Our  PRINT  statement
indicates what we want to see on the screen is the value
of A, the sum A+A, and finally the product A*A. Our
NEXT statement completes the loop and sends the computer
back  to  line  10  until  it  completes  the  loop  the
prescribed number of times.

In the above examples each time the loop was completed,
the variable increased its value by one. It repeats the
process  until  the  maximum  value  allowed  by  the  FOR
statement.

What do you do if you want to increase the value of the
FOR variable by more than one? What do you do if you
want to decrease the value of the FOR variable?

The way to make the computer count by increments other
than one is to use the keyword STEP. STEP, followed by
a number, tells the computer to count in a specific
increment.

Instruction:
------------
You type:                             Computer Responds:
---------                             ------------------

10 FOR M = 3 TO 15 STEP 3
20 PRINT M
30 NEXT M
RUN
                                             3
                                             6
                                             9
                                             12
                                             15

The STEP will force the computer to count by 3's. If
you are counting by anything other than positive one,
STEP becomes part of a FOR statement. (In fact, STEP
may only be used in a FOR statement.)

Examples:
---------

10 FOR X = 10 TO 100 STEP 5 (counts from 10 to 100 by
   five's)

10 FOR S = 2 to 40 STEP 2 (counts to 40 by two's)

COMMENTARY:

In our discussions to this point, we have used only
positive numbers. We have counted from zero or one in a
positive manner, ascending the number line.

However,  for  every  positive  number  there  is  also  a
negative number. Look at the number line below.

    <-----------------------0----------------------->
       -5  -4  -3  -2   -1     1   2   3   4   5

A negative number is named by a numeral and a minus (or
"negative") sign. Whenever the direction of movement on
a number line is to the right of the point of origin, we
say  the  movement  is  in  a  positive  or  ascending
direction. Whenever the direction of movement on a

number line is to the left of the point of origin, we
say  the  movement  is  in  a  negative  or  descending
direction. Remember, the point of origin on the number
scale  and  the  direction  of  the  movement  determines
positive and negative values.

Example:
--------

          <------------------+------------------>
              11   12   13   14   15   16   17

In the above example 15 is in a positive direction from
our point of origin; 13 is in a negative direction.

We often discuss negative numbers in terms of cities and
their altitudes. Sea level is our point of origin. The
altitude  of  a  particular  city  is  determined  by  its
relationship  to  sea  level.  The  city  has  a  positive
altitude if it is above sea level. Denver, Colorado, is
often  called  the  mile  high  city  because  it  is
approximately 5200 feet above sea level. On the other
hand, Death Valley, California, has a negative value.
It is approximately 280 feet below sea level.

When using a FOR...NEXT loop, it is possible to move in
a negative direction as well as a positive one. Again,
in order to count in an increment other than positive
one, STEP is added to the FOR statement. By using a
negative number, you indicate that the counting is to
take place in a negative manner.

Examples:
---------

10 FOR C=10 TO 1 STEP -1 (counts backwards from 10 to 1)

10 FOR D = 100 to 0 STEP -2 (counts backwards from l00
by two's)

Notice in our FOR statements that the upper limit of our
variable is stated first and then the lower limit. When
using STEP to decrease the value of the FOR variable,
the larger value must be stated first, then the smaller
value.

Exercises:
----------

1) Write a program to PRINT-your name 10 times.

2) Write a program which will count backwards from 10.
Instead of zero have the computer print "BLAST OFF".

3) Write a program which will count from 0 to 100 by
fives.

4) Write a program which will PRINT all numbers and
their squares, starting at 1 and continuing until it
reaches a maximum number, which should be specified
by a user INPUT.

CHAPTER XIX
-----------

                FOR LOOPS REVISITED:
                        ANOTHER STEP UP

Glossary:
---------

FOR          A statement which indicates a sequence that
---          the computer  is to  complete a  number of
             times. FOR is always used in conjunction
             with NEXT.

NEXT         A statement which when used with FOR forms
----         a  loop.  This  loop  provides  an  automatic
             method for making the computer count.

Nest         To fit together or within one another, as
----         boxes, mixing bowls or small tables (verb).

Nest         An assemblage of things lying or set close
----         together or within one another as a nest of
             tables (noun).

A  FOR...NEXT  loop  is  a  useful  programming  tool.
However,  it  can  become  more  powerful  when  used  in
conjunction with other FOR...NEXT loops.

The  process  of  placing  FOR...NEXT  loops  within  one
another  is  called  "nesting".  Nesting  is  a  desirable
and  beautifully  structured  programming  capability.
Remember, though, that the innermost level of a nest
must be completed before the next level outward can be
executed.

To illustrate the concept of nesting, we will start with
a non-computer example. Imagine that it is your job to
inspect  strawberries  for  quality.  There  are  twenty
strawberries in each basket and 8 baskets in each of
four crates. You must first open a crate, then remove a
basket, and finally, remove each strawberry and examine
it. After you have inspected each strawberry, you would
have  to  replace  them  in  the  basket.  After  you  have
inspected  each  strawberry  in  each  basket,  you  would
replace  the  baskets  into  the  crate.  Then,  you  would
follow the same procedure with the next crate.

If we were to write the above example in outline form,
it would look like this:

I.  For each of four crates and
       A.  For each of 8 baskets in each crate and
            1.  For each of 20 strawberries in
                  each basket,
                  i.  You must inspect a single
                      strawberry and
                  ii. Put it back in its
                      basket. Then,
            2.  When all 20 strawberries are
                checked and replaced, replace
                the basket in the crate. Then,
       B.  When all 8 baskets are checked and
             replaced, put the finished crate
             aside. Then,
II. When all four crates are checked, you are done.

Now we will "diagram" the same procedure in a paraphrase
of BASIC.

         FOR CRATE = 1 TO 4
             FOR BASKET = 1 TO 24
                 FOR STRAWBERRY = 1 TO 20
                     INSPECT A SINGLE STRAWBERRY
                     REPLACE IT IN ITS BASKET
                 NEXT STRAWBERRY
             REPLACE THE BASKET IN THE CRATE
             NEXT BASKET
         SET THE CRATE ASIDE
         NEXT CRATE

Notice in our example above, our FOR and NEXT loops are
paired from the inside out. First, our STRAWBERRY loop
is completed, then the BASKET, and finally the CRATE.

Was  this  example  a  legitimate  BASIC  program?  No,
because  BASIC  doesn't  understand  "REPLACE...",
"INSPECT...",  or  "SET  THE...".  The  FOR...NEXT
statements, though, were completely legitimate.

Suppose  we  modify  the  task  a  little.  Instead  of
inspecting strawberries, let's show a star on the screen
for each strawberry. And let's PRINT the words "BASKET"
and "CRATE" as we complete each task. Our program might
look like this:

Again,  this  is  perfectly  legal  as  long  as  you  nest
FOR...NEXT loops from the inside out.

Instruction:
------------
You type:
---------

NEW
10 FOR CRATE=1 TO 4
20 FOR BASKET=1 TO 8
30 FOR STRAWBERRY=1 TO 20
40 PRINT "*";
50 NEXT STRAWBERRY
60 PRINT "BASKET"
70 NEXT BASKET
80 PRINT: PRINT "CRATE": PRINT
90 NEXT CRATE
RUN

What does this program produce? How many times does the
computer  execute  the  FOR  CRATE  and  NEXT  CRATE  loop?
Which loop is executed more often, the FOR CRATE and
NEXT CRATE or the FOR BASKET and NEXT BASKET or FOR
STRAWBERRY AND NEXT STRAWBERRY loop?

Let's analyze this program:

Line 10 - FOR is followed by our numeric variable CRATE,
and the range of CRATE is set. The starting value of
CRATE will be 1. It's final value will be 4.

Line 20 - We begin another FOR...NEXT loop. BASKET is
our second numeric variable. It's range is 1 to 8.

Line 30 - Is the same as line 20 except STRAWBERRY is
the name of the numeric variable and its value is 1 to
20.

Line 40 - Is a PRINT statement. Each time the computer
executes  line  40,  it  will  produce  a·*.  Notice  the
semicolon. Remember the semicolon suppresses the start
of a new line.

Line 50 - Properly  completes  our  FOR  STRAWBERRY  loop
with  NEXT  STRAWBERRY.  FOR...NEXT  loops  are  automatic
counters.  The  NEXT  STRAWBERRY  command  forces  the
computer to go back to line 30. The return to line 30
continues until the value of STRAWBERRY is >20.

Line 60 - Tells the computer to PRINT the word BASKET
and move the cursor to the next line. Because of the
semicolon in line 40, the word BASKET is PRINTed next to
twenty stars.

Line 70 - Properly completes our FOR BASKET loop with
NEXT BASKET. This time the computer is forced to return

to line 20. Notice that this forces the computer to
re-execute our FOR STRAWBERRY and NEXT STRAWBERRY loop
each time the value of BASKET changes. When BASKET is
greater than 8, the computer will drop down to line 80.

Line 80 - Tells the computer to PRINT a blank line. By
using  the  colon,  we  are  able  to  put  more  than  one
command  on  this  line.  As  you  will  recall,  a  PRINT
statement by itself will force the computer to PRINT a
blank line. After the computer PRINTs a blank line, it
will PRINT the word CRATE. And since we used another
colon and another PRINT statement, the computer PRINTS
an additional blank line.

Line 90 - Properly  completes  our  FOR  CRATE  loop  with
NEXT CRATE. The NEXT CRATE command forces the computer
to go back to line 10. The return to line 10 continues
until the value of CRATE is >4. When CRATE is > 4, the
program will end.

Instruction:
------------
You type:
---------

NEW
100 FOR ROW = 1 TO 10
110 FOR STAR = 1 TO ROW
120 PRINT "*";
130 NEXT STAR
140 PRINT
150 NEXT ROW
RUN

The  above  program  is  essentially  a  variation  of  the
previous program. Once again, in line 100 we begin our
FOR ROW ... NEXT ROW loop. The value of the variable
ROW will have a range of 1 to 10. That is, the initial
value will be 1, and it's final value will be 10.

In line 120, we name our other numeric variable STAR.
Notice that the initial value of STAR is 1 and subsequent
values are determined by the value of the variable ROW.

Lines 130 to 150 are a repetition of lines 30 to 60 in
our previous program.

If you were to play computer, the following would be the
sequence of steps you would follow:

   (line number)

1.  (100) initialize the variable ROW at 1

2.  (110) initialize the variable STAR at 1

3.  (120) PRINT * without moving to the next line

4.  (130) add 1 to STAR and go to line 110

5.  (110) determine  that  STAR  >  ROW  ;  go  to  line
140 (the statement following NEXT STAR)

6.  (140) PRINT  a  blank  line  adjacent  to  the  *
(just  moves  to  the  beginning  of  the
next line)

7.  (150) add 1 to ROW and go to line 100

8.  (100) the  value  of  ROW  is  less  than  10  so
continue with the next statement (line 110)

9.  (110) set the value of STAR to 1

10. (120) PRINT * without moving to the next line

11. (130) add 1 to STAR and go to line 110

12. (110) the  value  of  STAR  (now  2)  is  still  <ROW
(currently 2) so continue with line 120

13. (120) PRINT * on the same line as previous *

14. (130) add 1 to STAR and go to line 110

15. (110) the  value  of  STAR  (now  3)  is  >  ROW  so
go to line 140

16. (140) PRINT  a  blank  line  next  to  the  second  *
(just moves to the start of the next line)

17. (150) add 1 to ROW and go to line 100

This program will finish executing when the value of ROW
is equal to 18. Each time the computer executes the
outer loop (FOR ROW...NEXT ROW) once, it executes the
inner loop (FOR STAR...NEXT STAR) 'ROW' number of times.

Exercise:
---------

1) Write a program using several FOR and NEXT loops to
create a multiplication table.

CHAPTER XX
----------

              STRING VARIABLES PART I:
                          REMEMBERING WORDS

Glossary:
---------

String        Is  a  variable  that  is  capable  of
Variable      containing  a  value  consisting  of  words,
--------      letters, or characters (in contrast to a
              numeric variable, Chapter III, which can
              hold only numbers).

DIM           Is the statement used to tell the computer
---           how much memory might be needed to hold
              the string value of a variable.

 $            Is the symbol which must be used as the
---           last character  in the  name of  a string
              variable.  The  $  symbols  tells  the
              computer that a particular variable is a
              string, not a numeric.

 =            A  comparison  operator  that  can  be  used
---           with strings.

 <>           A  comparison  operator  that  can  be  used
----          with strings.

As you have seen, computer programming with just numeric
constants (such as: 3, 7, etc.) was rather limiting.
When we added numeric variables to our repertoire of
programming statements,  we were  able to  produce more
useful programs.

Just as there were numeric constants, BASIC also uses
string constants. Our use of strings up to this point
has been rather limited. You will recall that a string
is any group of characters inside quotation marks. In
previous chapters we have used string constants only in
our PRINT statements. Formerly used examples include:

        PRINT "HELLO"
        PRINT "HI, MOM"
        PRINT "YOUR NAME"

It is also possible in BASIC to use string variables.
When using a string variable, the variable takes the
place  of  a  string  of  characters,  just  as  a  numeric
variable  is  used  in  place  of  a  number.  String
variables' names are distinguished from numeric names by

always being followed by a dollar sign {$}. The dollar
sign {$} indicates to the computer that the variable is
NOT a numeric variable. To the computer X and X$ are
very  different.  When  the  computer  encounters  X,  it
expects X to have a numerical value. However, when the
computer reads X$, it expects the value of X$ to be a
string of characters.

Since strings come in any size (while numbers are all
the same size as far as the computer is concerned), we
may tell the computer the largest possible number of
characters a given string variable might be allowed to
contain.

Thus string variables differ from numerics is in the
optional use of a DIMension statement. Before we can use
a string variable to hold a string value, we must first
indicate to the computer how long we want it to allow
the  string  value  to  become.  If  we  don't  tell  the
computer how long a string value a particular variable
must  hold,  BASIC  XL  assumes  that  it  will  hold  40
characters (enough for any string that can fit on a
single screen line).

Just as a numeric variable can remember a single numeric
value,  so  can  a  string  variable  remember  a  single
string. (Remember the definition of a string is zero or
more characters. Generally, when we see a string in a
program it is enclosed in quotation marks, as in chapter
I).

Instruction:
------------
You type:
---------

10 DIM B$(5)

In the above DIMension statement, we told the computer
that a string variable called B$ might have as many as
five characters.

You type:
---------

20 LET B$ = "JULIE"

This LET statement tells the computer that the value of
the string variable B$ is JULIE. Note that the quote
marks are NOT part of the string. Just as in English,
they are simply punctuations which show where the string
begins and ends.

One  note  of  caution:  When  you  DIMension  your  string
variable, be sure to allow enough room for all possible,
reasonable  answers.  If,  in  the  above  example,  you
wanted B$="WILLIAM", the computer would accept only the
first  five  characters,  and  B$  would  contain  "WILLI"
again, without the quote marks. On the other hand, B$
could = "SUE" or "JOE" or "MIKE" because they are less
than the maximum number of characters specified in our
DIMension statement.

In Chapter III we defined a numeric variable. Here is a
summary of the differences between a numeric variable
and  a  string  variable.  First,  a  numeric  variable
represents  a  number,  a  string  variable  represents  a
string--that is a group of characters or words. Second,
you  must  DIMension  a  string  variable,  telling  the
computer  how  much  memory  to  reserve  for  the  string
variable. Third, when using a string variable, you must
distinguish the variable by following the letter or last
letter in the variable name with a dollar sign {$}.
Finally,  just  like  other  strings  constants  used
previously in this book, you must enclose the value to
be  assigned  to  a  string  variable  (i. e.,  in  a  LET
statement).

All of this is particularly important in programs which
use INPUT statements. As stated in Chapter VIII nothing
will happen, after the execution of an INPUT statement,
until the computer receives an acceptable answer. If you
try  to  INPUT  a  string  value  where  the  computer  is
expecting  a  numeric,  the  computer  will  produce  a
message--usually  an  ERROR  S  -INPUT/READ.  The  INPUT
statement did not receive the type of data it expected.

The reverse is NOT true. If you use numbers where the
computer  is  expecting  INPUT  of  a  string  value,  the
computer  will  accept  the  numbers,  since  a  numeric
character  is  no  less  a  character  than  is  a  letter.
However, no arithmetic operations can take place with
those numbers. The computer treats them as a string.

Instruction:
------------
You type:
---------

10 PRINT "WHAT IS YOUR NAME";
20 INPUT N$
30 PRINT "HELLO," ; N$
40 PRINT "HOW OLD ARE YOU";
50 INPUT AGE
60 IF AGE <= 20 THEN PRINT "MY, HOW YOUNG YOU ARE," ; N$
70 IF AGE > 20 THEN PRINT "WHAT A NICE AGE TO BE," ; N$
80 PRINT "HAVE A NICE DAY," ; N$
RUN

Now try to answer the questions using various names and
ages. Also, try to reverse you answers putting your age
where your name belongs and vice versa. What happens?
The  computer  will  not  accept  letters  where  it  is
expecting  numbers.  However,  it  will  accept  numbers
where you might expect it to accept only letters. Why?
Because a string is made up of characters. Characters
can be either letters or numbers, or for that matter any
displayable character available from the ATARI keyboard.

String Comparisons
------------------

Just as we have compared two numbers using operators
such as <, >, >=, etc., so may we compare strings. In
this book, we will delve into the meaning of the equal
(".")  and  not-equal  ("<>")  operators  as  applied  to
strings.

+−−−−−−−−−−−−−−−−−−−−−−−−−−−−−−−−−−−−−+
| In fact, though, ALL the operators which are |
| available to numbers  are also available for |
| use   with   strings.   We  would   heartily |
| recommend that you experiment with the other |
| operators.   Is  "CAT">"cat"?  Is  "12">"2"? |
| The answers may surprise you.                |
+−−−−−−−−−−−−−−−−−−−−−−−−−−−−−−−−−−−−−+

Generally  speaking,  the  string  comparison  operators
produce a result by examining each of the two strings
being compared a character at a time. That is, the first
characters  of  each  string  are  compared.  If  they
are equal, then the second characters are compared. If
they are equal, then the third characters are compared.
And so on.

The  character-by-character  comparison  continues  until
the end of one or both strings is reached. Two strings
are  equal  if  and  only  if  they  are  exactly  the  same
length and if each pair of characters match identically.
They are always unequal if their lengths differ.

+−−−−−−−−−−−−−−−−−−−−−−−−−−−−−−−−−−−−+
| For  other comparisons,  the first  pair of |
| characters which differ determine the truth |
| of the comparison. If one string is shorter |
| than  the   other  but   otherwise  matches |
| exactly, it is considered to he "less" than |
| the longer string.                          |
+−−−−−−−−−−−−−−−−−−−−−−−−−−−−−−−−−−−−+

The implication of this is that "CAT" is equal to "CAT"
and only to "CAT". Of course, you can also use string
variables in comparisons, so in this statement

IF X$="CAT" THEN PRINT "THEY MATCH"

the message will be printed if and only if X$ has a
LENgth of 3 and contains the characters "CAT".

Since human responses to computer questions are often
unpredictable,  a  program  such  as  the  following  will
often NOT work:

    10 PRINT "Who is buried in Grant' s tomb";
    20 INPUT NAME$
    30 IF NAME$="GRANT" THEN PRINT "YOU BET YOUR LIFE!"

What happens if the person answers "grant" or "Grant" or
"U.S.Grant" or any of several equally valid responses?
Your program counts them as wrong, where a human would
count such variations as correct.

A  better  choice  in  such  situations  is  usually  the
multiple choice. And here BASIC XL provides a mechanism
to  ensure  "correct"  responses.  Recall  that  we  said
above that BASIC XL will only assign as many characters
to a string as the string is dimensioned for. So what
happens if you dimension a string to contain only one
character? Try the following program fragment:

        10 DIM CHOICE$(l)
        20 PRINT "WHO IS BURIED IN GRANT'S TOMB?"
        30 PRINT "A. GRANT"
        40 PRINT "B. GROUCHO"
        50 PRINT "C. LINCOLN"
        60 PRINT "YOUR CHOICE (A, B, OR C) ";
        70 INPUT CHOICE$
        80 IF CHOICE$="A" THEN PRINT "ok!" : END
        90 IF CHOICE$="B" THEN PRINT "Wow" : END
        l00 IF CHOICE$="C" THEN PRINT "Try again":GOTO 60
        110 PRINT "you must choose A, B, or C":GOTO 60

Do you see the difference? Even if the user answers
"AARDVARKS  ARE  BEAUTIFUL"  in  response  to  your  INPUT
prompt, BASIC XL will only store the first letter ("A")
in the variable CHOICE$, since it was dimensioned to
hold  only  one  character.  Sometimes,  restricting  the
size of a string in this way can make writing string
comparisons much easier.

Exercises:
----------
1) Write  a  program  to  produce  a  form  invitation  in

which a name is requested and placed in a string
variable.  When  the  invitation  is  printed  out,
including at least two usages of the name contained
in the string variable.

2) Ask the user a question which should be answered
with YES or NO. Have your program accept any answer
which  begins  with  "Y"  or  "N".  Respond  with  some
appropriate message(s).

CHAPTER XXI
-----------

           STRING VARIABLES, PART II:
                               EVEN WORDIER

Glossary:
---------

Destination   Is  a  string  to  which  a  value  is  being
String        assigned.
-----------

Source        Is  a  string  whose  value  was  previously
String        stored in the computer's memory.
------

Substring     Is a piece or part of a string's value.
---------

Subscripts    Are  numerical  expressions  which  specify
----------    the  beginning  and  (if  2  subscripts  are
              given) ending character positions within
              a  string  variable,  thus  defining  a
              substring.

LEN(     )    Is a special BASIC function that indicates
----------    the number of characters (including blank
              spaces and graphic characters in a string.

In the previous chapter we discussed string variables
and their use with LET and INPUT statements. There are
many other advanced programming techniques for handling
string  variables.  In  this  chapter  we  will  discuss
destination  strings,  source  strings,  substrings,
subscripts, and the LEN function. Mastering these terms
and  their  usage  should  provide  you,  as  a  beginning
programmer, with enough latitude to create interesting
programs without overwhelming you with other complicated
functions.

The  examples  of  LET  and  INPUT  used  in  our  previous
discussion  of  string  variables  really  discussed
destination  strings.  A  destination  string  is  one  to
which a value is assigned. By DIMensioning a string and
then assigning a value to it, we use it as a destination
string.

With  an  INPUT  statement,  the  keyboard  provides  the
assignment of the value of the destination string. A
destination string is one which is being changed in some
manner.

--111--



Conversely, a source string is one whose value has been
previously  stored  in  the  computer's  memory.  Thus,  a
source string is actually a string which was previously
used as a destination string (i. e., it was DIMensioned
and assigned to earlier in the program).

More information concerning source strings will follow
throughout this chapter. Try to keep in the mind the
differences between destination and source strings.

SUBSTRINGS
----------

With  the  string  variables  previously  employed,  the
entire  string  was  accessed.  We  had  no  method  for
accessing and using part of the string. A substring is a
division or a part of a string. In order to examine
and/or  manipulate  a  substring,  you  must  define  the
substring by using a subscript.

A subscript consists of one or two numbers, and it must
directly  follow  the  string  variable's  name.  The
number(s)  refer(s)  to  the  characters  in  a  string
variable.  Examples  of  subscripts  include  B$(20),
C$(l,15), GUESS$(50), and STATE$(5,20).

Single Subscripts:
------------------

If only one number is present in the subscript, the
computer will access the substring beginning with the
character in the specified, numbered position. It will
continue to access characters until it reaches the end
of the string.

Instruction:
------------
You type:                             Computer Responds:
---------                             ------------------

10 DIM STATE$(100)
20 LET STATE$="NEW YORK, NEW JERSEY,
NEW HAMPSHIRE, NEW MEXICO"
30 PRINT STATE$(5)
RUN
                                      YORK, NEW JERSEY,
                                      NEW HAMPSHIRE,
                                      NEW MEXICO

In the example above, the first statement at line 30
caused  the  computer  to  find  the  fifth  character.
Beginning from that point, it printed the rest of the
string. Now change line 30.

You type:                             Computer Responds:
---------                             ------------------

30 PRINT STATE$(15)
RUN
                                      JERSEY, NEW
                                      HAMPSHIRE, NEW
                                      MEXICO

Again, the computer searches for the fifteenth character
and executes the command. It PRINTS beginning with the
fifteenth character and ends with the last character in
the string.

Now change line 30 again.

You type:                             Computer Responds:
---------                             ------------------

30 PRINT STATE$(25)
RUN
                                      W HAMPSHIRE, NEW
                                      MEXICO

Notice that each character - 1etter, punctuation mark,
blank space, etc. - is assigned a position. If a blank
space is in the position that the computer is searching
for, the computer will PRINT a blank space. It will then
use  (or  continue  to  use)  the  rest  of  the  string.
Change line 30 again.

You type:                             Computer Responds:
---------                             ------------------

30 PRINT STATE$(37)
LIST
RUN
                                       NEW MEXICO

Notice that NEW MEXICO appears to be indented one space.
What really happened was we asked the computer to PRINT
the 37th character. In this example, a blank space is in
position 37.  Therefore, the  computer PRINTed  a blank
space.

Remember,  when  only  one  number  is  present  in  the
subscript, the computer will begin at that number and
will continue to access characters until it reaches the
end of the string.

Double Subscripts:
------------------

If  two  numbers  are  present  in  the  subscript,  the
substring will contain all characters within the stated
numerical bounds, inclusively.

DIET$(1,30) means the computer will begin to access the
substring  at  the  first  character  and  end  with  the
thirtieth inclusively.

ROAD$(15,35) means the computer will begin to access the
substring at the fifteenth character and continue until
it reaches the thirty-fifth character inclusively.

Let's go back to our example program.

Instruction:
------------
You type:                             Computer Responds:
---------                             ------------------

LIST
30 PRINT STATE$(1,8)
40 PRINT STATE$(11,20)
50 PRINT STATE$(23,35)
60 PRINT STATE$(38)
RUN
                                      NEW YORK
                                      NEW JERSEY
                                      NEW HAMPSHIRE
                                      NEW MEXICO

                                      READY

In the first three PRINT statements above, the subscript
included  two  numbers.  The  computer  goes  to  the
particular position designated by the first number in
the subscript. It uses that character as a beginning
point for the specified operation (in this case, a PRINT
operation). It  continues to  access characters  and to
perform the  operation until  it reaches  the character
numbered the same as the second number in the subscript.
It also  accesses that  last character.  (Remember, the
subscripted characters are accessed inclusively.)

STRING CONCATENATION
--------------------

Instruction:
------------
You type:
---------

10 A$="HELLO "
20 PRINT "WHAT IS YOUR NAME";
30 INPUT N$
40 A$=A$,N$
50 PRINT A$

RUN this program and answer the INPUT prompt. Were you
surprised with the results?

The  only  new  thing  here  is  line  40.  The  operation
performed  by  the  comma  in  this  line  is  called
"concatenation". This means that the string following
the comma is appended to the string to the left of the
comma. Both these strings are right of the equal sign
and hence are source strings. What is interesting about
line 40 is that A$ is both a source and a destination
string.

There is no real limit to the number of strings which
may  be  concatenated.  to  illustrate,  replace  line  40
above with this:

40 A$=A$,A$,N$," AND HI,ALSO."

and RUN the resultant program.

Sometimes,  you  do  NOT  know  the  LENgth  of  a  string
variable. For example, in a program which contains an
INPUT statement, the DIMension value is usually large
enough  to  allow  for  all  possibilities.  How  can  you
determine the actual LENgth of a string variable? LEN()
is a BASIC XL function which will tell you the number
of characters including blanks in a string variable. The
space in the parentheses is for the name of the string
variable.

Instruction:
------------
You type:
---------

10 DIM NAME$(31)
20 PRINT "WHAT IS YOUR NAME (30 CHARACTERS"
25 PRINT MAXIMUM)";
30 INPUT NAME$

40 IF LEN(NAME$) = 0 THEN GOTO 100
50 IF LEN(NAME$) >30 THEN GOTO 200
60 PRINT "THANK,"; NAME$:STOP
100 PRINT "DON'T YOU LIKE ME?"
110 PRINT "PLEASE GIVE ME YOUR NAME."
120 PRINT "PRETTY PLEASE?"
130 GOTO 30
200 PRINT "YOUR NAME IS TOO LONG FOR ME"
210 PRINT "TRY TO USE INITIALS OR A NICKNAME"
240 GOTO 20

Do you understand this program? Why did we DIM NAME$ to
equal 31 when we asked for a maximum of 30 characters?
The reason for DIMensioning NAME$ to equal 31 characters
is rather simple. If we used exactly 30, the computer
would reserve only 30 character's worth of space. The
computer would chop off anything beyond 30. There would
be no way to determine if the string variable NAME$
contained exactly  30 characters.  By allowing  for one
extra, the  computer can  make the  distinction between
exactly 30 and more than 30.

In lines 20 and 25, our PRINT statements indicate the
information we want. Notice that the semicolon at the
end of the line forces the question mark produced by our
INPUT statement to be PRINTed on the same line as the
PRINT statement. If the semicolon is not present, the
program itself is not affected. The semicolon allows for
a neater appearance.

By using an IF...THEN GOTO statement in lines 40 and 50,
we have forced the computer to make a decision. First,
it must determine the number of character positions in
our string  variable, NAME$.  The LEN  function enables
the computer to determine the number of positions. The
computer must decide if LEN(NAME$) is equal to 0. If
it  is,  then  the  computer  will  drop  to  line  100  and
execute that line and the ones that follow until it
reaches the next GOTO statement.

If  LEN(NAME$)  is  greater  than  30,  the  computer  will
GOTO 200 and PRINT lines 200 and 210 before it loops
back to line 20.

On the other hand, if LEN(NAME$) is equal to 30 or less
characters, the computer PRINTs line 60 and then STOPS.

We believe that learning to manipulate strings variables
is very important. That is why we have devoted so much
space  to  this  chapter.  Before  we  have  you  create  a
program which  examines string  variables, we  have one
more example.

Instruction:
------------
You Type:
---------

10 DIM NAME$(100)
20 PRINT "TYPE YOUR FIRST NAME AND LAST NAME"
25 PRINT "LEAVING ONE SPACE BETWEEN THEM"
30 INPUT NAMES$
40 FOR PTR = 1 TO LEN(NAMES$)
50 IF NAMES$(PTR,PTR) = " " THEN GOTO 90
60 NEXT PTR
70 PRINT "ONLY ONE NAME?"
80 GOTO 20
90 PRINT NAMES$(PTR+1)
95 PRINT NAMES$(1,PTR-1)
100 GOTO 20

Line 10 - Our program begins by DIMensioning our string
variable NAMES$. Notice, we have told the computer that
NAMES$ might be as large as 100 characters.

Lines 20 & 25 - Next, our PRINT statements indicate the
information we desire.

Line 30 - The INPUT statement provides an opportunity
for  the  user  to  type  into  the  computer  the  desired
information.

Line 40 - We set up a new variable called PTR (pointer).
You will recall in Chapter XVIII we used our FOR...NEXT
loop  as  an  automatic  counter.  We  are  using  it  here
again in that same manner. In line 40 we initialized
the variable at one to continue for the LENgth of our
string variable  NAMES$. Because  the user  might enter
any name with any number of characters to a maximum of
100,  we  have  no  way  to  determine  the  exact  ending
character  position  in  NAMES$  unless  we  use  the  LEN
(NAME$) function.

Remember a string variable with a subscript identifies a
character  position  in  the  string.  Since  we  have
initialized our variable in line 40 to 1, the first time
through NAMES$(1,1) or the first character position. If
that  position  does  not  contain  a  blank  space,  the
computer drops to line 60.

Line 50 -Each time the computer reaches line 50 it tests
to see if NAME$(PRT,PRT) is equal to a blank space. The
computer continues in the FOR...NEXT loop until it finds
the blank space.

Line 60 - Contains the NEXT statement. The value of the
variable  PTR  increases  by  1.  In  essence,  this  means
PTR=PTR+1 GOTO 40. The computer then determines if the
new value of PTR is still within the range of 1 to
LEN(NAMES$).

Line 70 - This PRINT statement was included in case the
user entered only one name.

Line 80 - Our GOTO statement sends the computer back to
line 20 to get the user's first and last name separated
by a blank space.

Line 90 - The computer executes this line when it finds
the  blank  space.  Line  90  means  PRINT  our  string
variable NAMES$  from the  character position  which is
equal to the blank space plus 1. For example, if the
user's name  was WOLFGANG  MOZART, the  FOR...NEXT loop
would begin with the W. It would determine that W is
not a blank space and thus it would move to the next PTR
character  position,  O.  The  procedure  would  continue
until the blank space is found.

Count the  character positions  in WOLFGANG  MOZART. In
what character position is the blank space? The blank is
in the ninth position. Therefore, in line 90 PTR + 1
is equal to 10. The letter in the tenth position is M.
When  a  variable  has  only  one  subscript  number,  the
computer will access all characters beginning with that
number (10 in our example) and PRINT until it comes to
the last character in the string. Because subscripts are
inclusive,  the  last  character  is  also  PRINTed.  Up
to this point the computer would PRINT MOZART.

Line  95  -  Adjacent  to  the  blank  space,  we  tell  the
computer to PRINT NAMES$(1,PTR-1). This means PRINT our
variable NAMES$ beginning with the first position and
continuing until the position of PTR-1 is reached. In
our WOLFGANG MOZART example, PTR was position nine so
PTR-1 is equal to position 8. When a string variable is
followed  by  a  subscript  containing  two  numbers,  the
computer uses the first number as a beginning point and
the second as an ending. Our program produces MOZART
WOLFGANG, or any name which was properly entered. Each
time a name is entered the results will be last name and
first name on separate lines.

If you  have trouble  understanding this  program, play
computer. Using paper and pencil, go through each step
just the way the computer would do it. That should make
the explanation easier for you to understand.

Also please note: The above program is not perfect. We
are aware of the flaws. By placing a space before the
first  name,  a  user  could  cause  problems  with  this
program.  However,  we  are  using  it  to  explain  string
variables. If you noticed the flaws, that's great.

------------------------------------------
|   SPECIAL NOTE:   Actually  BASIC XL   |
|   gives an  easier way  to find  the   |
|   space in NAME$  if it exists.  Try   |
|   replacing lines 40 through 60 with   |
|   this:                                |
|                                        |
|      40 PTR=FIND (NAMES$," ",0)        |
|      50 IF PTR THEN GOTO 90            |
|      (DELETE LINE 60)                  |
|                                        |
|   For more information  on FIND, see   |
|   your BASIC XL Reference Manual.      |
--------------------------------------------

Exercise:
---------

1) Write a program to permit INPUT of a name and then
PRINT the name backwards.

CHAPTER XXII
------------

                 SOUND:
                     YOU CALL THAT MUSIC

Glossary:
---------

SOUND         A BASIC statement which tells the computer
-----         to generate a tone or noise

Sound         Are numbered  from 0  through 3  and each
Channels      channel  represents  a  separate  tone  or
--------      noise.

Pitch         Is the depth of a tone or of a sound. The
-----         range of pitch is numbered from 0 through
              255.

Sound         Can be selected values between 0 and 15
Quality       and allows the ATARI to produce pure or
-------       distorted notes or silence.

Volume        The  degree  of  loudness  or  audibility,
------        represented by a number from 0 to 15.

The ATARI's capacity to produce sound provides another
dimension  for  your  programming  creativity.  With
elementary  programming  skills,  you  can  force  the
computer to execute commands and produce sounds which
would  require  advanced  skills  on  some  other  home
computers.

In order to make a sound, you must first indicate to the
computer  that  you  want  a  sound.  The  SOUND  statement
tells  the  computer  to  generate  a  tone.  A  SOUND
statement is followed by four numbers. Each number is
separated from the others by commas.

Example:
--------

10 SOUND 1,15,10,5

The first  number following  SOUND indicates  the sound
channel. Your ATARI gives you a choice of four channels
numbered from 0 through 3•. Each channel is independent
of the others. Because of this, it is possible to blend
the  sound  channels  together.  This  blending  of  sound
would  be  similar  to  voices  in  a  chorus  or  musical
instruments  in  an  orchestra.  Each  sound  channel  is
controlled by a separate SOUND statement.

Following the sound channel in a SOUND statement is the
pitch. The ATARI is capable of producing 256 different
pitches, numbered 0 to 255, but not all pitches produce
sounds on all modes of "quality".

The  third  number  in  a  SOUND  statement  regulates  the
sound quality. With a value between 9 and 15, the sound
quality  is  capable  of  producing  pure  tones  or  of
producing tones mixed with different amounts of noise.
Pure tones are generated by the values 10 and 14. These
tones are similar to those produced by a flute. Value 12
is also useful when programming music. It simulates a
reed sound similar to a clarinet.

Other even numbered values introduce various amounts of
noise into the pure tone. These values are useful for
creating sounds like car crashes, explosions, and video
games.  Odd-numbered  distortion  values(1,3,5,7,9,11,and
13) produce silence.

SPECIAL NOTE: When the sound quality is 10 or 14, the
pitches  generated  increase  their  frequency  regularly
with 255 producing the lowest pitch. With a range of
one octave below and two octaves above middle C, the
ATARI  can  produce  all  notes:  sharps,  flats,  and
naturals. In addition there are other tones available
that do not correspond to notes on the musical scale.
Although they are of no value in programming music, they
can be used as alarms or sound effects to accompany
other programs.

Instruction:
------------
You type:
---------

10 FOR FREQ = 10 TO 200 STEP 10
20 FOR VOL = 15 TO 0 STEP -1
30 SOUND 0,FREQ,10, VOL
40 NEXT VOL
50 NEXT FREQ

What did you hear? Did you notice the variations in
sound?

In line 10 of the above program, we used the variable
FREQ for determining the pitch. The range of values for
FREQ  were  selected  to  be  between  10  and  200.  In
addition, we selected STEP 10 to increase the pitch. By
using STEP 10, we made the variations in pitch more
apparent than it would have been had we selected STEP 2
or STEP 4, etc.

We also chose to decrease the volume. In line 20 we set
the range of values for the volume to decrease 15 to 0,
STEP-1.  Remember,  variables  may  decrease  as  well  as
increase.

Lines 40 and 50 simply close our FOR and NEXT loops.
This  allows  the  computer  to  automatically  make  the
changes in pitch and volume.

Working with the sound element of your Atari requires
practice. If you are musically inclined, the differences
in pitch and sound quality will be more apparent to you
than to someone with a less sensitive ear.

Even  if  you  cannot  distinguish  small  differences  in
sounds,  you  can  still  create  interesting  tones  to
accompany your programs.

Actually, the Atari permits such a wide range of sounds
that your imagination and creativity should be your only
limitation.

The following program simulates a space ship taking off.

Instruction:
------------
You type:
---------

10 FOR P=150 TO 100 STEP -1
20 FOR N=P TO P-100 STEP -5
30 SOUND 0,N,10,5
40 NEXT N
50 NEXT P
60 GOTO 10

In order to create the whirling sound of the space ship
ascending,  it  was  necessary  to  form  two  FOR...NEXT
loops.  In  line  10  we  establish  the  first  loop.  Our
variable  P  has  a  range  of  150  to  100,  and  it  will
decrease one number each time the loop is executed.

Line  20  starts  our  second  FOR...NEXT  loop.  A  new
variable N is initialized to have the same value as P.
Each time through the loop, the value of N will decrease
by 5 until it reaches the value of P-100.

In  order  to  produce  noise,  you  must  use  a  SOUND
statement, like the one shown in line 30. The first
number indicates the sound channel, in this case 0. The
next  number  is  our  variable  N,  which  indicates  the
pitch. The third number is sound quality; in this case
we used 10 which produces a pure tone. Finally, the last
number indicates volume.

In this example, the FOR...NEXT loops are nested. That
is the FOR N loop is inside the FOR P loop. Remember,
the  last  FOR's  variable  name  gets  the  first  NEXT's
variable name. In lines 40 and 50 we complete the loops
we began in lines 10 and 20. Notice that the value of N
will  change  more  often  than  the  value  of  P.  The
innermost FOR...NEXT loop must be completed before the
value of P is changed.

Let's  play  computer  to  see  how  this  program  would
actually work. The first time through the program the
value of P would equal 150 and the value of N would equal
150. When the computer reaches the SOUND statement in
line 30, it replaces N with the value of 150. When the
computer reaches line 40, it goes back to line 20 and
changes the value of N to decrease by 5. N then has the
value 145. The value of N will continue to decrease by 5
until it reaches a value of P-100 or 50. Once the value
of N equals 59, the computer will drop down to line 50.
Line 50 will cause the computer to return to line 10.
The value of P will now decrease by 1, and the value of P
will equal 149. When the computer drops down to line 20,
the value of N is now equal to 149. Again, the computer
will repeat the FOR N loop until the value of N is equal
to P-100. The value of N would now equal 49.

By using the GOTO statement in line 60, we create an
endless loop which will cause the computer to continue to
return to line 10. Because line 10 contains the FOR P
loop, the value of P is reinitialized each time the
computer executes line 60.

Exercises:
----------

1) Write a program which creates random sounds.

2) Write a program which creates random pure tones.

CHAPTER XXIII
-------------

            GRAPICS PART II:
                  I CAN WRITE BIGGER THAN YOU

Glossary:
---------

Graphics      The  large  area  of  the  monitor  or
Window        television screen in which graphics words,
--------      designs, or pictures can be displayed.

Text          An area at the bottom of the monitor or
Window        television  screen  which  contains  enough
------        area for four lines of text.

PRINT # 6;    A command used in GRAPHICS modes 1 and 2
-----------   that causes the characters to appear in
              the graphics window.

POSITION      A statement which moves the cursor to any
--------      specified  location  on  the  screen.  The
              location  is  designated  by  the  column
              number and the row number.

As stated earlier, the ATARI Home Computer is equipped
with  a  large  range  of  graphic  features.  Chapter  XIV
was intended to introduce you to the most common of
those aspects. This chapter and the ones that follow
will further explain the ATARI'S graphic capabilities
and the concepts necessary to create your own advanced
designs.

Like GRAPHICS 0, GRAPHICS 1 and GRAPHICS 2 are referred
to as the "text modes". They are used to display words
and characters. The difference is in the size of the
text.  Think  of  GRAPHICS  0  as  small,  GRAPHICS  1  as
medium, and GRAPHICS 2 as large.

Enter the new modes by typing GRAPHICS 1 or GRAPHICS 2.
Press {RETURN}. Notice that the display screen is now
split  into  two  segments.  The  first  segment  occupies
about five/sixths of the screen and is referred to as
the graphics window. The second section has enough area
to  hold  four  lines  of  text  and  is  called  the  text
window.

Instruction:
------------
You Type:
---------

GRAPHICS 1
PRINT "VARIETY IS THE SPICE OF LIFE"
PRESS {RETURN}

Your window text screen should look like this:

VARIETY IS THE SPICE OF LIFE
READY

These instructions allowed you to PRINT a line of normal
size text along the top of the text window. In order to
increase the size of the characters and to have them
displayed in the larger segment of the screen or the
graphics window, you must use a PRINT '67 statement.

Now type:
---------

PRINT #6;·"VARIETY IS THE SPICE OF LIFE"

Notice  the  size  of  the  characters  in  the  graphics
window.  To  see  even  larger  characters,  change  to
GRAPHICS 2.

You Type:
---------

GRAPHICS 2
PRINT #6; "VARIETY IS THE SPICE OF LIFE"

Again, you will see the characters are now even larger
than they were in GRAPHICS 1.

In both of our examples, the characters appeared in the
upper  left  hand  corner  of  the  screen.  For  creative
purposes  this  placement  may  not  suit  your  needs.  In
order  to  change  the  location  of  the  characters,  a
POSITION statement is used.

As you will recall in our discussion of GRAPHICS 0 we
compared the display screen to a piece of graph paper.
In GRAPHICS 1, the larger characters limit the screen
size  to  20  columns  by  20  rows.  (Remember  rows  run
horizontally while columns run vertically.) Because of
the even larger sized characters in GRAPHICS 2, there is
room for 20 columns and only 10 rows.

Now  you  have  the  amount  of  area  in  which  you  can
POSITION the characters. A POSITION statement is used in
conjunction  with  column  and  row  numbers.  The
POSITION  statement  does  not  visibly  move  the  cursor
location when it is executed. It does move, however,
when a subsequent statement such as PRINT accesses the
display screen.

You Type:
---------

GRAPHICS 1
POSITION 7,11
PRINT #6; "MIDDLE"

These instructions should place the word "middle" in the
center of the graphics window.

There are other variations you can use with GRAPHICS 1
and 2. In the text window, lower-case and/or inverse
video  characters  as  well  as  upper-case  may  be  used.
When  these  same  characters  are  used  in  the  graphics
window, however, they translate into color variations.
In the exercise below, first type A, the lower case A,
control A, inverse upper case A, inverse lower case A.
(Remember the ATARI key produces an inverse image.)

Instructions:
-------------

You type:
---------

GRAPHICS 1
PRINT #6; "AaAAa"

What did you see? Notice all the variations of color
available in GRAPHIC 1 and 2.

GRAPHICS  modes  1  and  2  give  your  creative  urges  an
additional outlet.  Although all  our examples  in this
chapter have been in direct mode, it is also possible to
use programming mode.

10 GRAPHICS 2
20 PRINT #6; "WHAT a difference this MAKES"
30 PRINT #6; "IT'S ALL RIGHT TO HAVE A SPLIT PERSONALITY"

In our program above, some information is displayed in 
the graphics window and some appears in the text window.

Exercises:
----------

1) Write a program which will PRINT in the graphics
window a person's name and "YOU ARE A STAR".

2) Write a program using the multiplication sign that
will look like a marquee. PRINT your name in the
middle of the marquee and center the marquee on the
screen.

CHAPTER XXIV
------------

           GRAPHICS PART III:
                     MY PICTURE IS IMPROVING

Glossary:
---------

Graphic       A state in which the computer responds to
Mode          instructions for the purposes of drawing
-------       pictures, designs, graphs, or variations
              of the standard characters.

Graphics      The  large  area  of  the  monitor  or
Window        television screen in which graphic words,
--------      designs, or pictures can be displayed.

Text          An area at the bottom of the monitor or
Window        television  screen  which  contains  enough
------        area for four lines of text.

Pixels        The shaded blocks of color used in graphic
------        modes  number  3  and  above  to  create
              pictures, designs, and graphs.

COLOR         A  statement  which  selects  one  of  the
-----         available color registers to be used with
              subsequent PLOTs and DRAWTOs.

PLOT          A  statement  which  illuminates  a  single
----          point on the screen.

DRAWTO        A  statement  which  causes  a  line  to  be
------        drawn  from  a  point  (the  last  plotted
              point) to a specified location.

LOCATE        A statement which allows a program to find
------        out what color is already plotted at any
              given point on the screen.

This chapter  discusses the  GRAPHICS statement  in two
parts. First, we discuss graphic modes 3, 5, and 7.
Although GRAPHICS 7 was previously described in Chapter
XIV, it is worth repeating here for comparison to modes
3  and  5  and  for  the  sake  of  completeness.  Second,
graphics modes 4 and 6 are introduced a little later
because of differences between them and the other modes.

Pixels in Modes 3, 5, and 7
---------------------------

The  big  difference  between  modes  3,  5,  and  7  when
compared to modes 0, 1 and 2 is that we are now dealing
with pixels instead of characters. Recall that pixels

are shaded blocks of color. Earlier we described the
screen in graphics mode as being similar to a piece of
graph paper. We noted that a pixel is equivalent to one
block being filled with color.

In addition, graphics modes 3, 5, and 7 differ from each
other in the number of pixels available. In general,
the  higher the number in graphics mode, the more pixels
or shaded squares are available. In terms of picture
quality GRAPHICS 7 provides finer lines than GRAPHICS 3.
GRAPHICS 7 has better resolution, but GRAPHICS 3 makes
it easier to produce bar graphs. Each graphic mode has
its own special abilities. As you become more familiar
with the various modes, you will have an easier time
choosing which mode best suits your needs.

The  chart  below  summarizes  the  pixel  size
characteristics of each of the five modes discussed in
this  chapter  (modes  4  and  6  are  included  here  for
convenience, even though they are not discussed until
later).

Graphics       Number of       Number of       Total #
  Mode           Rows           Columns       of Pixels
--------       ---------       ---------      ---------

   3               20              40            800
   4               40              80           3200
   5               40              80           3200
   6               80             160          12800
   7               80             160          12800

Pictures in Modes 3, 5, and 7
-----------------------------

The choices of colors available in graphics modes 3, 5,
and 7 are the same. These colors are as follows:

                  COLOR 0 = black
                  COLOR 1 = orange
                  COLOR 2 = aqua
                  COLOR 3 = blue

The color black is used for background.

When using GRAPHICS modes 3·through 7, PLOT and DRAWTO
statements are usually used. As stated in Chapter XIV,
the  PLOT  statement  enables  the  user  to  select  a
particular point position. The first number after PLOT
tells the computer the column desired: the second number
indicates the desired row. Numbers in a PLOT statement
are separated by a comma.

When  PLOT  and  DRAWTO  are  used  in  pairs,  the  PLOT
statement indicates the starting point of a line while
the  DRAWTO  statement  indicates  the  ending  point.  As
with PLOT, when you use DRAWTO you must specify two
numbers (separated by a comma) to indicate the desired
end position as a column and row position.

When  a  DRAWTO  follows  a  previous  DRAWTO  without  an
intervening PLOT, the ending position specified by 'the
first  DRAWTO  becomes  the  starting  position  for  the
second one.

(If  you  need  more  information  concerning  PLOT  and
DRAWTO,  refer  to  Chapter  XIII  and/or  your  BASIC  XL
Reference Manual.)

Instruction:
------------
You type:
---------

NEW
10 GRAPHICS 3
20 COLOR 1
30 PLOT 0,0
40 DRAWTO 39,0
50 COLOR 2
60 DRAWTO 39,19
70 COLOR 3
80 DRAWTO 0,19
90 COLOR 0
100 DRAWTO 0,0
RUN

The  program  above  gives  directions  for  drawing  a
rectangle. Was the image on your screen a three-sided
or  a four-sided object? Actually, all four sides were
drawn. Why did the fourth side appear to be missing?
In  GRAPHICS  3,  5,  and  7  there  are  three  available
foreground colors. The fourth color (COLOR 0) is always
the background color and so is indistinguishable from
the background when it is plotted or drawn.

Notice  also  how  much  screen  area  is  covered  by  our
program. Try to keep that image in your mind.

Instruction:
------------
You Type:
---------

10 GRAPHICS 5 {RETURN}

We are going to change graphics modes.

In  order  to  do  that,  we  need  only  change  line  10.
Remember, the computer disregards our previous line 10
and keeps the rest of the program intact.

Instruction:
------------
You Type:
---------

RUN

How does our object change? It becomes smaller. Why?
In GRAPHICS 3, the individual pixels covered a large
area, in GRAPHICS 5 there are more pixels available, but
each individual pixel is smaller.

Instruction:
------------
You Type:
---------

10 GRAPHICS 7 {RETURN}
RUN

Again  our  object  became  smaller.  The  lines  of  color
created in GRAPHICS 7 are finer than those in GRAPHICS
3. However, in GRAPHICS 7 more lines would have to be
drawn to occupy the area of one line in GRAPHICS 3.
GRAPHICS 3 is similar to drawing with a magic marker
while GRAPHICS 7 represents a fine point pen. GRAPHICS
5 fits in the middle like a felt tip pen.

Graphics Modes 4 and 6
----------------------

Graphics modes 4 and 6 have the same screen size as
graphics  modes  5  and  7,  respectively  (see  our  chart
above). The big difference between these modes is in
the  number of available colors. GRAPHICS 4 and 6 have
only  two  available  colors.  One  color  is  used  for
foreground and one is used for background.

Instruction:
------------
You Type:
---------

10 GRAPHICS 4
RUN

What happened to the object? How many sides can you

see?  Although  our  original  program  specifies  four
colors, only two are available in GRAPHICS 4. Each time
the computer encounters an even numbered color, it uses
the  available  foreground  color;  odd  numbered  colors
become black.

Instruction:
------------
You Type:
---------

10 GRAPHICS 6
RUN

You will notice a smaller sized object in one color.
Because of the limited number of colors, GRAPHICS 4 and
GRAPHICS 6 are rarely used.

The LOCATE Statement
--------------------

When we have a picture in mind which we want to draw on
the screen, we usually use COLOR, PLOT, and DRAWTO to
accomplish  our  ends.  But  suppose  there  is  a  picture
already on the screen and we want to "describe" it to
the  computer.  Of  course,  one  way  would  be  to  get  a
magnifying glass and start typing in pixel locations and
colors  in  response  to  INPUT  statements,  but  this  is
obviously ridiculously slow, tedious, and error-prone.

Fortunately, BASIC XL provides a method - the LOCATE
statement - by which the computer can directly "read"
the screen itself. The form of this statement is:

        LOCATE column, row, color

and examples of use might include:

        LOCATE X,Y,C
        LOCATE HORIZONTAL/2+10,VERTICAL/2+5,LOCCOLOR

While the column and row may be specified by arithmetic
expressions (just as with PLOT and DRAWTO), the "color"
MUST be an arithmetic variable. When a LOCATE statement
is executed, BASIC XL causes the computer to "read" the
pixel  located  at  the  specified  column  and  row.  It
places the value it reads into the variable which you
supply.  And,  not  surprisingly,  the  data  read  is
actually the color of the screen at that point. Note
that this color value is the same value used in the
COLOR statement which is then subsequently placed on the
screen by PLOT or DRAWTO.

Instruction:
------------
You type:
---------

10 GRAPHICS 3
20 FOR COUNT=1 TO 400
30 COLOR RANDOM(1,3)
40 PLOT RANDOM(40),RANDOM(20)
50 NEXT COUNT
60 PRINT "NOW CONVERTING"
70 FOR COLUMN=0 TO 39
80 FOR ROW=0 TO 19
90 LOCATE COLUMN,ROW,SCREENCOLOR
100 COLOR 2
110 IF SCREENCOLOR=O THEN PLOT COLUMN,ROW
120 NEXT ROW
130 NEXT COLUMN
140 PRINT "FINISHED"

Let's analyze this program by line number.

10 - We select the largest pixels available, to make our
program more visible.

20 to 50 - We simply plot a lot of points on the screen
using randomly chosen colors and locations.

60 - Just a message (which will be displayed in the text
window) to tell us when the program gets to this point.

70 and 130 - A loop, whereby we check all the columns on
the screen.

80 and 120 - Another loop, but here we check all the
rows within each of the columns.

90 - We check a particular location to find out what
color is on the screen.

100  and  110  -  If  the  location  we  checked  had  not
previously been plotted, we change it to COLOR 2 via the
PLOT statement.

140 - We're done.

The effect of this program is to sprinkle various colors
on  the  screen.  Then  we  go  back  and  change  all  the
"background"  pixels  to  another  color.  A  relatively
useless program, but it illustrates the principles of
PLOT, COLOR, and LOCATE well.

Exercises:
----------

1) Modify the exercise at the end of Chapter XIV so
that  the  programs  RUN  in  GRAPHICS  3  and  then  in
GRAPHICS 5.

2) Modify lines 100 and 110 of the last example above
to change the color of EVERY pixel on the screen.
The new colors should depend on the old colors as
follows:

                old color       new color
                    0               3
                    1               2
                    2               1
                    3               0

Hint: Don't use an IF statement to determine the new
color. Work out an arithmetic expression instead.

CHAPTER XXV
-----------

               GRAPHICS PART IV:
                    ALL THE COLORS OF THE RAINBOW

Glossary:
---------

Color      The memory locations which set the colors used
Registers  for foreground, background, and border colors.
----------

COLOR      A statement which selects one of the available
-----      color registers and with that color draws.

SETCOLOR   A statement which allows the user to change
--------   colors on the screen at any time.

Luminance  The brightness or the radiation or reflection
---------  of light.

Hue        The color or the particular shade or tint of a
           color.

For many people, this chapter will prove difficult and
confusing.  Try  to  stick  with  us  through  the  charts,
explanations,  and  examples.  We  believe  that  it  will
become clear in the end.

In our previous chapters on graphics we have discussed
the  COLOR  statement.  In  review,  we  said  a  COLOR
statement  is  one  which  selects  an  available  color
number and that subsequent PLOTs and DRAWTOs use that
color number to place pixels on the screen. For example,
in  modes  3,  5,  and  7,  it  was  possible  to  select
three foreground  colors and  one background  color for
PLOT and DRAWTO statements.

But the Atari Computer and BASIC XL are capable of much
more complex operations. To try to find a real word
analogy, let's think of a COLOR  number as selecting a
particular  spot on an artist's palette, regardless of
what shade of paint is on the spot. Well, just as an
artist might not place the same color on the same spot
on his palette each time he paints, so does BASIC XL
(or,  more  properly,  the  Atari  Computer's  hardware)
change which shade a particular COLOR number refers to.

To continue our analogy, we might say that the computer
simply changes which jar of paint it dumps onto each
spot.  The  situation  gets  more  complex:  you,  as  the
programmer, can change the color of the paint in the jar

(by  using  the  SETCOLOR  statement).  And,  just  as  the
COLOR statement is not really aware of what shade is
actually "painting", so is the computer not really aware
of What shade is in the jar.

The net result of all this is that you really do have
control of what color goes where on the screen. The
only problem is that you have to follow the rules about
what COLOR number refers to what palette spot (which can
vary depending on the GRAPHICS mode chosen) and what
SETCOLOR refers the "jar" which is used for that palette
spot. All of the options for graphics modes 3 through 8
are  presented  in  table  form  below  (Figure  2).  But
before we discuss the table, we will investigate the
workings of SETCOLOR, the statement which lets us change
the colors in the jars.

The SETCOLOR Statement
----------------------

Following  SETCOLOR  are  numeric  expressions.  Separated
by commas,' these numbers determine which color register
to change, to which hue to change it, and what degree of
brightness the hue will be.

The  first  number  following  SETCOLOR  indicates  which
register to set. The registers are numbered from 0 to 4
and relate to the jars of paint in our analogy. To find
out which palette spot they refer to you must consult
Figure 2, below.

The second  number after  SETCOLOR determines  the hue.
The sixteen hues are numbered 0 to 15 and are usually
visibly as follows, though there can be variation from
one television screen or monitor to another.

0 Grey                          8 Light Blue  
1 Gold                          9 Blue-Green  
2 Orange                       10 Aqua        
3 Red                          11 Green-Blue  
4 Pink                         12 Green       
5 Violet                       13 Yellow-Green
6 Blue-Purple                  14 Orange-Green
7 Blue                         15 Orange      

Figure 1: Available Hues
------------------------

The  last  number  following  SETCOLOR  represents  the
brightness or  luminance. Although  numbered from  0 to
15, the even-numbered values are the only settings which
are meaningful. The 0 represents the darkest value and
14  represents  the  brightest  value.  (Odd  numbers  are
treated the same as the next lower even number.)

Instruction:
------------
You Type:
---------

100 GRAPHICS 3
110 COLOR 1
120 PLOT 0,10
130 DRAWTO 39,10
140   FOR HUE=0 TO 15
150     FOR LUM=0 to 14 STEP 2
160     PRINT HUE, LUM
170     SETCOLOR 0, HUE, LUM
180       FOR WAIT=l TO 500
190       NEXT WAIT
200     NEXT LUM
210   NEXT HUE
220 END
RUN

If you have a color monitor or television screen for
your ATARI home computer, the above program will produce
each of the sixteen colors available and in addition
will produce  eight different  degrees of  luminance or
brightness.

Let's analyze the structure of this program. To begin
with we must choose a graphics mode. Graphics 3 was
selected  for  two  reasons:  1)  because  it's  degree  of
resolution allows for excellent bars to be drawn: and 2)
because it contains many color choices.

Next,  we  must  PLOT  a  point  and  also  use  the  DRAWTO
statement.  These  two  statements,  used  in  conjunction
with COLOR, select a point and create a bar of color. We
could have used only one or two points. However, it
would almost be impossible to see the changes in the
color and luminance.

In  line  140  we  have  begun  a  FOR...NEXT  loop.  Our
variable  is  HUE  and  represents  the  various  color
selections available. We could have named the variable
almost anything, but HUE is more meaningful than just a
letter or another word. Also, note the value of our
variable HUE is equal to the sixteen available color
choices.

Again, in line 150, we have begun a FOR...NEXT loop.
This  time  our  variable  is  named  LUM  and  represents
luminance  or  the  various  degrees  of  brightness
available. Earlier we said that only the even values of
luminance were meaningful. In our program, the value of
our variable LUM is equal to the degrees of brightness
available. We have added STEP 2 to our FOR statement so
that the value of LUM will increase in even increments.

In  order  to  see  which  color  and  which  degree  of
brightness is currently on the screen, we have included
the PRINT statement in line 160. Finally, in line 170 we
get to the point of all of this. Here is our SETCOLOR
statement.  As  the  values  of  our  variables  HUE
and LUM change, so will the colors of the bar on the
screen.  This  is  possible  because  of  our  SETCOLOR
statement. Notice, though, that the SETCOLOR uses color
register 0 while our PLOT statement used COLOR 1. How
come? Again, it all has to do with palette "spots",
paint jars, etc. Be sure and see Figure 2, below, for a
clarification of these confusing numbers.

In line 180 we begin yet another FOR loop. We called
our variable WAIT and made its value equal to from one
to  500.  Although  the  program  will  display  a  bar  of
color and the various changes in HUE and LUM, without
this FOR...NEXT  loop, these  changes would  occur very
rapidly.  By  adding  a  FOR...NEXT  loop  whose  only
function is to keep the image on the screen, we allow
the user an opportunity to view the bar of color and its
changes  for  a  reasonable  amount  of  time.  We  could
change the value of our variable WAIT depending on our
desire to lengthen or shorten the amount of time the
various colors are shown.

Whenever you use more than one FOR...NEXT, you must nest
them. That is, you put one inside of another, like a
nest of tables or a set of mixing bowls. Although you
may use as many FOR •••NEXT loops as you desire, they
must be built from the inside out. The last FOR gets the
first NEXT.

                   FOR HUE
                     FOR LUM
                       FOR WAIT
                       NEXT WAIT
                     NEXT LUM
                   NEXT HUE

In  lines  190,  200,  and  210,  we  complete  all  of  our
FOR...NEXT loops. Our automatic counters are complete
and the values on our variables automatically change.
Thus, we have the changes in color and brightness.

The Relationship of COLOR to SETCOLOR
-------------------------------------

Finally, we  present the  long-awaited Figure  2, which
shows in  tabular form  the relationship  between COLOR
number  and  SETCOLOR  register.  We  will  present  the
figure first and then explain it.

Graphics    Color   Setcolor  Default  Default
  Mode      Number  Register  Hue,Lum   Color
--------    ------  --------  -------  -------
3, 5, 7        0        4       0,0     Black
               1        0       2,0     Orange
               2        1      12,10    Green
               3        2       4,6     Dark Blue
-------------------------------------------------
4, 6           0        4       0,0     Black
               1        0       2,8     Orange
-------------------------------------------------
8, 0           0        2       4,6     Dark Blue
               1        1*      4,10*   Light Blue
--------------------------------------------------
               (* In mode 8, the hue of SETCOLOR
                  register 1 is NOT selectable -
                  it will always be the same as
                  the hue or register 2.)

       Figure 2: COLOR/SETCOLOR Table
       ------------------------------

To use this chart, follow these steps.

1. Determine the GRAPHICS mode you are going to use.
Use only the portion of the table which applies to
that mode.

2. If  the  colors  given  in  the  rightmost  column  are
adequate to your needs, simply use the digit from
the  Color  Number  column  in  your  program  (in
conjunction with COLOR, of course).

3. If you like some of the colors, reserve their  Color
Numbers for use in your program.

4. Use SETCOLOR to choose a new color(s) by varying the
hue and luminance to your choice(s) using any of the
Setcolor  Registers  which  are  available  in  your
chosen  graphics  mode.  Then  use  the  Color  Number
which relates to that register in your program.

Instruction:
------------
You Type:
---------

10 GRAPHICS 3
15 LET LCOLOR=1
20   FOR HORIZ=0 TO 39
30   COLOR LCOLOR
40   PLOT HORIZ,0
50   DRAWTO HORIZ,23
60   LET LCOLOR=LCOLOR+1
70   IF LCOLOR>3 THEN LCOLOR=1
80 NEXT HORIZ
100 LET BAR1=4
110 LET BAR2=8
120 LET BAR3=12
200 SETCOLOR 0,8,BAR1
210 SETCOLOR 1,8,BAR2
220 SETCOLOR 2,8,BAR3
230 LET TEMP=BAR1
240 LET BAR1=BAR2
250 LET BAR2=BAR3
260 LET BAR3=TEMP
300   FOR WAIT=1 TO 50
310   NEXT WAIT
400 GOTO 200
RUN

What results do you get from this program? How was the
movement created? As with marquee lights, the bars are
not moving although they appear to be. The movement is
created by changing the colors of the vertical bars.

Because we are using bars, we again use GRAPHICS 3. As
you  become  more  familiar  with  each  of  the  graphics
modes, it will become easier for you to determine which
mode will best suit your needs.

In  line  15  we  initialize  our  variable  LCOLOR  at  1.
Because we are in GRAPHICS 3, we have a choice of three
foreground colors. In order to create the movement we
will want the colors to change. Therefore, we use a
variable.

Next, we begin a FOR...NEXT loop. Since the screen size
in  GRAPHICS  3  is  39  columns,  our  automatic  counting
variable HORIZ is set for 0 to 39. This accounts for
all the columns on the screen.

--139--



Line 30 represents our COLOR statement. The value of
COLOR will be set to the value of our variable LCOLOR.
Each  time  the  value  of  LCOLOR  increased,  our  COLOR
statement will change.

In order to create an image, in lines 40 and 50, we PLOT
and  DRAWTO.  Notice  that  the  point  that  we  PLOT  is
dependent upon the value of our variable HORIZ. Also,
you can see by using 0 as the row value, the points
PLOTed will be across the top of the screen. By using
the value of the variable HORIZ in the DRAWTO statement,
we cause a line to be drawn from each value of HORIZ.
In other words, we cause a line to be drawn from the top
of each column to the bottom of each row.

We use our non-algebraic expression
         LET LCOLOR = LCOLOR + 1
to change the value of the variable LCOLOR. Each time
the computer executes line 60, it increases the value of
LCOLOR by1.

Why did we include line 70? As you may remember, in
GRAPHICS 3, there is a choice of three foreground colors
and one background color. If we did not include line
70, sometimes (every fourth time), we would get black
bars.  To  avoid  using  black,  we  use  an  IF...THEN
statement along with our relational operator > (greater
than). When the value of LCOLOR is greater than 3, the
computer resets its value to equal 1.

In line 80, we complete our FOR...NEXT loop which was
started in line 20. The value of our variable HORIZ
changes from 0 to 39 in increments of one.

If you add the line, 90 STOP, you will be able to see
the bars of color across the screen. Remember to erase
line 90 typing:

         90 [RETURN}

In lines 100, 110, and 120 we are initializing three new
variables  called  BAR  1,  BAR  2,  BAR  3  respectively.
Notice  the  values  chosen  for  these  new  variables.
Remember  in  a  SETCOLOR  statement  the  last  number
represents brightness or luminance, and only even values
between 0 and 14 are meaningful. The values for the
variables BAR 1, BAR 2, and BAR 3 were chosen on that
basis. To change colors within the program we use our
SETCOLOR  statement.  The  first  number  represents  the
color  register.  The  second  delineates  the  color
desired; in this case, color 8, light blue. The last
number determines brightness which depends on the value
of our variables BAR 1, BAR 2, and BAR 3.

In order to get the movement, we are going to change the
brightness value. To do this lines 230, 240, 250 and 260
are set up to rotate the luminance value. Each time
these  lines  are  executed,  their  values  will  change.
However, the values will always be either 4, 8, or 12.

Finally, in line 300 and 310 we set up another FOR/NEXT
loop. The purpose of this loop is to allow the image to
stay on the screen so that the viewer can enjoy the
color changes.

In line 400, the GOTO statement continues the process.
The bars of color remain the same with the SETCOLOR
statements continuously re-executed (executed again and
again).

Exercises:
----------

1) Write a program to randomly change the background
color for text mode (GRAPHICS 0) continuously.

2) Write the same program for a display in GRAPHICS 3.

CHAPTER XXVI
------------

            GRAPHICS PART V:
                  A REVOLUTION IN RESOLUTION

Glossary:
---------

GRAPHICS 8    A special two color graphics mode with a
----------    high level of resolution.

Luminance     Brightness or the radiation or reflection
---------     of light.

Hue           Color or the particular shade or tint of
---           a color.

GRAPHICS B is discussed as a separate unit because it is
significantly  different  from  the  other  graphic  modes
previously covered.

As you might imagine, GRAPHICS B provides the highest
level of resolution. In split-screen mode - that is the
graphics window with text window - the screen size is 320
rows by 160 columns. (Remember, rows run horizontally
and columns run vertically.) GRAPHICS 8 is most useful
for detailed drawings and pictures. This mode may also
be used for games which require a large area to be used
as a playfield. Thus, GRAPHICS a is most useful where
detailed drawings are important.

The reason for the high level of resolution is due to
the size of each pixel. In graphics mode a, each pixel
is  smaller  than  it  was  in  the  previously  discussed
modes. The pixels in this mode look more like points
than squares (even though they really are still square).
This difference allows for more detailed drawings.

Instruction:
------------
You type:
---------

NEW
10 FOR MODE = 3 TO 8
20 GRAPHICS MODE
30 PRINT "MODE: ";MODE
40 SETCOLOR 2,8,0
50 COLOR 1
55 PLOT 0,0
60 DRAWTO 0,19
70 DRAWTO 39,19
80 DRAWTO 0,0
90 FOR I=1 TO 1000: NEXT I
100 NEXT MODE
RUN

The  above  program  displays  the  same  triangle  in  six
different  modes.  Notice  that  in  mode  3  almost  the
entire  screen  is  filled.  As  the  number  of  the  mode
increases, the resolution also increases. The size of
the triangle becomes smaller, and the lines are finer
and more distinct.

When the program reached GRAPHICS 8, what differences
did you notice? First, the lines were very fine, and
the  triangle  as  the  object  was  clearer  and  more
distinct. What happened to the split screen mode? Did
it  seem  to  disappear?  GRAPHICS  8  has  special
characteristics.  Because  of  these  special  traits,  it
seems as though the screen is no longer split.

Unlike  other  graphics  modes,  the  foreground  and
background colors are NOT independent. By foreground we
mean  the  area  in  which  you  would  see  the  text  in
GRAPHICS 0 or where you would see the design, picture,
or drawing in modes 3 through 7.

In Graphics mode 8, the foreground and background colors
are always the same hue. Obviously, this means that if
the luminance of the two is the same or similar, it is
hard  to  distinguish  one  from  the  other.  However,  by
using luminances which are as different as possible, the
result can be quite "distinct".

--------------------------------------------
| Note:   Actually,   the    colors    and |
| luminances   that  can  be  selected  in |
| GRAPHICS 8 are identical with the colors |
| and luminances available in mode 0,  the |
| primary text mode of the Atari computer. |
| You   might  try  some  of  the  various |
| SETCOLORs we did here in mode 0 as well. |
--------------------------------------------

Instruction:
------------
You type:
---------

GRAPHICS 8
PLOT 10,10
DRAWTO 100,100

What appeared on your monitor or screen? It should have
been a diagonal line. If you have a hard time seeing it,
it's not your eyes. The line really does not show up
very well.

Instruction:
------------
You type:
---------

SETCOLOR 2,8,0

Can  you  see  the  line  more  clearly  now?  How  did  the
SETCOLOR statement change the line?

The SETCOLOR statement in GRAPHICS 8 is very important.
As you will recall from Chapter XXV, the first number in
a  SETCOLOR  statement  fills  the  color  register.  In
graphics mode 8, the user has a choice of three SETCOLOR
registers: 1,2 or 4.

Instruction:
------------
You type:
---------

GRAPHICS 8
PLOT 10,10
DRAWTO 100,100
SETCOLOR 1,8,15

Can you still see the line? With the above SETCOLOR
statement,  you  can  see  the  line,  but  it  is  not  as
dramatically clear as it was in the previous SETCOLOR
statement.  Why?  SETCOLOR  1  affects  the  foreground
luminance only. In this case, the hue is ignored. That
means the brightness of the foreground can be altered,
but nothing else can be changed.

A SETCOLOR 2 statement affects the background hue and
luminance.  That  means,  the  background  can  be  made
darker or lighter. In our statement, SETCOLOR 2,8,9, we
made the background as dark as possible. The line looks
brighter and  more distinct  because the  background is
darker.

A SETCOLOR 4 affects only the border hue and luminance.
Note this is exactly the same manner in which hue and
luminance work on characters in graphics mode 0.

Exercise:
---------

1) Write a program to draw a border around the edge of
the mode 8 screen.

CHAPTER XXVII
-------------

              PROGRAMMING AIDS:
                           COSMETIC SURGERY

Glossary:
---------

REM     Is  a  statement  which  provides  information
---     concerning the program in a LISTing.

NUM     Is a statement which provides automatic program
---     line numbering.

RENUM   Is  a  statement  which  automatically  renumbers
-----   the program lines of an existing program.

The people who write languages like BASIC XL (systems
programmers)  understand  the  may  problems  beginning
programmers may  encounter. To  help alleviate  some of
those troublesome areas, they include special features
in  their  languages.  BASIC  XL  has  many  built-in
features. In this chapter we will discuss a few which
are  particularly  useful  to  the  beginning  BASIC  XL
programmer.

Have you ever found a note written to yourself? You
look at it and for the life of you, you just cannot
imagine what you meant by it? The same thing can happen
with programs you write.

You may not have any trouble remembering what a program
of  less  than  10  lines  does.  For  example,  do  you
remember this program from Chapter 10?

10 LET I = 1
20 PRINT RANDOM(1,100)
30 LET I = I+1
40 IF I < 7 THEN GOTO 20

In case you do not, this program will produce 6 random
numbers.  A  much  easier  way  to  remember  would  be  to
write the program like this:

5 REM THIS PROGRAM PICKS 6 RANDOM NUMBERS
10 LET I = 1
20 PRINT RANDOM(1,100)
30 LET I = I+1
40 IF I < 7 THEN GOTO 20

As you continue to write more and more programs and the
length of those programs grows, it will become more and
more difficult for you to remember the details of each
program.

When the length of your programs are 100 statements or
more, it is likely that you will forget exactly what
each program does.

A REM or REMARK statement is one in which the programmer
makes a note concerning the nature of a program or a
certain section of the program. The only time a REM
statement is seen is when a LISTing of the program is
produced.

When the computer encounters a REM statement, it ignores
everything on  the line  following REM  and immediately
skips to to the next computer line.

Examples of REM statements include:

10 REM THIS IS A COMMENT
15 REM A COUNTING PROGRAM
100 REM THIS SECTION BEGINS AN ENDLESS LOOP

The length of a REMark or REM statement may be as short
as you wish or as long as three lines. If your comment
is  longer  than  three  lines,  you  must  begin  another
program line with REM. As long as you begin a program
line with REM, the information entered will not affect
the program's operation.

Often, beginning programmers will forget to include line
numbers. They are so intent on writing the program that
they forget the number. The line is executed instead of
being added to the program, sometimes with disastrous
results. Boy, does this cause some frustration!

This may have happened to you already. Perhaps, you were
typing  in  one  of  our  examples  and  made  a  mistake.
You corrected the mistake, but forgot to include the
line number.

To  help  avoid  this  problem  and  to  make  programming
easier,  BASIC  XL  comes  with  a  statement  called  NUM.
This statement will automatically number each program
line. To use NUM, simply type NUM and push {RETURN}.
First, you will notice the READY prompt, and where you
would normally expect to find the cursor, will be a line
number. Next, on the same line, you will see a blank
space and the cursor. You may now type in a program
statement.

NUM may be used in four different ways. First, NUM may
be  used  alone.  This  will  will  produce  program  lines
beginning with 10 and will continue in increments of 10.

NUM

Or you may use NUM followed by a single number. When
the computer finds NUM and a single number, it will
begin the first program line with the specified number.
For example:

NUM 3

The computer begins numbering with 3 and adds lines in
increments of 10. The first line would be 3 followed by
13, 23, 33, etc.

If NUM is followed by two numbers separated by a comma,
the computer will begin numbering line with the first
specified number and increase in increments specified
by the second number.

NUM 1000,2

begins  numbering  with  line  1000  and  increases  in
increments of 2. The first line would be 1000 followed
by 1002, 1004, 1006, etc.

Finally, NUM may be followed by a comma and a single
number.  In  this  case,  the  computer  would  start
numbering program lines with 10 and would increase in
increments specified by the number following the comma.

NUM ,4

would begin numbering with 4 and increase in increments
of  4.  The  first  line  would  be  4  followed  by
8, 12, 16, 20, etc.

What happens if a program is in the computer's memory
and you want to use NUM?

Instruction:
------------
You type:
---------

10 GOTO 20: REM LINE 10
20 GOSUB 30 :REM LINE 20
NUM

What happens to the program lines? By using the command
NUM, you force the computer to NUMber your program lines
for you. The computer begins with line 10 and increases
in  increments  of  10.  The  number  30  and  the  cursor
should  now  indicate  where  to  begin  the  next  program
line.

Instruction:
------------
You type:
---------

30 IF TRUE THEN GOTO 10 :REM LINE 30
NUM 15,5
15 []
REM THIS IS THE ORIGINAL LINE 15

(Please keep this program in the computer's memory until
you complete the section on RENUM.)

When you type in NUM 15,5, the computer displays the
number 15 followed by a blank space and the cursor. You
then add the REM statement shown above. When you press
{RETURN}, what  happens? The  ready prompt  appears and
the computer does not continue to number lines. This is
due to the fact that a line 20 already exists in the
program.

NUM will stop to avoid a collision with any program line
already in the computer's memory. You will recall from
previous  discussions  that  each  line  number  must  be
distinct.  If  a  line  number  is  repeated  the  computer
will only store the line most recently entered.

RENUM is closely related to NUM.

Instruction:
------------
You type:                              Computer Responds
---------                              -----------------

RENUM
LIST
                                        10 GOTO 30
                                        20 REM THIS IS
                                           THE ORIGINAL
                                           LINE 15
                                        30 GOSUB 40
                                        40 IF TRUE THEN
                                           GOTO 10

                                        READY
                                        []

Notice what happens to the line numbers. The original
line 10 is still there. However, the GOTO 20 statement
has been changed to GOTO 30. Why? Because originally,
the GOTO specified line 20, but it (line 20) has been
RENUMbered  to  line  30.  Also,  what  was  line  15  has
become line 20.

RENUM automatically changes the numbers of all GOTO and
GOSUB statements.

When used alone, RENUM begins with line number 10 and
increases in increments of 10.

When  RENUM  is  followed  by  one  number,  the  computer
begins with the specified number and continues to number
program lines in increments of 10.

RENUM 1

begins with line 1 and increases in increments of 10.
In  the  above  example,  the  first  line  would  be
1, 11, 21, 3l, etc.

RENUM, when followed by a comma and a number begins
numbering line with 10 and continues to number lines in
increments of the specified number.

RENUM ,8

begins with 10 and increases in increments of 8. The
first line would be 10 followed by 18, 26, 34, etc.

Finally, RENUM followed by two numbers separated by a
comma  will  begin  numbering  lines  with  the  first
specified number and will continue to number lines in
increments of the specified second number.

RENUM 1000, 4

begins with the number 1000 and increases in increments
of 4. The first line would be 1000 followed by 1004,
1008, 1012, etc.

Exercises:
----------

1) Use NUM to enter a program. The program doesn't
need to make sense, but be sure to include some GOTO or
GOSUB statements.

2) Use  RENUM  on  your  program.  Try  various
comhinations of starting line number and increment. Be
sure to LIST your program each time to see the effects
of RENUM.

CHAPTER XXVIII
--------------

              THE JOYSTICK:
                      MANUAL OR AUTOMATIC

Glossary:
---------

HSTICK       A function which reads the positions of the
------       joystick in terms of horizontal positions.

VSTICK       A function which reads the positions of the
------       joystick in terms of vertical positions.

STRIG        A  function  which  determines  if  the  red
-----        button or trigger on the joystick is being
             depressed.

Throughout this book we have attempted to increase your
program skills. In our introduction we indicated that a
video game would be the final chapter of this book. But
as  you  probably  know,  a  video  game  often  uses  a
joystick. A joystick is a device which may be attached
to  your  Atari  Home  Computer.  Actually,  your  home
computer  is  capable  of  handling  four  joysticks.  The
joysticks are attached in the front part of the computer
just  below  the  keyboard.  Each  has  its  own  number
starting with 0, 1, 2, and 3. The joystick is often used
to  play  games  on  your  home  computer  and  was  first
popularized with the Atari 2600.

In our explanations of the various programs throughout
this book, we have stressed the need to consider all
possibilities. We have asked you to RUN your programs
with diverse responses. This was done to insure that you
had allowed for all possible answers.

By using the joystick, you remove the keyboard as the
means of interaction between the user and the computer.
This limits the number of possible extraneous choices.
Each potential choice requires that the programmer check
for an inappropriate answer. If this is not done, your
program will have "bugs" in it. In case you have not
heard this term before, a bug is a programming problem.
Good programs do not have "bugs".

--------------------------------------
|  But  remember  'Richard's  Rule:  |
|  Every significant  program has a  |
|  bug in it.                        |
|  There's also Card's Corollary: A  |
|  program   without   a   bug   is  |
|  insignificant.                    |
--------------------------------------

VSTICK  and  HSTICK  are  functions  which  allow  the
programmer to  use the  joystick. These  functions read
the joystick positions. The joystick permits movement
that is up and down or right and left. In addition, the
computer interprets the movement of the joystick as a
series of numbers between 5 and 15. This limits the
number of user choices.

VSTICK indicates the vertical position of the joystick,
while  HSTICK  indicates  the  horizontal  position.
HSTICK(0) reads joystick number zero (which, strangely,
is marked joystick 1 on the computer). If the stick is
pushed left, the function "reads" a minus one for use by
your program. If it is pushed right, your program reads
a plus one. If it is in the middle, you get a zero.
Try this with a joystick plugged into the first joystick
socket:

10 PRINT HSTICK(0)
20 GOTO 10
RUN

Push the stick left and right and see what the program
PRINTs.  (Hit  {BREAK}  to  quit.)  Of  course,  HSTICK(1)
reads the second joystick, HSTICK(2) reads the third and
HSTICK(3) reads the fourth (if your machine has a third
and a fourth socket).

VSTICK is similar in its workings. It reads plus one,
zero,  or  minus  one  if  the  joystick  is  pushed  up,
centered,  or  pushed  down,  respectively.  Again,  try
this:

10 PRINT VSTICK(0)
20 GOTO 10
RUN

Another function used in conjunction with the joystick
is  STRIG.  The  STRIG  function  is  used  to  indicate
whether the red button or trigger of the joystick is
depressed.

For reasons having to do with the hardware of the Atari,
the STRIG function returns a zero when the stick trigger
is pushed and a one when it is not. Try this little
experiment.

10 PRINT STRIG(0)
20 GOTO 10
RUN

Push the button on your joystick and see what happens.

Using these functions, it is relatively easy to write
programs using the joystick.

Instruction:
------------
You type:
---------

NEW
5 REM THIS PROGRAM DRAWS USING THE JOYSTICK
10 GRAPHICS 7
20 COLOR 1
30 LET X=0 : LET Y=0
40 PLOT X,Y
50 LET X= X+HSTICK(0)
60 LET Y=Y-VSTICK(0)
70 IF X < 0 THEN X=0
80 IF X > 159 THEN X=159
90 IF Y < 0 THEN Y=0
100 IF Y > 79 THEN Y=79
110 IF STRIG(0)=0 THEN GOTO 10
120 GOTO 40

Let's examine this program:

Line 5 - is our REM statement to remind us what this
program produces.

Line 10 - is our GRAPHICS statement and permits us to
use mode 7.

Line 20 - is our COLOR statement and permits us to PLOT
with light green.

Line 30 - initializes our variables X and Y at 0.

Line 40 - is our PLOT statement.

Line 50 - changes the value of X to be the same as the
value of X plus the horizontal reading of the joystick.

Line 60 - changes the value of Y to be the same as the
value of Y plus the vertical reading of the joystick.

Line 70 - limits the value of X to numbers that are not
less than 0.

Line 80 - limits the value of X to numbers that are not
greater than 159.

line 90 - limits the value of Y to numbers that are not
less than 0.

Line 100 - limits the value of Y to numbers that are not
greater than 79.

Lines 70 - 100 are there to prevent ERROR 141. If the
value of X and/or Y go beyond the limits of graphics
mode 7 an error will result. The error will prevent the
program from continuing.

Line 110 - if the trigger is pushed, clears the screen
and starts the program again.

Line 120 - forces the computer to return to line 40 and
allows for the continuous execution of this program.

Exercises:
----------

1) Modify the above example so color is plotted only
when the trigger is pushed. (Remove line 110, which
clears the screen if the trigger is pushed, since
the two uses are incompatible.)

2) Change the program to simply show where the joystick
has "moved" to by "flashing" the spot which would
have been plotted (in the original program).

3) This is tricky: add to exercise 2 to allow the user
to change colors by pushing the joystick button.

CHAPTER XXIX
------------

              A Real Live Video Computer Game:
                      SNAILS' TRAILS

Well, the hard work is all done. If you have read and
understood all the previous chapters, you are ready for
a  break.  And,  rather  than  present  you  with  more
statements, functions, keywords, etc., we thought that a
game would be both instructive and fun. So here, in its
entirety,  is  the  BASIC  XL  version  of  a  perennially
popular video game.

Games similar to "SNAILS' TRAILS" have been around for a
long time. A version called "SURROUND" was one of the
first games available for the Atari 2688 game machine.
But, in the tradition of the video game industry, we
should present a story line:

You are a giant, mutant snail. Wherever you travel, you
leave a trail of radioactive slime behind. So poisonous
and impenetrable is this slime that should any being
(including you, yourself) touch it, it dies instantly.
(Yes, yes. If it's that poisonous, how could you lay
the trail in the first place? How should we know...YOU
are the mutant.)

Further,  the  scientists  of  far  off  H'tra-E  have
discovered your kind and have imprisoned you and another
of  your  race  in  a  large  rectangular  arena.
Unfortunately, both of you are neither male or female.
Instead,  you  are  each  a  St'i,  specially  bred  to  do
battle until death! You do not know the meaning of the
word P'ot's ("stop")!

So, as the scientists release you from stasis (you can
hear the three bells as the stasis field is lifted), you
begin by  charging straight  toward your  opponent. But
wait! A bit of intelligence enters your crazed brain.
If your slime trail is so deadly, perhaps you can entice
your enemy to run into your trail, thus killing st'i
without damage to yourself. Great strategy!

What's this,  though? Your  opponent has  developed the
same strategy. Now you and st'i must race around the
arena, with the strategic goal of forcing each other to
touch a poisonous trail or to run into the electrified
outer fence. (Well, we had to keep you in the arena
somehow, didn't we?) But tactics can be important here
as  well.  Look,  you  are  running  straight  across  the
arena. At the last second, you veer in front of your
enemy! He can't avoid your trail in time! He's going to
•••Oops. You forgot about the wall. Too bad. R.I.P.

To make a long story into a short game, you and another
human opponent must each use an Atari joystick (plugged
into joystick ports 1 and 2) to control your snai1. The
firet snail to run into a s1ime trail or a wall loses,
and the other enai1 scores a point. The first snai1 to
score 10 points wins the game. Good luck.

10 REM ---SNAILS' TRAILS---
20 REM
30 REM FROM THE MOVIE OF THE SAME NAME
40 REM
50 REM THIS GAME REQUIRES 2 JOYSTICKS PLUGGED INTO CONNECTORS
60 REM NUMBER 1 AND 2 (STICKS 0 AND 1)
70 REM
80 REM
90 FAST
100 REM ROUND INITIALIZATION
110 GRAPHICS 5
120 REM DO SETCOLORS HERE IF DESIRED
130 COLOR 1:PLOT 0,0:DRAWTO 0,39:DRAWTO 79,39
140 DRAWTO 79,0:DRAWTO 0,0
150 H0=20:V0=20:H1=20:V1=20
160 HMV0=1:VMV0=0:HMV1=-1:VMV1=0
170 COLOR 2:PLOT H0,V0:COLOR 3:PLOT H1,V1
180 PRINT "  SCORE":PRINT "PLAYER 1",,"PLAYER 2"
190 PRINT SCORE0,,SCORE1
200 REM START A ROUND
210 FOR FREQ =50 TO 150 STEP 50
220 FOR VOLUME=15 TO 0 STEP -0.1
230 SOUND 0,FREQ,10,VOLUME
240 NEXT VOLUME
250 NEXT FREQ
300 REM MAIN MOVE LOOP
310 FOR MOVE=1 TO 255 STEP 3
320 REM SENSE AND MOVE PLAYER 0
330 IF HSTICK(0) THEN HMV0=HSTICK(0):VMV0=0
340 IF VSTICK(0) THEN VMV0=VSTICK(0):HMV0=0
350 H0=H0+HMV0:V0=V0-VMV0
360 LOCATE H0,V0,BANG0:IF BANG0 THEN 400
370 COLOR 2:PLOT H0,V0
400 REM SENSE AND MOVE PLAYER 1
410 IF HSTICK(1) THEN HMV1=HSTICK(1):VMV1=0
420 IF VSTICK(1) THEN VMV1=STICK(1):HMV1=0
430 H1=H1+HMV1:V1=V1-VMV1
440 LOCATE H1,V1,BANG1:IF BANG1 THEN 500
450 COLOR 3:PLOT H1,V1
500 IF BANG0 OR BANG1 THEN 600
510 FOR VOLUME=14 TO 0 STEP -2
520 SOUND 0,MOVE,10,VOLUME
530 NEXT VOLUME
590 NEXT MOVE:GOTO 300
600 REM SOMEBODY GOT BANGED
610 IF BANG0 AND BANG1 THEN 100

620 IF BANG0 THEN SCORE1=SCORE1+1
630 IF BANG1 THEN SCORE0=SCORE0+1
640 FOR VOLUME=15 TO 0 STEP -0.25
650 IF BANG0 THEN SETCOLOR 1,4,VOLUME
660 IF BANG1 THEN SETCOLOR 2,4,VOLUME
670 SOUND 0,23,0,VOLUME
680 NEXT VOLUME
700 REM CHECK FOR END OF GAME
710 IF SCORE0<10 AND SCORE1<10 THEN 100
720 GRAPHICS 2
730 IF SCORE0>SCORE1 THEN 770
740 PRINT #6;"PLAYER 2 WINS:"
750 PRINT #6;SCORE1;" TO ";SCORE0
760 GOTO 800
770 PRINT #6;"PLAYER 1 WINS:"
780 PRINT #6; SCORE0;" TO ";SCORE1
800 REM END OF A GAME
810 PRINT "PUSH JOYSTICK BUTTON TO PLAY AGAIN"
820 REM WAIT FOR A BUTTON TO BE PUSHED
830 IF STRIG(0) AND STRIG(1) THEN 820
840 RUN

How SNAILS' TRAILS Works
------------------------
You don't have to understand how the program works to
enjoy SNAILS' TRAILS. Just type in the listing and RUN
it. But, if you believe you are ready to understand the
programming  techniques  involved,  read  on.  We  will
explain the program a line (or short group of lines) at
a time. The numbers in front of each paragraph below
indicate the lines which are being explained.

10-80   Just some REMarks, to remind us what the name of
the program is and to give some instructions for running
the game. These lines are unnecessary, but it is good
practice to always have some such similar comments.

90   Actually, we cheated. This is a new statement, not
introduced previously. FAST has the unique ability to
speed up your program considerably. May we suggest that
you leave out this line until the program is working
properly. Then simply add this line and RUN the game
again.  Look  at  how  much  faster  it  plays.  We  would
STRONGLY recommend that you ,do NOT use FAST in your own
programs until after you have read the description of
how it works in the BASIC XL REFERENCE MANUAL.

100   This REMark denotes the beginning of the real work
of each round (or turn) in the game. (Remember, the
first to win 10 rounds wins the game.)

110   We  chose  an  intermediate  pixel  size  for  our
display. This program will work in GRAPHICS 7 instead,
but you will have to adjust the values in lines 130
through 150.

l20   The colors  chosen by  default when  the GRAPHICS
statement  is  executed  seemed  adequate  to  us.  If  you
would like, though, you could place one or more SETCOLOR
statements on this line.

130-140   We  draw  a  line  around  the  outside  of  the
screen to define the "arena".

150   The H'tra-E scientists decided to start the snails
at the same positions each time: halfway down the arena
and about a quarter of the way in from each side. H0 and
V0  designate  the  starting  Horizontal  and  Vertical
position of the first player while H1 and V1 refer to
the second player. (We use 0 and 1 to correspond to
joysticks  0  and  1,  even  though  we  call  the  players
"first" and "second". The English language doesn't like
"zeroeth" and "oneth" very well.)

l60   Since the  fighting instinct  is strong  when the
snails are first released, they start moving directly
toward each other. The left hand snail (player) moves to
the right, as indicated by HMV0 (Horizontal Movement of
player 0) being set to plus one. The other player moves
left (HMV1 = -1). Neither snail moves vertically, so the
two vertical movements are set to zero.

NOTE: Lines 100 through 160 have for all intents
and purposes, defined the limits and scope of our
game. From here on out, the game simply follows
the rules we have defined.

Often, the hardest part about writing a program is
designing the initialization code. Consider, as an
example,  the  task  of  writing  a  version  of  the
classic PONG game.

The movement, per se, is easy: two paddles move up
and down the left and right edges of the screen and
a ball bounces between them. But consider how much
the decisions made at set up time affect the game:
How big is the ball? How big are the paddles?
Where  does  the  ball  start  from?  Where  do  the
paddles start? Are,there boundary walls? If so,
where?

When you design and write your own programs, try to
pay special attention" to the "start up" code. Any
extra  time  spent  on  this  phase  will  pay  you
dividends in the form of better looking and more
readable programs. And, who knows, your program
may even work the first time!

170 - We place the snails on their starting spots and
draw the first pixel of their slime trails. See how easy
this  was,  once  the  initialization  code  had  set  up
everything? Note that player 1 will use color 2 while
player 2 uses color 3. Any reason? Not really. Change
the colors if you like.

180-190 - We display the scores in the text window at the
bottom of the screen. The commas and spaces used in
these lines are not necessary, but they make the display
look better.

Note that we depend upon SCORE0 and SCORE1 being zero
when the program is RUN. This is a legitimate use of our
knowledge of how BASIC XL works.

200-250 - Just before the stasis field is lifted, three
bells ring. Remember?

210 and 250 - This  outer  loop  generates  three
FREOuencies. Do you see why they are 50, 100, and 150?
Do  you  remember  that  higher  numbers  imply  lower
frequencies when pure tones (sound quality = 10) are
used?

220-240 - This inner loop simply makes a tone using sound
register zero. Notice how the STEP of -0.1 implies that
the volume will take on values of 15.0, 14.9, l4.8, etc.,
in succession, until it reaches zero. When a pure tone
is  sounded  like  this,  a  rapidly  decreasing  volume
produces a ringing, bell-like note.

300-310 - The players are moved once each time we go
through this loop. For an explanation of why MOVE is
STEPped by 3, see the discussion of line 520. Also, see
the discussion of line 590 for more information about
this loop.

320-370 - This section of the code controls the movement
of the first player's snail.

330 - If the first joystick (number 0) is pushed left or
right, we change the horizontal movement (HMV0) of the
snail  to  correspond.  Since  we  don't  allow  diagonal
movement, we cancel any vertical movement there might
have been (we set VMV0 to zero).

340 - Similarly, if the joystick is pushed up or down, we
change  the  vertical  movement  (VMV0)  and  cancel  the
horizontal movement (set HMV0 to zero). Notice that this
implies that a stick pushed diagonally will produce only
vertical  motion.  If  you  would  rather  let  horizontal
movement have priority over vertical, simply swap the
order of lines 330 and 340.

350 - The most  important line  of the  player movement
code. The new player position (H0 and V0) is found by
adding the old position (also H0 and V0) to the requested
movements (HMV0 and VMV0).

NOTE  the  minus  sign  used  in  determining  the
vertical  position.  It  is  there  because  VTICK
returns a plus one when the stick is pushed up,
while the Atari graphics system insists that larger
numbers are further "down" the screen. This simple
little trick keeps everything running smoothly and
as the player(s) would expect.

360 - Remember that the pixel at H0 and V0 is where this
player's  snail  expects  to  move  to  next.  What  if
something is already at that spot, such as a wall or a
slime trail? Then the player must "die".

The LOCATE statement puts the color of the pixel at H0
and  V0  into  the  variable  BANG0.  If  nothing  already
exists  at  the  location,  the  color  will  be  0  (the
background color), and the program will fall through to
line  370.  If  something  exists,  though,  we  are  done
moving this player, so we go directly to line 400.

370 - If we get here, there was no conflicting object in
our pixel, so we lay down another piece of our slime
trail.

400-450 - This is the same code as lines 320 through 370
except that all the references are now to the second
player (H1, V1, etc.). We will not explain the code line
by line since it is virtually identical.

500 - If either player hit something, this must be the
end of a round. We exit the main code by simply GOing TO
line 600.

510-530 - If we get here, both players are still alive.
We make a little bell tone to indicate that we have
moved. Eliminating these lines will speed the game up
significantly. But it won't sound as good.

520 - The use of MOVE here needs a special comment. In
line 310, we began a FOR...NEXT loop involving MOVE:
        FOR MOVE = 1 to 255 STEP 3
But that only allows 85 moves! Surely we expect that
some games will last longer than that, don't we? Sure,
but we wanted to use the MOVE variable to determine the
frequency of the short bell tone, and SOUND only allows
frequency values from 0 through 255. By using the loop
shown, we are guaranteed legal sound values.

590 - And this, then, is the end of the MOVE loop. But
what happens if both players are still alive at the end
of 85 turns? Simple answer: GO back TO line 300 and
start the FOR...NEXT loop over again. Actually, if we
didn't need sounds, this line could consist of just the
GOTO and we could eliminate the FOR...NEXT entirely.

600-680 - When  one  or  both  of  the  players  runs  into
something, we execute this code.

610 - If both players hit at the same time, don't do
anything. Just go remove the stasis field again.

620-630 - Add a point to the score of the player that did
NOT hit anything.

640-680 - Another decreasing volume loop. This time, the
sound quality (line 670) is set to zero so we get noise.
The  effect  is  something  like  an  explosion.  650-660.
Depending on which player hit something, we flash one or
the  other  of  the  slime  trails  red.  Why  do  we  use
"VOLUME" for what is normally luminance? Simply because
it happened to be changing and in the range required for
luminance.  The  result:  a  bright  flash  which  quickly
fades out. Nice.

700-710 - If neither player has yet scored 10 points, we
start another round at line 100.

720 - If somebody lost, we might as well broadcast the
fact in great big letters. Remember GRAPHICS 2 produces
large characters in the graphics window and retains the
text window.

730 - What message is displayed depends on who won.

740-750 - The message, in nice big characters. Notice
the semicolon in 740 forces the first line to the left
edge of the screen. The comma in 750 places the scores
in  the  middle.  The  result  is  an  acceptable  score
display.

760 - We  bypass  the  other  message  by  GOing  TO  the
end-of-game code.

770-780 - The same code as lines 740 and 750, changing
who won.

800-810 - Just in case, we allow the H'tra-Ethlings to
start with more fresh snails. Note that the message in
line 810 is printed at the bottom of the screen, in the
text window.

820-830 - This is  actually a  very small  program loop
which will last forever. Unless, of course, one or the
other player pushes his/her joystick button.

840 - Yes, you really can use RUN as a program statement.
In situations such as this, it has the advantage of
forcing a clear of all variables, etc.

Of course, if we had wanted to do something like keep
track of high scores (inapplicable in this particular
game), we would have to use GOTO 10 here. The further
implication is that we could not assume any variables to
be zero when a game started. (See the comments about
line 190.)

Some Last Comments
------------------
Are you really still with us? If you got lost in that
explanation, perhaps the best thing to do would be to
type in the game and try it. May we strongly suggest
that you SAVE or CSAVE the program before RUNning it. If
you typed it in exactly as shown, no problems should
occur. But...

This kind of game is what programming is all about. Not
that everyone should write games. Rather that everyone
who  is  serious  about  programming  should  seek  a  well
defined programming goal, design the means to accomplish
it, and start coding. And whether a program plays a game
or calculates the residual value of fully depreciated
Edsels, if it does what it was designed to do, it is a
success.

Remember this, then, when you tackle exercise number 3,
below.

EXERCISES
---------

1. Change line 120 so that the program will draw a red
border around the arena.

2. As  this  game  is  written,  a  player  may  wipe
himself/herself  out  too  easily  by  backtracking  on
his/her own slime trail. Obviously, nobody intends to
do this, but sometimes when you turn fast you will
end up going diagonally and the program's logic will
force you back into yourself.

Try  to  "fix"  the  program  so  that  a  player  cannot
backtrack so easily. Fair warning: the answer which we
provide for this exercise still makes it possible - though
much harder - to backtrack. If you do it better, great!

3. Think of a problem which a computer can solve. Write
a program which will solve the problem.

The "problem" might be a game you would like to see. Or
it might be a loan amortization calculation. Or perhaps
a checkbook balancer. Don't limit your horizons.

If you really can't think of a problem you want solved,
try some of our suggestions:

Games:     TIC-TAC-TOE
           PONG
           CRAPS
               (A  very  easy   game  to  write  without
                graphics.  So why not put in some really
                good pictures?)

Business:  LOAN AMORTIZATION
           RATE OF RETURN ON INVESTMENT
           INVOICE PRINTING
                (Filing invoices is MUCH more difficult.
                 This book  has NOT given you sufficient
                 background to write such a program.)

Home use:  CHECKBOOK BALANCER
           RECIPE MULTIPLIER
                (What is 6 times 1/3 cup - in quarts?)

Education: ADDITION DRILL AND PRACTICE
                (Also very easy without graphics.
                 But how about showing 3 red apples
                 plus 4 orange oranges?)

CHAPTER XXX
-----------

Congratulations! You've made it to the final chapter of
our book. Good for you! It took patience, and an open
mind and hard work to get this far. Now what? What have
you learned? What good does it do you?

In the process of working through this book, you have
learned to operate your ATARI HOME COMPUTER. That and a
dollar will buy you a cup of coffee. Right? Actually,
there is more to it than that.

By learning to operate your ATARI HOME COMPUTER, you
need never fear computers again. You now know that a
computer will not byte, and in fact, the computer is
always READY for your next command.

You  have  also  opened  the  door  to  many  other
possibilities. Your familiarity with the computer and
simple  BASIC  statements  will  allow  you  to  use
commercially produced software with a minimal amount of
effort. With each new day more software programs come
on the market. You can increase your use of your home
computer  and  at  the  same  time  learn  a  new  foreign
language like French or Italian. Or you can choose from
word processing, home management, home phone lists, and
address label programs. Actually, the possibilities are
almost endless.

At social gatherings, you can be· part of the group
discussing  programming.  Or,  you  can  show  your
brilliance  by  word  dropping.  For  example:  you  could
tell  your  friends  about  favorite  variables:  or  how
IF...THEN  and  GOSUB...RETURN  statements  have  changed
your life. Why, you'll be the center of attention.

Maybe what you have learned is that you aren't meant to
be a programmer at all. Knowing that isn't a waste of
time.  Now  that  you  know,  you  again  can  rely  on
commercially  produced  software  for  all  your  needs.
Believe  it  or  not,  not  everyone  was  intended  to  be
programmer. However, everyone should know enough to be
computer  literate.·Assuming  you  have  read  and
understood  most  of  this  book,  you  are  now  computer
literate.

Finally, if all this programming stuff really got to
you, if night after night you spent hours with our book
and your ATARI, you may have a future in programming.
There  is  more  BASIC  to  learn  and  other  computer
languages.

--163--



One  final  note:  if  you  want  to  pursue  a  career  in
programming  or  if  you  want  to  become  a  better
programmer, there are two things you should do.

First, start reading. Begin by reading your BASIC XL
Reference Manual. This will give you information about
computer  functions  and  commands  not  covered  in  this
book. There are many other books on the market that
will provide  additional information  on BASIC.  If you
don't  have  an  excellent  mathematics  background,
carefully check the books you use. Be sure each book
deals with functions and commands and not just higher
levels of math. Don't buy a book sight unseen.

Second, start programming. Don't worry about what you
programs you write or whether they are useful or not.
Any  programming  experience  will  be  valuable.  Also,
there are many magazines and user groups that publish
programs for you to try. The better publications also
include explanations of their programs. Programming is
like tennis or golf or music or any other learned skill,
the more you do it, the better you'll get at it.

ANSWERS
-------

Answers: Chapter I
------------------

1.  PRINT "JEFFREY R. SMITH"

2.  PRINT "GIVE ME LIBERTY OR GIVE ME DEATH"

Please note: the above answers are correct. However, 
your screen will look like this:

PRINT "JEFFREY R. SMITH"
JEFFREY R. SMITH

READY
PRINT "GIVE ME LIBERTY OR GIVE ME DEATH"
GIVE ME LIBERTY OR GIVE ME DEATH

Answers: Chapter II
-------------------

1.  PRINT "JOHN Q. PUBLIC"
    PRINT "924 MAIN STREET, ANYWHERE, USA"
    PRINT "8/1lB-555-1212"

2.  PRINT 15+25
    PRINT 38-14
    PRINT 680*12
    PRINT 25/5

3.  PRINT "000-11-2222"

Answers: Chapter III
--------------------

1)  a)  Assigns the value of 15 to X.

    b)  Assigns the value of 144 to B.

    c)  Assigns the value of 338 to F2.

    d)  Assigns the value of 323 to Z.
        (which is 338-15)

2)  LET TOTAL = 7 + 5
    PRINT TOTAL + 1

    (N.B.: You could have coded PRINT 7 + 5 + 1, but the
    answer  shown  keeps  better  to  the  spirit  of  this
    chapter.)

3)  198

Answers: Chapter IV
-------------------

1)  10 LET X = 2
    20 LET Y = 4
    30 PRINT X + Y
    40 PRINT X * Y

When you typed RUN, did your program work? Good. If
not compare your answer to the one given. It is a good
idea  to  get  in  the  habit  of  testing  your  programs.
Really,  that  is  the  only  way  to  know  if  they  work
properly.

Since variable names, the value of a variable and line
numbers chosen in a program are arbitrary, your program
is probably correct if it is similar to the one above,
and if it RUNs properly.

2)  Assuming the same program is used as was used in the
    answer to exercise 1:

    10 LET X•= 2
    30 PRINT X+Y
    40 PRINT X*Y

    OR

    10 LET X = 2
    20 PRINT X+Y
    30 PRINT X*Y

Did you correctly note that the computer assumes Y has a
value of zero? In Basic XL, all variables have a value
of zero until something else is assigned to them.

3)  You should have typed in a single line, such as:
    20 LET Y = 4 OR 15 LET Y = 4

The important point here is that you used a line number
somewhere between the line numbers for the first and
third lines. When you RUN the program, you should get
the same results as you did in exercise 1.

--166--



Answers: Chapter V
------------------

You Type:                         Computer Responds:
---------                         ------------------

1)  NEW
    10 LET X = 12
    20 LET Y = 8
    30 PRINT X ; Y
    RUN
                                     128

2)  30 PRINT X,Y
    RUN
                                     12     8

3)  30 PRINT X+Y,X-Y,X*Y
    RUN
                                     20     4     96

PLEASE  NOTE:  Since  variables  names,  the  value  of  a
variable  and  line  numbers  chosen  in  a  program  are
arbitrary, your programs are probably correct if they
look similar to the answers shown. Be sure to check that
your programs RUN properly.

ANSWERS: Chapter VI
-------------------

1)  10 PRINT "N", "N-SQUARED", "N-DOUBLED"
    20 LET N = 1
    30 PRINT N, N*N, N+N
    40 LET N=N + 1
    50 LET N = X
    60 GOTO 30

2)  10 LET X = 1
    20 PRINT X
    30 LET X = X + 1
    40 GOTO 20

3)  10 LET X = 1
    20 PRINT "YOUR NAME",X
             or
    20 PRINT "YOUR NAME";X
    30 LET X = X + 1
    40 GOTO 20

4)  10 LET X = 1
    20 PRINT "YOUR NAME",X
    30 LET X = X + 2
    40 GOTO 20

5)  10 LET X = 5
    20 PRINT "YOUR NAME", X
    30 LET X = X + 2
    40 GOTO 20

RUN each of your programs.

If your programs look similar to the ones above and if
they RUN properly, they are probably correct.

Answers: Chapter VII
--------------------

1)  10 LET C = 5
    20 PRINT C
    30 LET C = C + 5
    40 IF C <= 100 THEN GOTO 20

2)  10 LET T = 1
    20 PRINT "A COUNTING COMPUTER IS EASY"
    30 LET T = T + 1
    40 IF T <= 10 THEN GOTO 20

3)  10 LET K = 937
    20 LET M • 486
    30 IF K = M THEN PRINT "THEY ARE EQUAL"
    40 IF M>K THEN PRINT M;" IS LARGER"
    50 IF K>M THEN PRINT X;" IS LARGER"

If your programs look similar to the ones above and if
they RUN properly, they are probably correct.

Answers: Chapter VIII
---------------------

1)  10 PRINT "WHAT IS THE BASE"
    20 INPUT BASE
    30 PRINT "WHAT IS THE HEIGHT"
    40 INPUT HEIGHT
    50 PRINT "AREA = ";BASE*HEIGHT/2

2)  10 PRINT "ENTER ONE TEST SCORE"
    20 INPUT TEST1
    30 PRINT "ENTER ANOTHER TEST SCORE"
    40 INPUT TEST2
    50 PRINT "ENTER ANOTHER TEST SCORE"
    60 INPUT TEST3
    70 LET TOTAL = TEST1+TEST2+TEST3
    80 LET AVERAGE=TOTAL/3
    90 PRINT "THE AVERAGE OF THE THREE TEST SCORES IS ";
       AVERAGE

3)  10 PRINT "WHAT IS THE MAXIMUM NUMBER"
    20 INPUT MAX
    30 LET NUM = 1
    40 PRINT NUM, NUM*NUM
    50LET NUM = NUM + 1
    60 IF NUM <= MAX THEN GOTO 40

Answers: Chapter IX
-------------------

1)  10 PRINT "ENTER YOUR BOWLING SCORES"
    20 INPUT YOU1, YOU2
    30 PRINT "ENTER YOUR OPPONENT'S BOWLING SCORES"
    40 INPUT OPP1, OPP2
    50 IF YOU1>OPP1 AND YOU2>OPP2 THEN PRINT "YOU WON
       BOTH GAMES"
    60 IF YOU1<OPP1 AND YOU2<OPP2 THEN PRINT "YOU LOST
       BOTH GAMES"

2)  10 PRINT "PICK A NUMBER"
    20 INPUT X
    30 IF X>10 AND X<20 THEN PRINT "YOUR NUMBER IS
       BETWEEN 10 AND 20"

3)  10 PRINT "GIVE ME 3 NUMBERS, IN ORDER"
    20 INPUT NUM1
    30 INPUT NUM2
    40 INPUT NUM3
    50 IF NUM1 < NUM2 AND NUM2 < NUM3 THEN PRINT
       NUM1,NUM2,NUM3

Answers: Chapter X
------------------

1)  10 LET D = 1
    20 PRINT RANDOM(1,50)
    30 LET D = D + 1
    40 IF D <= 10 GOTO 20

2)  10 LET D = 1
    20 PRINT RANDOM(1000)
    30 LET D = D + I
    40 IF D<= 10 GOTO 20

3)  10 LET NUMBER = RANDOM(1,50)
    20 PRINT "PICK A NUMBER BETWEEN 1 AND 50";
    30 INPUT GUESS
    40 PRINT GUESS, NUMBER

4)  10 LET NUMBER = RANDOM(1,50)
    20 PRINT "PICK A NUMBER BETWEEN 1 AND 50";
    30 INPUT GUESS
    40 IF GUESS<NUMBER THEN PRINT "TOO LOW"
    50 IF GUESS>NUMBER THEN PRINT "TOO HIGH"
    60 IF GUESS<>NUMBER THEN GOTO 20
    70 PRINT "YOU GOT IT"

------------------------------------------
| Note:   If   you   solved  exercise  4 |
| correctly,   congratulations   are  in |
| order.  This is  the classic "computer |
| number guessing game" and is often the |
| target   or   end   of    introductory |
| programming classes.                   |
------------------------------------------

Answer: Chapter XI
------------------

1) If you are able to copy a program and to retrieve
it, you probably understand the use of the program
recorder.

Answers: Chapter XII
--------------------

1)  If not, try again.

2)  ERROR -NOT SAVED FILE

3)  ERROR (unpredictable, but often 137)

Answers: Chapter XIII
---------------------

1) Obviously, the contents of your letter will differ,
but here's an example of proper form. Note how we
used commas successfully but did NOT use any at the
ends of lines.

10 LPRINT "DEAR JOHN,"
20 LPRINT "HOW ARE YOU?"
30 LPRINT "I'M SORRY I HAVE WRITTEN SOONER."
40 LPRINT "RECENTLY, I HAVE BEEN SPENDING MY SPARE
   TIME LEARNING TO PROGRAM MY ATARI."
50 LPRINT "I WILL WRITE YOU A LONG LETTER SOON."
60 LPRINT,," YOUR FRIEND,"
70 LPRINT,," JIM"

2) By  using  LIST"P:",  the  program  you  typed  in
should appear on the printer.
3)

Answers: Chapter XIV
--------------------

1)  10 GRAPHICS 7
    20 PLOT RANDOM(0,159), RANDOM(0,79)
    30 FOR I = 1 TO 25
    40 COLOR RANDOM(1,3)
    50 DRAWTO RANDOM(0,159), RANDOM(0,79)
    60 NEXT I
    70 GOTO 10

2)  10 GRAPHICS 7
    20 FOR I = 1 TO 15
    30 COLOR RANDOM(1,3)
    40 COL1 = RANDOM(160): COL2 = RANDOM(160)
    50 ROW1 = RANDOM(80): ROW 2 = RANDOM(80)
    60 PLOT COL1, ROW1 : DRAWTO COL1, ROW2
    70 DRAWTO COL2, ROW2: DRAWTO COL2, ROW1
    80 DRAWTO COL1, ROW1
    90 NEXT I
    100 GOTO 10

3)  10 GRAPHICS 7 : PLOT 0,0
    20 FOR X = 0 TO 39 STEP 2
    30 COLOR RANDOM(1,3)
    40 DRAWTO 159-X,X
    50 DRAWTO l59-X,79-X
    60 DRAWTO X,79-X: DRAWTO X,X+2
    70 NEXT X
    80 GOTO 10

----------------------------------------
| NOTE:  By  now,  we  are  presenting |
| exercises   so  complex  that  many, |
| many  different  "answers"  are pos- |
| sible.  We  would  suggest  that you |
| type in  our  programs only  if  you |
| cannot solve  the exercises yourself |
| or  if you simply  want to see  what |
| kind  of  result  we  expected.  The |
| rule here is simple: If your program |
| works, it is probably right.         |
----------------------------------------

Answers: Chapter XVI
--------------------

1)  10 PRINT "ENTER TOTAL WEEKLY SALES"
    20 INPUT SALES
    30 IF SALES >=20000 THEN LET PAY=2000
    40 IF SALES <20000 THEN LET PAY=1000
    50 PRINT PAY

2)  Add the following 1ines:
    30 IF SALES>=20000 THEN LET PAY=2000
    50 LET FEDTAX=105 : 0.067*PAY
    60 LET NETPAY=PAY-FEDTAX-SOCSECRTY
    70 PRINT NETPAY

ANSWER: Chapter XVII
--------------------

1)  150 PRINT "ONE PINT EQUALS 1) 8 OZS. 2) 16 OZS.
        3) 24 OZS"
    160 LET CORRECTANSR = 2
    170 GOSUB 400
    180 PRINT "ONE GALLON = 1) 4 QUARTS 2) 4 CUPS
        3) 4 PINTS"
    190 LET CORRECTANSR = 1
    200 GOSUB 400

If your answers are similar in format to the ones above
and if your program runs properly then your program is
probably correct.

Answers. Chapter XVIII
----------------------

1)  10 FOR I = I TO 10
    20 PRINT "JOHN SMITH"
    30 NEXT I

2)  10 FOR X = 10 TO 1 STEP -1
    20 PRINT X
    30 NEXT X
    40 PRINT "BLAST OFF"

3)  10 FOR G = 0 TO 100 STEP 5
    20 PRINT G
    3  NEXT G

4)  10 PRINT "HOW FAR SHOULD I GO";
    20 INPUT MAX
    30 FOR N=1 TO MAX
    40 PRINT N, N*N
    50 NEXT N

Answers. Chapter XIX
--------------------

1)  10 FOR I=1 TO 9
    20   FOR J=1 TO 9
    30   PRINT I*J;
    40   PRINT"  ";
    50   NEXT J
    60 PRINT
    70 PRINT
    80 NEXT I

Answers: Chapter XX
-------------------

1)  10 INPUT INVITEE$
    20 PRINT "DEAR    "; INVITEE
    30 PRINT "ON JUNE 15TH I WILL BE 39 ONE MORE TIME."
    40 PRINT "PLEASE   ";
    50 PRINT INVITEE$;
    60 PRINT "HELP ME CELEBRATE THIS OCCASION."
    70 PRINT "MY PARTY WILL BE HELD AT THE GREEN OAKS
       COUNTRY CLUB."
    80 PRINT "BE PROMPT";
    90 PRINT INVITEE$;
    100 PRINT "THE PARTY STARTS AT 8:00."
    110 PRINT "I'LL NEED HELP BLOWING OUT THE CANDLES."
    120 PRINT
    130 PRINT "SINCERELY,"
    140 PRINT "JACK BENNY"
    150 PRINT
    160 GOTO 10

2)  10 DIM ANSWER$(1)
    20 PRINT "DO YOU LIKE ME?";
    30 INPUT ANSWER$
    40 IF ANSWER$="Y" THEN PRINT "I LIKE YOU, TOO"
    50 IF ANSWER$="N" THEN PRINT "BRRRRAP!"

Answer: Chapter XXI
-------------------

1)  10 DIM NAME$(35)
    20 PRINT "ENTER YOUR NAME"
    30 INPUT NAME$
    40 LET A= LEN(NAME$)
    50 FOR I=A TO 1 STEP -1
    60 PRINT NAME$(I,I);
    70 NEXT I

ANSWERS: Chapter XXII
----------------------

1)  10 SOUND RANDOM(4),RANDOM(256),RANDOM(16),RANDOM(16)
    20 FOR I=1 TO 200: NEXT I
    30 GOTO 10

2)  10 P=RANDOM(256)
    20 SOUND 0,P,10,8
    30 FOR I=1 TO 10 : NEXTI
    40 GOTO 10

Answers: Chapter XXIII
----------------------

1)  10 GRAPHICS 1
    20 PRINT #6; "STEVEN LEWIS"
    30 PRINT #6; "YOU ARE A STAR"

2)  10 GRAPHICS 2
    20 PRINT #6; "     THE PALACE     ";
    30 POSITION 0,2
    40 PRINT #6; "********************";
    50 PRINT #6; "*     jonathon     *";
    60 PRINT #6; "*     michaels     *";
    100 PRINT #6; "*                  *";
    110 PRINT #6; "*  5 performances  *";
    120 PRINT #6; "*      a   day     *";
    130 PRINT #6; "*       LIVE       *";
    140 PRINT #6; "********************";

Answers: Chapter XXIV
---------------------

1)  10 GRAPHICS 3

    10 GRAPHICS 5

The above program line should replace the first line in
each of the three exercises listed at the end of Chapter
XIV. First, RUN the program in mode 7 and then change to
mode 3 and then to mode 5.

2)  100 COLOR 3-SCREENCOLOR
    110 PLOT COLUMN,ROW

Answers: Chapter XXV
--------------------

1)  10 SETCOLOR 1, RANDOM(16), RANDOM(16)
    20 FOR WAIT=1 TO 1000: NEXT WAIT
    30 GOTO 10

2)  10 GRAPHICS 3
    20 SETCOLOR 4, RANDOM(16), RANDOM(16)
    30 FOR WAIT=1 TO 1000: NEXT WAIT
    40 GOTO 20

Line  20  in  answer  1  and  line  30  in  answer  2  are
arbitrary  delays  and  do  not  affect  the  correctness
of  the  answer.  Without  some  such  delay,  though,  the
display color will flicker too fast to be discernible.

Answer: Chapter XXVI
--------------------

1)  10 GRAPHICS 8
    20 COLOR 1
    30 PLOT 0,0
    40 DRAWTO 159,0
    50 DRAWTO 159,79
    60 DRAWTO 0,79
    70 DRAWTO 0,0

Answers: Chapter XXVIII
-----------------------

1)  40 IF STRIG(0)=0 THEN PLOT X,Y
    110 {RETURN} (deletes line 110)

2)  40 COLOR 1: PLOT X,Y
    46 FOR I=1 TO 200: NEXT I
           (The above line is just a delay loop and is
            not strictly necessary.)
    48 COLOR 0: PLOT X,Y

3)  20 C=1
    40 IF STRIG(0)=0 THEN C=C+1
    42 IF C>3 THEN C=0
    44 COLOR C: PLOT X,Y

ANSWERS: Chapter XXIX
---------------------

1)  Change line 120 to read:
        120 SETCOLOR 0,4,4

The hue and luminance numbers ("4,4") might need to be
changed  slightly  for  different  television  sets  or
monitors.

Did you remember that SETCOLOR 0 applied to COLOR 1,
which is used to draw the border?

2) Change the following lines as shown:
       330 IF HSTICK(0)<>0 AND HMV0=0 THEN
           HMV0=HSTICK(0) : VMV0=0
       340 IF VSTICK(0)<>0 AND VMV0=0 THEN
           VMV0=VSTICK(0) : HMV0=0
       410 IF HSTICK(1)<>0 AND HMV1=0 THEN
           HMV1=HSTICK(1) : VMV1=0
       420 IF VSTICK(1)<>0 AND VMV1-0 THEN
           VMV1=VSTICK(1) : HMV1=0

Do you see Why those changes work? Basically, we can
read line 33" to say ..if the first user is pushing the
stick left or right AND if he was NOT moving left or
right before now, then change his movement to be left or
right (according to how the stick is pushed) and cancel
his vertical movement." The other lines may be read in
a similar fashion.

How does this deceptively simple change work? Well, if
the user's snail was going right (for example), then
pushing  the  joystick  left  OR  right  has  no  effect.
Reasons: if he is already going right, why do anything;
if he is going left, we don't want him to run into
himself  (the  conditions  of  this  exercise),  so  we
disallow the movement.

The  flaw  in  this  logic  is  complex.  If  the  snail  is
moving up and the user pushes the joystick diagonally
down and to the left (for example), our program will see
and allow the left movement first (HMV0 will be set to
-1 and VMV0 will be set to 0 in line 330). But then
when the program reaches the next line, it will also see
and allow the down movement (incidentally canceling the
left movement), and our snail will seem to double back
on  himself.  Note,  however,  that  as  this  program  is
written the flaw does not exist for horizontal movement.

Can  the  flaw  be  fixed?  Yes,  at  the  expense  of  more
complex logic, resulting in a program which runs slower.
Is it worth fixing? Probably not, since the game is very
playable as is.

A final comment. There were several possible answers to
this exercise, so if your version worked it is probably
correct. We present here several possible choices for
line 330 (the other lines would be similar).

        330 IF HMV0=0 AND HSTICK(0)<>0 THEN
                        HMV0=HSTICK(0) : VMV0=0
        330 IF HSTICK(0) AND NOT HMV0 THEN
                        HMV0=HSTICK(0) : VMV0=0
        330 IF HSTICK(0) THEN IF NOT HMV0 THEN
                        HMV0=HSTICK(0) : VMV0=0
        330 IF NOT HMV0 THEN IF HSTICK(0) THEN
                        HMV0=HSTICK(0) : VMV0=0

Incidentally, the last two examples will probably run a
little faster than the others, since the second function
will not even be checked unless the first expression is
true, thanks to the extra THEN we used.

Get Introduced To Programming:  The fundamental 
concepts of programming a computer are explained in this book 
using BASIC XL as the programming language. 

Start And Finish:  To be familiar with the basic handling of 
your Atari computer is the prerequisite of success when working 
through this book. The goal is your ability to create elementary 
programs in BASIC XL but also a better understanding of your 
Atari computer.

Future Outlook:  Once you have mastered BASIC XL you 
might want to explore on your Atari computer more powerful 
programming languages like Action! or MAC/65.

About The Book:  This is a nice and easy to read learning 
book to teach yourself about Atari computers and programming 
in BASIC. It is a good example how to attract people to make 
more use of their home computer.

                                                  -- GoodByteXL

BASIC XL

The Right Tool For
All Atari® BASIC Programming

